

JIM GREGORY

LIVRO DE JOGOS EM
MSX



EDITORAL LUTÉCIA

LIVRO DE JOGOS
EM
MSX

Jim Gregory

**LIVRO DE JOGOS
EM
MSX**

Tradução e adaptação de
RONALDO SERGIO DE BIASI

EDITORAL LUTÉCIA

Título original inglês
THE MSX GAMES BOOK

Copyright © 1985 by Jim Gregory

Direitos de publicação exclusiva em língua portuguesa no Brasil
adquiridos pela
EDITORIA LUTÉCIA LTDA
Rua Argentina 171 — 20921 Rio de Janeiro, RJ — Tel.: 580-3668
que se reserva a propriedade literária desta tradução

Impresso no Brasil

ISBN 85-19-23813-0

PEDIDOS PELO REEMBOLSO POSTAL
Caixa Postal 23.052 — Rio de Janeiro, RJ — 20922

Sumário

<i>Prefácio</i>	9
<i>Introdução</i>	11
<i>Técnicas Usadas nos Programas</i>	17
Jogos de Texto	
1 Marie Celeste: <i>Um jogo de aventura inspirado em um famoso navio-fantasma.</i>	19
2 Diga, Mestre: <i>Um jogo de adivinhação em que as respostas só podem ser “sim” ou “não”.</i>	33
3 Eu Acuso: <i>Um jogo de detetive no qual você tenta descobrir quem é o criminoso.</i>	40
Jogos Clássicos	
4 Star Trek: <i>Enfrente os perigosos klingons em uma batalha de vida ou morte.</i>	45
5 Super Breakout: <i>O conhecido jogo de breakout com uma opção para o tamanho da raquete.</i>	56
6 Labirinto: <i>Tente encontrar a saída de um labirinto tridimensional a tempo de escapar de um louco.</i>	60
Jogos para Duas Pessoas	
7 Damas: <i>Um tabuleiro de damas com verificação dos lances.</i>	67
8 Xadrez: <i>Um tabuleiro de xadrez com verificação dos lances e opção para salvar a partida em fita ou em disquete.</i>	77
9 Gustavus: <i>Um jogo tradicional de lápis e papel adaptado para o computador.</i>	97

Jogos de Dados

- | | |
|--|-----|
| 10 Corrida de Dados: <i>Uma corrida em que a velocidade dos competidores depende dos pontos dos dados.</i> | 102 |
| 11 Olhos de Cobra: <i>Um jogo de dados simples e divertido, para duas a quatro pessoas.</i> | 107 |
| 12 Craps: <i>O mesmo jogo dos cassinos de Las Vegas no seu microcomputador.</i> | 110 |

Jogos de Cartas

- | | |
|---|-----|
| 13 Vinte-e-Um: <i>O computador dá as cartas e os jogadores fazem apostas, tentando quebrar a banca.</i> | 115 |
| 14 Maior-Menor: <i>A próxima carta será maior ou menor que a anterior? Arrisque a sorte e poderá ganhar uma fortuna em microcruzados.</i> | 129 |
| 15 Memória: <i>Dois jogadores procuram encontrar o maior número possível de pares de cartas.</i> | 137 |

Jogos de Raciocínio

- | | |
|--|-----|
| 16 Senha: <i>Use as pistas fornecidas pelo computador para descobrir quais são as cores escondidas.</i> | 145 |
| 17 Alberto Diz: <i>Obedeça às instruções para movimentar os bonecos na tela...mas só quando quem estiver mandando for o Alberto!</i> | 151 |
| 18 Papagaios: <i>Repita a seqüência de notas musicais produzidas por quatro papagaios.</i> | 156 |

Jogos de Fliperama

- | | |
|--|-----|
| 19 Pic Man: <i>Faça o herói percorrer um labirinto, comendo biscoitos.</i> | 162 |
| 20 Frigger: <i>Ajude um canguru a atravessar uma estrada movimentada e um rio cheio de perigos.</i> | 169 |
| 21 Dinkey Kong: <i>Para salvar a namorada, o herói tem que subir até o topo do edifício, evitando os obstáculos que aparecem no caminho.</i> | 174 |

Jogos Educativos

- | | |
|--|-----|
| 22 Banho de Palavras: <i>Descubra qual é a palavra a tempo de evitar um caldo.</i> | 181 |
| 23 Aritmética: <i>Testes das quatro operações, com diferentes graus de dificuldade.</i> | 185 |
| 24 Hoje, o Brasil: <i>Um teste de geografia do Brasil, que pode ser estendido a outros países.</i> | 190 |

Programas Utilitários	
25 Gerador de Caracteres: <i>Desenhe seus próprios caracteres e salve-os em fita para uso futuro.</i>	199
26 Gerador de Figuras: <i>Use uma caneta simulada para desenhar na tela.</i>	211
27 Monitor da Memória: <i>Observe o conteúdo da RAM, modifique-o e verifique se entrou corretamente com todas as instruções de um programa.</i>	216
Observações Finais	221
Apêndices	
<i>Apêndice 1: Gráficos para Desenhar Sprites e Caracteres</i>	224
<i>Apêndice 2: Gráfico das Coordenadas da Tela</i>	225
<i>Apêndice 3: Mapa da Memória do MSX</i>	226
<i>Apêndice 4: Tabela de Cores do MSX</i>	227

Prefácio

Um livro como este só se tornou possível graças à existência do Microsoft Extended Basic ou MSX.

Em virtude da compatibilidade entre os micros MSX, o dono de qualquer micro dessa família poderá carregar os programas deste livro no seu microcomputador e entreter-se com eles.

Os jogos para microcomputadores não são apenas divertidos, mas também educativos. Em primeiro lugar, o desejo de ver o jogo "pronto" e funcionando desperta o interesse pelo computador. Em seguida, ao entrar com o programa, o usuário se familiariza com a sintaxe da linguagem em que o programa foi escrito. Além disso, como os programas raramente funcionam da primeira vez, devido a erros de digitação, torna-se necessário acompanhar a lógica do programa e assim descobrir como ele funciona. Os dois últimos estágios na evolução de um programador consistem em modificar programas já existentes e finalmente poder afirmar: "Eu sou capaz de fazer um programa melhor do que este!". Espero sinceramente que, depois de usar este livro, você seja capaz de desenvolver programas em BASIC MSX ainda melhores do que os apresentados aqui.

Entretanto, o objetivo principal desta obra não é ensinar ou explicar. A intenção foi produzir bons programas, que custariam dez vezes o preço do livro se fossem vendidos separadamente em fita cassete. Este é um livro de jogos e trabalhamos arduamente para incluir jogos de vários tipos, que interessassem a todos os leitores.

Infelizmente, embora o BASIC MSX seja uma linguagem rápida e que dispõe de comandos muito convenientes para programar jogos, o BASIC às vezes necessita de um pouquinho de ajuda. Por isso, incluímos pequenas rotinas em linguagem de máquina em alguns programas. Isto não deve ser motivo de preocupação, já que as rotinas são digitadas como parte da listagem. O resultado é um programa híbrido de BASIC e

linguagem de máquina que atende ao nosso objetivo principal, ou seja, oferecer ao leitor jogos agradáveis e interessantes.

O desenvolvimento dos programas apresentados neste livro foi um trabalho de equipe que exigiu um grande esforço e dedicação. Tenho a satisfação de registrar meus agradecimentos e apreciação a todos que trabalharam comigo neste projeto. Aqueles que participaram diretamente da elaboração dos programas, e cujos pseudônimos aparecem nas listagens, são “ISSI”, “PAULG”, “ANDY” e “BOOTSY”.

Os outros membros do grupo foram Janet e minha esposa Val, que se encarregaram da preparação do original através das sucessivas revisões.

Finalmente, meus agradecimentos a Richard Miles, da editora Collins, que encomendou o livro e se encarregou de todas as etapas de sua preparação até chegar a você — a razão final de todo o processo — a pessoa a quem agradeço por comprá-lo.

Jim Gregory

Introdução

Os programas aqui apresentados foram escritos especialmente para este livro, o que resultou em uma série de programas que cobrem todos os tipos de jogos. Esses jogos ilustram muitas técnicas úteis de programação.

Para ajudar a reduzir os erros de digitação, todas as listagens foram impressas com uma largura-padrão de 39 colunas. Isso quer dizer que se você usar SCREENØ para carregar os programas, cada linha na tela corresponderá exatamente a uma linha no livro; este é apenas um exemplo do esforço que fizemos para assegurar que todos os programas deste livro funcionem corretamente.

Depois de serem exaustivamente testados, os programas foram listados diretamente a partir da memória do computador. Assim, caso um dos programas não funcione, o mais provável é que o leitor tenha cometido um erro na hora de digitá-lo.

Falando com a experiência de quem já carregou muitos programas a partir de listagens publicadas em livros e revistas, sei que haverá ocasiões em que você será capaz de jurar que copiou o programa corretamente, embora ele não esteja funcionando. Leia com atenção as notas que se seguem e cometerá muito menos erros de digitação. Procure seguir os conselhos à risca e verá que eles lhe pouparão tempo e trabalho.

Alguns conselhos úteis

1. Instale o equipamento sobre uma mesa de tamanho adequado, sente-se em uma cadeira confortável e procure manter as costas retas. Assegure-se de que a iluminação é adequada e coloque o livro em uma posição conveniente para a leitura.

2. Antes de começar a digitar, certifique-se de que todos os equipamentos estão ligados.
3. Certifique-se de que dispõe de um suprimento adequado de fitas ou de um disquete formatado.
4. Se as suas fitas têm um trecho não-magnético no início, prepare-as para a gravação dos programas, fazendo girar o carretel até a parte castanha aparecer.
5. Para certificar-se de que tudo está em ordem, coloque no gravador uma fita C90, que será sua fita de trabalho, e execute um teste de "salvar e carregar", digitando

```
10 REM 'TESTE'  
20 PRINT "TUDO BEM"  
30 GOTO 10
```

ou um programa semelhante.

Salve o programa de teste, depois carregue-o e execute-o, entrando com RUN no modo direto. Se o programa funcionar corretamente, não torne a enrolar a fita; continue de onde ela parou. Se o programa não funcionar, rebobine a fita e comece de novo, verificando com atenção todos os passos. Você então poderá usar esta fita para salvar seu programa, seja quando terminar ou quando quiser interromper a digitação para continuar mais tarde.

Experimente primeiro usar a maior velocidade de gravação do MSX para verificar se ela é confiável para a combinação de computador, gravador e fita que você está utilizando. Caso os resultados não sejam satisfatórios, use a velocidade menor.

Se estiver trabalhando com disquetes, faça um teste semelhante, salvando e tornando a carregar um programa em disquete.

6. Quando salvar apenas uma parte do programa, use um nome apropriado, como "V1", "V2" etc. Se preferir, use a hora e a data em que o programa foi salvo, como por exemplo "8:15-7/1/87".

Depois de salvar um programa, não se esqueça de anotar o nome do programa e o número do contador; assim, poderá localizá-lo rapidamente no futuro.

7. Não deixe de salvar o programa em fita ou disquete *antes* de entrar com RUN. O mundo dos micros está cheio de gente apressada que per-

deu muitas horas de trabalho porque se esqueceu de tomar esta precaução elementar.

Existem várias razões pelas quais um programa pode ser perdido enquanto está sendo executado. Apenas para dar alguns exemplos, pode haver um erro de digitação tal que o interpretador execute a instrução NEW, ou pode ser que o programa entre em um loop para o qual não haja saída a não ser desligar o computador. Os programas que utilizam rotinas em linguagem de máquina, instruções POKE ou instruções para o sistema operacional estão particularmente sujeitos a erros irreversíveis.

8. Depois que o programa estiver funcionando corretamente, é recomendável guardar uma cópia de reserva em uma fita C90, juntamente com outros programas. A cópia principal deve ser guardada em uma fita mais curta, como a C15. Guarde apenas um programa em cada lado da fita; assim, ficará muito mais fácil carregar os programas na memória do computador. Não se esqueça de rotular todas as fitas gravadas.

O objetivo de guardar uma cópia de reserva é evitar a necessidade de digitar novamente todo o programa caso uma das fitas sofra algum tipo de dano.

Se estiver trabalhando com disquetes, tenha sempre um disquete de reserva com as cópias de todos os seus programas definitivos.

9. Muitos usuários guardam suas fitas sem nenhum cuidado e depois se admiram quando elas começam a apresentar defeito. Para assegurar uma longa vida útil às suas fitas, siga estas regras simples (quase todas elas também se aplicam aos disquetes):

- Rotule claramente todas as suas fitas.
- Guarde as fitas nos estojos quando não estiverem sendo usadas.
- Guarde as fitas em um lugar fresco e seco.
- Evite deixar as fitas nas proximidades de aparelhos capazes de gerar campos magnéticos, como receptores de TV, computadores, alto-falantes, motores, ventiladores e especialmente telefones.
- Não deixe uma fita dentro do gravador sem necessidade e não deixe o gravador em "play" a não ser que o programa o exija.
- Depois de gravar a versão definitiva do programa, quebre a lingüeta de plástico do cassete para impedir que o programa seja apagado por engano. Se você quiser tornar a gravar alguma coisa na fita, bastará tapar o buraco com fita adesiva.

10. Não passe mais que três horas digitando. Faça uma pausa, dê uma volta, descance a cabeça antes de continuar!

Outros conselhos

TOMEI UM BANHO DE DE CHUVEIRO

Leia a frase acima em voz alta. Talvez fique surpreso ao saber que a maioria das pessoas lê “Tomei um banho de chuveiro”. Será que o mesmo aconteceu com você? Na verdade, a frase é “Tomei um banho de de chuveiro”. A palavra “de” aparece duas vezes seguidas, mas como a repetição não é esperada, o cérebro não “vê” o segundo “de”.

Um problema muito semelhante ocorre na hora de copiar a listagem de um programa. Muitas vezes você digita o que espera encontrar, em lugar daquilo que realmente está impresso. Outros erros muito comuns são entrar com a letra O em lugar do algarismo zero (que nas listagens deste livro aparece como Ø) e com a letra S no lugar do cífrão (\$).

Basta um momento de distração para misturar duas linhas consecutivas com textos semelhantes. Assim, por exemplo,

```
1Ø PRINT "Entre com o primeiro número"
2Ø PRINT "Entre com o número seguinte"
```

pode facilmente ser transformado em:

```
1Ø PRINT "Entre com o primeiro seguinte"
```

Olhe sempre para a tela para verificar se os caracteres que exigem o uso da tecla **Shift** foram digitados corretamente. Se não fizer isso, poderá produzir jóias como:

```
1Ø PRINT 6ONDE ESTAO AS ASPAS/6
```

Jamais tente alterar um programa enquanto está sendo digitado. Deixe para fazer as mudanças depois que o programa estiver funcionando. Não se sinta tentado a pular linhas apenas porque elas lhe parecem dispensáveis. Se em algum ponto o controle do programa for transferido para uma linha que você omitiu, o programa não funcionará.

As linhas que começam com REM contêm comentários que o ajudarão a compreender o funcionamento dos programas. Essas linhas devem ser digitadas exatamente como aparecem na listagem. Depois que o programa estiver funcionando e a versão final estiver salva em fita ou disquete, você poderá, se quiser, criar uma outra cópia sem as linhas de REM.

Não há, porém, nenhuma vantagem real em fazer isso, já que a pequena economia de memória resultante não justifica o trabalho que você terá para localizar e suprimir as linhas. Nos programas que aparecem neste livro, as linhas de REM podem ser omitidas sem problemas, porque tivemos o cuidado de não transferir o controle do programa para nenhuma linha de REM. Entretanto, não se pode dizer o mesmo de outros programas; existem alguns autores que não têm nenhum escrúpulo em transferir o controle para linhas de REM.

Os espaços podem ser de importância vital para um programa de computador. Procure sempre certificar-se de que está entrando com o número correto de espaços. Para facilitar o trabalho, alinhe a borda de um pedaço de papel com os caracteres das linhas acima e abaixo da linha que está sendo digitada e use esses caracteres para contar os espaços, deslocando o papel coluna por coluna.

Todas as listagens deste livro foram impressas com 39 colunas de largura, para corresponderem à largura da tela do computador no modo SCREEN Ø. Isto o ajudará a verificar se o número de espaços está certo. Assim, por exemplo, se uma linha da listagem termina com quatro espaços, a linha na tela também deve terminar com quatro espaços. As instruções que ocupam mais de uma linha na tela exigem uma atenção especial. Aperte a tecla RETURN apenas quando chegar no final da instrução. Se você se distrair e apertar a tecla RETURN ao chegar ao final da primeira linha, a instrução será carregada de forma truncada.

Usando uma régua para marcar a linha em que está, você reduzirá ainda mais a possibilidade de erros.

Se você seguir todos os conselhos aqui enunciados, não deverá encontrar grandes problemas. É inevitável, porém, que, apesar de todas as precauções, apareçam alguns erros nos seus programas. Nesse caso, consulte o último capítulo do livro, que contém alguns conselhos úteis para localizar erros.

Acentos, cedilha e letras minúsculas

Embora os micros brasileiros disponham de acentos, estes em geral são usados para mostrar mensagens na tela e não para entrada de dados no computador. Esta é a filosofia adotada neste livro. Assim, *não use acentos ou cedilha* para responder às perguntas do computador. Os acentos devem ser simplesmente omitidos; o “ç”, substituído pela letra “c”. A única exceção é o jogo “Hoje, o Brasil”, em que, por se tratar de um jogo didático, os nomes dos estados e capitais só são aceitos pelo computador quando digitados corretamente, com todos os acentos que possuem.

Por outro lado, com exceção do jogo Maior-Menor, em que a letra “M” é usada para indicar “maior” e a letra “m” para indicar “menor”, as letras maiúsculas e minúsculas podem ser usadas indistintamente para responder às perguntas formuladas pelo computador.

Técnicas Usadas nos Programas

Ponteiros de cor (cores lógicas)

O sistema de atribuição de cores adotado no MSX pode ser usado para criar a ilusão de um movimento muito rápido. O método consiste em definir separadamente cada elemento de uma imagem através de diferentes ponteiros de cor. Em seguida, os ponteiros são trocados pelo próprio programa. Quando as cores são modificadas, o efeito na tela é instantâneo e cria a ilusão de movimento.

O programa *Labirinto* é um excelente exemplo do uso desta técnica. O resultado é um labirinto em que o ponto de vista da pessoa que está jogando muda rapidamente. Não seria possível conseguir o mesmo efeito de outra forma, a não ser recorrendo a um extenso programa em linguagem de máquina.

Ou exclusivo (XOR)

Esta é outra técnica que é usada neste livro. O uso do XOR permite superpor informações a uma imagem que já esteja presente na tela. Ao ser executada uma segunda vez, a operação XOR remove a informação adicional, deixando a imagem antiga inalterada.

Exemplos dos comandos necessários para usar esta técnica podem ser encontrados em muitos dos programas. O programa *Xadrez* contém alguns ótimos exemplos.

O objetivo principal deste livro, como já dissemos, é oferecer pro-


gramas agradáveis e divertidos. Para descrever com pormenores como cada um desses programas funciona, teríamos que reduzir à metade o número de jogos que o livro contém.

Se você se dispuser a estudar a estrutura lógica de cada programa, descobrirá muitas soluções interessantes e originais para problemas práticos de programação. Tente compreender a razão pela qual certas rotinas são usadas e a forma como funcionam. Depois que salvar em fita a versão final de cada programa, experimente modificá-lo de diversas maneiras para ver o que acontece; assim, aprenderá muito mais a respeito de programação e poderá incorporar o que aprendeu aos seus próprios programas.

Agora chega de conversa e vamos começar a carregar nossos jogos.

1

Marie Celeste



Uma Aventura no Mar

Neste jogo de aventura, você se encontra ancorado ao lado do lendário navio-fantasma.

Qual é o segredo do *Marie Celeste*? Onde está a tripulação?

Para resolver este mistério marítimo, basta fornecer as instruções corretas ao computador. As mensagens foram misturadas na listagem para que você não descubra as respostas enquanto estiver digitando o programa.

Na maioria dos jogos de aventura, as direções são indicadas como norte, sul, leste e oeste. Neste jogo, porém, como se trata de uma aventura náutica, as direções são proa, popa, bombordo e boreste, que correspondem, respectivamente, à parte dianteira do navio, à parte traseira, ao lado esquerdo e ao lado direito.

Enquanto estiver jogando, é aconselhável desenhar um mapa dos lu-

gares por onde já passou. O computador aceita instruções com duas ou três palavras, como “pegar balde” ou “subir pela corda”, e abreviações de três letras para as direções: PRO, POP, BOM e BOR. Para saber em que direções você pode se movimentar, entre com “SAIDAS”.

Em muitos jogos de aventura, o jogador tem que entrar com a palavra “INVENTARIO” para saber o que está carregando no momento. Neste jogo, porém, a lista dos objetos é mostrada na tela depois de cada movimento.

No caso dos jogos de aventura, uma das táticas mais eficazes (mas que exige uma boa dose de paciência) é experimentar vários “verbos diferentes para descobrir quais os que fazem algum efeito. Alguns verbos muito usados em jogos de aventura são: pegar, subir, descer e abrir. Tente usar esses e outros semelhantes enquanto estiver a bordo do *Marie Celeste*. Experimente também dar instruções como “Examinar a chave” no momento apropriado para ver o que acontece. Para abandonar o jogo, entre com “DESISTO”. O computador perguntará a você se deseja começar de novo.

Depois que resolver o mistério, convide os amigos para jogar também. Você se divertirá vendo outras pessoas passarem pelas mesmas dificuldades que teve que enfrentar.

```

1000 REM <<< MARIE CELESTE - ISSI/JIM/
PAULG >>>
1010 SCREEN 0:COLOR 15,4,7:KEY OFF
1020 WIDTH 38:GOTO 2600
1030 RESTORE 1120
1040 FOR N=1 TO 35
1050 FOR M=1 TO 6
1060 READ A:LOCATE N,M
1070 PRINT CHR$(1);CHR$(A);
1080 NEXT M
1090 NEXT N
1100 PRINT:PRINT
1110 RETURN
1120 DATA 82,86,86,86,86,81,87,32,32,32
,32,32,82,86,86,86,86,B1,87,32,32,32,32
,32,89,86,86,86,86,81
1130 DATA 32,32,88,88,86,90,32,32,87,87
,32,87,32,32,89,83,86
1140 DATA 81,32,32,84,86,86,81,32,32,87
,32,32,32,32,89,32,32,32

```

```
1150 DATA 32,71,89,86,86,81
1160 DATA 32,32,88,84,86,90,32,32,87,87
,32,87,32,32,89,91,32,91
1170 DATA 32,32,32,32,32,32
1180 DATA 88,86,86,86,86,90,87,32,32,32
,32,87,89,32,32,32,32,91
1190 DATA 32,32,88,84,86,90,32,32,87,87
,32,87,32,32,89,91,32,91
1200 DATA 89,86,86,86,86,81
1210 DATA 32,32,88,84,86,90,32,32,87,87
,32,87,32,32,89,91,32,91
1220 DATA 32,32,88,90,32,90,32,32,87,87
,32,87,32,32,89,89,86,91
1230 DATA 32,88,84,86,86,90,32,32,87,32
,32,87,32,32,32,32,32,91
1240 DATA 32,32,88,84,86,86,90,32,32,87,87
,32,87,32,32,89,91,32,91
1250 REM <<< INICIALIZAÇÃO >>>
1260 DEF FNA$(N)=RIGHT$(O$(N),LEN(O$(N))
)-2)
1270 DIM D(10,7)
1280 DIM O$(11)
1290 DIM M(10,4)
1300 DIM O(11)
1310 RESTORE 1410
1320 FOR N=1 TO 10
1330 READ D(N,1),D(N,2),D(N,3),D(N,4)
1340 READ D(N,5),D(N,6),D(N,7)
1350 READ M(N,1),M(N,2),M(N,3),M(N,4)
1360 NEXT N
1370 FOR N=1 TO 11
1380 READ O$(N),O(N)
1390 NEXT N
1400 RETURN
1410 DATA 1,8,10,14,0,0,0
1420 DATA 0,0,0,0
1430 DATA 1,4,6,20,36,21,31
1440 DATA 1,1,0,0
1450 DATA 1,4,6,20,0,0,0
```

```
1460 DATA 1,0,1,0
1470 DATA 1,4,6,20,32,22,34
1480 DATA 0,1,1,0
1490 DATA 1,18,33,19,3,20,0
1500 DATA 0,1,0,0
1510 DATA 1.5,8,9,11,20,0,0
1520 DATA 0,0,0,0
1530 DATA 1,2,20,25,6,15,16
1540 DATA 1,0,0,1
1550 DATA 1,19,24,30,26,0,0
1560 DATA 0,0,1,0
1570 DATA 1,12,22,34,29,3,0
1580 DATA 1,0,0,1
1590 DATA 3,20,1,17,33,27,0
1600 DATA 0,0,1,0
1610 DATA A FACAO,8
1620 DATA A LONA,1
1630 DATA A POCA DE AGUA,2
1640 DATA A CORDAO,6
1650 DATA O BILHETE,8
1660 DATA O BALDE,4
1670 DATA A CHAVE,0
1680 DATA O GATO,9
1690 DATA O ESFREGAO,5
1700 DATA O PAR DE GALOCHAS,5
1710 DATA O GANCHO,0
1720 REM <<< ENTRADA DE DADOS >>>
1730 M$="" :PRINT:PRINT:PRINT ">" ;
1740 A=ASC(INPUT$(1)) AND 223:A=A OR 32
AND A=0:A$=CHR$(A)
1750 IF A=8 AND M$>"" THEN M$=LEFT$(M$,
LEN(M$)-1):PRINT CHR$(127);
1760 IF A=13 AND M$>"" THEN PRINT A$:RE
TURN
1770 IF (A$<"A" OR A$>"Z") AND A<>32 OR
LEN(M$)=40 THEN 1740
1780 M$=M$+A$:PRINT A$:GOTO 1740
1790 REM <<< INVENTARIO >>>
1800 PRINT:PRINT
```

```

1810 FOR N=1 TO 11:IF O(N)=13 THEN GOTO
1830
1820 NEXT N:PRINT "Você não está carreg
ando nada.":RETURN
1830 PRINT "Você está carregando:"
1840 FOR N=1 TO 11
1850 IF O(N)=13 THEN PRINT FNA$(N)
1860 NEXT N
1870 RETURN
1880 REM <<< SEPARAR >>>
1890 V$="" :N$=""
1900 FOR N=1 TO LEN(M$)
1910 IF MID$(M$,N,1)="" THEN GOTO 1950
1920 NEXT N
1930 V$=M$
1940 RETURN
1950 V$=LEFT$(M$,N-1)
1960 N$=RIGHT$(M$,LEN(M$)-N)
1970 RETURN
1980 REM <<< COMANDOS >>>
1990 GOSUB 1730
2000 GOSUB 1890
2010 IF V$="DESISTO" THEN GOTO 3940
2020 IF V$="SAIDAS" THEN GOSUB 3790
2030 C$=LEFT$(V$,3)
2040 IF C$="BOM" THEN GOSUB 2640
2050 IF C$="BOR" THEN GOSUB 2670
2060 IF C$="PRO" THEN GOSUB 2700
2070 IF C$="POP" THEN GOSUB 2730
2080 IF C$="PEG" OR C$="CAR" OR C$="LEV
" OR C$="SEG" THEN GOSUB 2770
2090 IF C$="SOL" OR C$="LAR" THEN GOSUB
2970
2100 IF C$="ENT" THEN GOSUB 3050
2110 IF C$="SAI" THEN GOSUB 3110
2120 IF C$="ABR" OR C$="FEC" OR C$="TRA
" THEN GOSUB 3410
2130 IF C$="COR" THEN GOSUB 3520
2140 IF C$="SUB" OR C$="DES" THEN GOSUB
3650

```

```
2150 IF C$="MAT" THEN GOSUB 3770
2160 IF C$="ENX" OR C$="SEC" THEN GOSUB
3870
2170 IF C$="AMA" OR C$="PRE" THEN GOSUB
3470
2180 IF C$="CAL" OR C$="VES" THEN GOSUB
2930
2190 C$=LEFT$(V$,2)
2200 IF C$="IR" OR C$="VA" OR C$="VO" T
HEN V$=N$:GOTO 2010
2210 IF C$="EX" OR C$="LE" THEN GOSUB 3
170
2220 IF V$="*" THEN GOTO 2240
2230 PRINT:PRINT "Não entendi..."
2240 GOSUB 2450
2250 GOTO 1990
2260 REM <<< APRESENTAÇÃO >>>
2270 GOSUB 2330
2280 GOSUB 1260
2290 GOSUB 1030
2300 P0=6
2310 RETURN
2320 <<< STRINGS >>>
2330 DIM W$(36)
2340 RESTORE 2390
2350 FOR N=1 TO 36
2360 READ W$(N)
2370 NEXT N
2380 RETURN
2390 DATA VOCE ESTA,NA POPA,PROA,NO,BOR
ESTE,CONVES,DEBAIXO,EM UM,PEQUENO,GRAND
E
2400 DATA "BARCO, AO LADO",EM,UM,BARCO
SALVA-VIDAS,RANGE
2410 DATA SOB SEUS PASSOS.,DO,LADO DE F
ORA,DENTRO,NA,DO MARIE CELESTE.
2420 DATA ESTA,UMA,NO MAR.,CABINE,O,CAP
ITAO.,TRANCADO.,DESTRANCA DO PARA,DO,MO
LHADO.,VOCE VE,DE UM DEPOSITO
```

```

2430 DATA PASSAGEM,DA,O CHAO,DEMONIO,TR
IANGULO DAS BERMUDAS, ATLANTIDA
2440 REM <<< DESCRIÇÕES >>>
2450 IF PO=2 AND O(10)<>13 THEN PRINT "
O chão está molhado e você escorrega na
direção da proa.":PO=3
2460 IF PO=1 AND O(2)=0 THEN GOTO 3980
2470 PRINT
2480 FOR N=1 TO 7
2490 PRINT W$(D(PO,N));" ";
2500 NEXT N
2510 GOSUB 1800
2520 PRINT:FOR N=1 TO 11:IF O(N)=PO THE
N 2540
2530 NEXT N:PRINT "Você não vê nenhum o
bjeto.":RETURN
2540 PRINT "Você vê:"
2550 FOR N=1 TO 11
2560 IF O(N)=PO THEN PRINT "UM";:IF LEF
T$(O$(N),1)<>"O" THEN PRINT O$(N) ELSE
PRINT RIGHT$(O$(N),LEN(O$(N))-1)
2570 NEXT N
2580 RETURN
2590 REM <<< COMEÇO DO JOGO >>>
2600 GOSUB 2270
2610 GOSUB 2450
2620 GOSUB 1990
2630 REM <<< EXECUÇÃO DOS COMANDOS >>>
>
2640 V$="*"
2650 IF M(PO,4)=1 THEN PO=PO-5:RETURN
2660 PRINT "Você não pode ir nessa dire
ção.":RETURN
2670 V$="*"
2680 IF M(PO,2)=1 THEN PO=PO+5:RETURN
2690 PRINT "Você não pode ir nessa dire
ção.":RETURN
2700 V$="*"
2710 IF M(PO,1)=1 THEN PO=PO+1:RETURN

```

```
2720 PRINT "Você não pode ir nessa direção.":RETURN
2730 V$="*"
2740 IF (PO=2 OR PO=7) AND D(PO,1)<>INT(D(PO,1)) AND O(4)=PO THEN GOTO 3700
2750 IF M(PO,3)=1 THEN PO=PO-1:RETURN
2760 PRINT "Você não pode ir nessa direção.":RETURN
2770 V$="*"
2780 FOR N=1 TO 11
2790 IF N$=O$(N) OR N$=FNA$(N) THEN GOT O 2840
2800 NEXT N
2810 IF C$="PEG" THEN PRINT "Pegar "; ELSE IF C$="LEV" THEN PRINT "Levar "; ELSE IF C$="CAR" THEN PRINT "Carregar "; ELSE PRINT "Segurar ";
2820 PRINT "o quê ?"
2830 RETURN
2840 IF O(N)=13 THEN PRINT "Você já está com ";O$(N):RETURN
2850 IF O(N)<>PO THEN PRINT "Que ";FNA$(N); "?":RETURN
2860 IF N=4 AND D(PO,1)<>INT(D(PO,1)) THEN PRINT "Está amarrada apertado demais !!!":RETURN
2870 IF N=6 AND O(7)=0 THEN GOSUB 2900: :GOTO 3340
2880 IF N=2 THEN PRINT "Ela está presa no barco !":RETURN
2890 IF N=11 THEN PRINT "E pesado demais para você !":RETURN
2900 O(N)=13
2910 PRINT "Você pega ";O$(N)
2920 RETURN
2930 V$="*":IF N$<>O$(10) AND N$<>FNA$(10) THEN IF C$="YES" THEN PRINT "Vestir o quê ?":RETURN ELSE PRINT "Calçar o quê ?":RETURN
```

```

2940 IF O(10)=13 THEN PRINT "Você já ca
lçou ";O$(10):RETURN
2950 IF O(10)<>PO THEN PRINT "Que ";FNA
$(10);"?":RETURN
2960 PRINT "Você calça ";O$(10):O(10)=1
3:RETURN
2970 V$="*"
2980 FOR N=1 TO 10
2990 IF N$=O$(N) OR N$=FNA$(N) THEN 303
0
3000 NEXT N
3010 IF C$="SOL" THEN PRINT "Soltar ";
ELSE PRINT "Largar ";
3020 PRINT "o quê ?":RETURN
3030 IF O(N)<>13 THEN PRINT "Você não e
stá carregando esse objeto !":RETURN
3040 PRINT "Você solta ";O$(N):O(N)=PO:
RETURN
3050 V$="*"
3060 IF (N$<>"DEPOSITO" AND N$<>"NO DEP
OSITO") THEN PRINT "Entrar onde ?":RETU
RN
3070 IF PO=5 THEN PRINT "Você já está d
entro dele !":RETURN
3080 IF PO<>10 THEN PRINT "Que depósito
?":RETURN
3090 IF M(10,4)=0 THEN PRINT "Está tran
cado !":RETURN
3100 PO=5:PRINT "Você entra no depósito.
":RETURN
3110 V$="*"
3120 IF (N$<>"DEPOSITO" AND N$<>"DO DEP
OSITO") THEN PRINT "Sair de onde ?":RET
URN
3130 IF PO=10 THEN PRINT "Você já está
fora dele !":RETURN
3140 IF PO<>5 THEN PRINT "Que depósito
?":RETURN
3150 IF M(10,4)=0 THEN PRINT "Está tran
cado !":RETURN

```

```

3160 PO=10:PRINT"Você sai 'do depósito."
:RETURN
3170 V$="*"
3180 IF N$="DEPOSITO" OR N$="O DEPOSITO"
" THEN IF PO=10 THEN PRINT "Ele está";W
$(D(PO,6)):RETURN ELSE IF PO=5 THEN PRI
NT "Ele é escuro e empoeirado.":RETURN
ELSE PRINT "Que depósito ?":RETURN
3190 FOR N=1 TO 11
3200 IF N$=O$(N) OR N$=FNA$(N) THEN GOT
O 3250
3210 NEXT N
3220 IF C$="LE" THEN PRINT "Ler "; ELSE
PRINT "Examinar ";
3230 PRINT "o quê ?"
3240 RETURN
3250 IF C$="LE" AND N<>5 THEN PRINT "Qu
e bobagem !":RETURN
3260 IF O(N)<>PO AND O(N)<> 13 THEN PRI
NT "Que ";FNA$(N);"?":RETURN
3270 IF N=1 THEN PRINT "E afiada !"
3280 IF N=2 THEN PRINT "E verde !"
3290 IF N=3 THEN PRINT "E de água salga
da !"
3300 IF N=4 THEN PRINT "E bem grossa e
resistente !"
3310 IF N=4 AND D(PO,1)<>INT(D(PO,1)) T
HEN PRINT "Está amarrada na amurada do
navio !"
3320 IF N=5 THEN PRINT "Está escrito:":
PRINT "'Não adianta examinar o gato'"
3330 IF N=6 AND O(7)<>0 THEN PRINT "Est
á vazio !"
3340 IF N=6 AND O(7)=0 THEN PRINT "Caiu
algo de dentro dele !":O(7)=PO
3350 IF N=7 THEN PRINT "Está escrito 'D
EPOSITO' !"
3360 IF N=8 THEN PRINT "Ele sai corrend
o atrás de um rato !":O(8)=0

```

```

3370 IF N=9 THEN PRINT "Está muito gasto !"
3380 IF N=10 THEN PRINT "São fedorentas !"
3390 IF N=11 THEN PRINT "Parece que é para uma corda !"
3400 RETURN
3410 V$="*":IF N$<>"DEPOSITO" AND N$<>"O DEPOSITO" THEN IF C$="ABR" THEN PRINT "Abrir o quê ?":RETURN ELSE. IF C$="FEC" THEN PRINT "Fechar o quê ?":RETURN ELSE PRINT "Trancar o quê ?":RETURN
3420 IF PO<>10 AND PO<>5 THEN PRINT "Que depósito ?":RETURN
3430 IF O(7)<>13 THEN PRINT "VOCE NAO TEM A CHAVE !":RETURN
3440 IF C$<>"ABR" THEN M(PO,4)=0:M(PO,2)=0:PRINT "Você tranca o depósito.":D(10,6)=27:RETURN
3450 IF PO=10 THEN M(PO,4)=1 ELSE M(PO,2)=1
3460 PRINT "Você abre o depósito.":D(10,6)=28:RETURN
3470 V$="*":IF N$<>O$(4) AND N$<>FNA$(4) THEN IF C$="AMA" THEN PRINT "Amarrar o quê ?":RETURN ELSE PRINT "Prender o quê ?":RETURN
3480 IF O(4)<>13 AND O(4)<>PO THEN PRINT "Que ";FNA$(4); "?":RETURN
3490 IF D(PO,1)<>INT(D(PO,1)) THEN PRINT O$(4); " já está amarrada !":RETURN
3500 IF PO<>6 AND PO<>7 AND (PO<>2 OR O(11)<>2) AND PO<>1 THEN PRINT "Não há onde amarrá-la aqui !":RETURN
3510 D(PO,1)=D(PO,1)+.5:O(4)=PO:PRINT "Você amarra ";O$(4):RETURN
3520 V$="*":IF O(1)<>13 THEN PRINT "Com o quê ?":RETURN
3530 FOR N=1 TO 11

```

```

3540 IF N$=0$(N) OR N$=FNA$(N) THEN 358
0
3550 NEXT N
3560 PRINT "Cortar o quê ?"
3570 RETURN
3580 IF O(N)<>PO AND O(N)<>13 THEN PRIN
T "Que ";FNA$(N);"?":RETURN
3590 IF N=4 THEN IF D(PO,1)<>INT(D(PO,1
)) THEN D(PO,1)=D(PO,1)-.5
3600 IF N=2 THEN PRINT "Você corta ";O$(
2):GOTO 3980
3610 IF N=8 THEN GOTO 3770
3620 IF N=5 OR N=4 THEN PRINT "Você cor
ta ";O$(N);".":RETURN
3630 PRINT "Que besteira !"
3640 RETURN
3650 V$="*":IF N$<>"FEL"+O$(4) AND N$<>
"N"+O$(4) THEN PRINT "Por onde ?":RETUR
N
3660 IF O(4)<>PO AND O(4)<>13 THEN PRIN
T "Que corda ?":RETURN
3670 IF O(4)=13 THEN PRINT "Como, se voc
é a está carregando ?":RETURN
3680 IF D(PO,1)=INT(D(PO,1)) THEN PRINT
"A corda não está presa a nada !":RETU
RN
3690 IF(C$="DES" AND (PO=1 OR PO=6)) OR
(C$="SUB" AND (PO=2 OR PO=7)) THEN PRI
NT "Impossivel.":RETURN
3700 PRINT "Você ";:IF PO=6 OR PO=1 THE
N PRINT "sobe"; ELSE PRINT "desce";
3710 PRINT " pela corda."
3720 IF PO=6 THEN PO=7:O(4)=7:D(7,1)=D(
7,1)+.5:D(6,1)=D(6,1)-.5:RETURN
3730 IF PO=7 THEN PO=6:O(4)=6:D(6,1)=D(
6,1)+.5:D(7,1)=D(7,1)-.5:RETURN
3740 IF PO=2 THEN PO=1:O(4)=1:D(1,1)=D(
1,1)+.5:D(2,1)=D(2,1)-.5:RETURN

```

```


3750 IF PO=1 THEN PO=2:O(4)=2:D(2,1)=D(
2,1)+.5:D(1,1)=D(1,1)-.5:RETURN
3760 RETURN
3770 V$="*":IF (N$=0$(8) OR N$=FNA$(8))
THEN IF O(8)=PO OR O(8)=13 THEN PRINT
O$(8); " fugiu.":O(8)=0:RETURN ELSE PRIN
T "Que ";FNA$(8); "?":RETURN
3780 PRINT "Que bobagem !":RETURN
3790 V$="*"
3800 PRINT:PRINT "As saídas são:"
3810 IF M(PO,1)<>0 THEN PRINT "PROA ";
3820 IF M(PO,3)<>0 OR D(PO,1)<>INT(D(PO
,1)) THEN PRINT "POPA ";
3830 IF M(PO,2)<>0 THEN PRINT "BORESTE
";
3840 IF M(PO,4)<>0 THEN PRINT "BOMBORDO
";
3850 IF M(PO,1)=0 AND M(PO,2)=0 AND M(P
O,3)=0 AND M(PO,4)=0 THEN PRINT "NENHUM
A.";
3860 PRINT:RETURN
3870 V$="*"
3880 IF PO<>2 THEN IF C$="ENX" THEN PRI
NT "Enxugar o quê ?":RETURN ELSE PRINT
"Secar o quê ?":RETURN
3890 IF O(9)<>13 THEN PRINT "Como, seu t
clo ?":RETURN
3900 IF O(6)<>13 THEN PRINT "Onde você
vai espremer o esfregão ?":RETURN
3910 O(3)=0:PRINT "Muito bem ! Agora es
tá seguro."
3920 D(2,1)=1:D(2,2)=2:D(2,3)=20:D(2,4)
=32:D(2,5)=13:D(2,6)=14:D(2,7)=23
3930 O(11)=2:RETURN
3940 PRINT "Quer jogar de novo ?"
3950 GOSUB 1730
3960 IF M$="SIM" OR M$="S" OR M$="QUERO
" THEN RUN
3970 CLS:END

```

```
3980 PRINT:PRINT:PRINT "SURPRESA !!!!!  
SURPRESA !!!!!":PRINT "NOS NOS ESCONDEM  
OS ";W$(7);" DE UM ";0$(2);"EM ";W$(13);  
" ";W$(11);" ";W$(20)  
3990 PRINT "Nós achávamos que você"  
4000 PRINT "NUNCA conseguiria descobrir  
!":PRINT:PRINT "***** FIM DA AVEN  
TLRA *****"  
4010 END
```

2

Diga, Mestre



Um Jogo de Adivinhação

Este jogo não é bem o que parece! Ao carregar o programa, você pode ter a impressão de que comprehende perfeitamente o que ele faz, mas durante a execução verá que as coisas não são tão simples assim...

Em *Diga, Mestre*, o computador é um “professor” que escolhe um assunto e pede a você palavras apropriadas. Suponhamos, por exemplo, que o computador escolha “filmes”. Você então pode entrar com “Diga, mestre, serve ALIEN?” O professor pode responder “sim” ou “não”. Cabe a você tentar descobrir a razão da resposta antes de entrar com outro exemplo, como “O EXTERMINADOR DO FUTURO”, para o qual a resposta também poderá ser “sim” ou “não”. O jogo continua até você conseguir obter três respostas “sim” seguidas, caso em que o computador supõe que você já sabe por que alguns filmes “servem” e outros não.

Mais tarde, quando já estiver dominando perfeitamente o programa, talvez você queira modificá-lo para mudar as regras ou acrescentar novos assuntos, o que poderá constituir um novo desafio!

Mostre a um amigo a listagem do programa. Ele na certa ficará surpreso com a falta de dados. Invente uma explicação qualquer ou então diga-lhe o seguinte:

“O programa utiliza um sistema avançado de análise de texto com base nas características matemáticas das palavras da língua portuguesa. Assim, a seleção criteriosa de parâmetros permite aplicar rotinas de teste relativamente gerais a respostas específicas.”

Naturalmente, você sabe que esta explicação é fantasiosa, já que você mesmo entrou com as linhas em que as palavras são verificadas! Por que, então, na categoria “insetos”, *mosquito* serve e *mosca* e *percevejo* não servem?

Notas:

- 1) Se você cometer um erro quando estiver entrando com uma palavra, aperte a tecla de retrocesso. A palavra inteira será cancelada e você poderá digitá-la de novo.
- 2) Se desistir de descobrir a regra e quiser continuar jogando, entre com a palavra “DESISTO”.

```

1000 REM <<< DIGA,MESTRE - ANDY >>>
1010 CLEAR 2000
1020 DIM S$(11)
1030 GOSUB 2190
1040 RESTORE 1830
1050 SCREEN 1,3:COLOR 15,13,13:KEY OFF
1060 FOR S=0 TO 9
1070 FOR A=1 TO 32
1080 READ A$
1090 S$(S)=S$(S)+CHR$(VAL("&H"+A$))
1100 NEXT A
1110 SPRITE$(S)=S$(S)
1120 NEXT S:Y=56
1130 FOR X=0 TO 160
1140 GOSUB 2030
1150 NEXT
1160 GOSUB 2080
1170 LOCATE 11,5:PRINT "DIGA,"
```

```

1180 LOCATE 11,7:PRINT "MESTRE"
1190 FOR T=1 TO 2500:NEXT T
1200 CLS
1210 X=80:Y=40:GOSUB 2030:GOSUB 2080
1220 FOR YY=0 TO 23:LOCATE 20,YY:PRINT
"!":NEXT YY
1230 FOR X=21 TO 30:LOCATE X,8:PRINT "="
":LOCATE X,15:PRINT "=":NEXT X
1240 LOCATE 21,2:PRINT "DIGA O"
1250 LOCATE 21,3:PRINT "NOME DE"
1260 LOCATE 21,4:PRINT "UM"
1270 LOCATE 21,10:PRINT "ACERTOS"
1280 LOCATE 21,11:PRINT "SEGUIDOS"
1290 LOCATE 21,17:PRINT "JOGO NO."
1300 GOSUB 2130
1310 LOCATE 2,16:PRINT "DIGA,MESTRE:SER
VE" >
1320 REM <<< COMEÇO DO JOGO >>>
1330 WIN=0:GO=GO+1
1340 S=INT(RND(-TIME)*18)+1
1350 LOCATE 23,4:PRINT LEFT$(T$(S),1)
1360 LOCATE 21,5:PRINT RIGHT$(T$(S),LEN
(T$(S))-1)
1370 LOCATE 23,13:PRINT WIN
1380 LOCATE 23,20:PRINT GO
1390 A=T(S,1):B=T(S,2):C=T(S,3)
1400 DEF FN OK=ASC(LEFT$(A$,1))
1410 IF A=2 THEN DEF FN OK=ASC(RIGHT$(A
$,1))
1420 IF A=3 THEN DEF FN OK=ASC(MID$(A$/,
B,1))
1430 IF A=4 THEN DEF FN OK=LEN(A$)
1440 IF A=5 THEN DEF FN OK=SP+B
1450 IF A=6 THEN DEF FN OK=ASC(LEFT$(RI
GHT$(A$,2),1))
1460 REM <<< LOOP PRINCIPAL >>>
1470 GOSUB 1550
1480 IF A$="DESISTO" THEN 1200
1490 IF RE=C THEN GOSUB 1690 ELSE GOSUB
1720

```

```
1500 FOR S=1 TO 1000:NEXT S:PUT SPRITE  
8,(-32,-32)  
1510 LOCATE 23,13:PRINT WIN  
1520 IF WIN=3 THEN FOR K=1 TO 1000:NEXT  
K:GOTO 1750  
1530 GOTO 1470  
1540 REM <<< ENTRADA DE DADOS >>>  
1550 A$="":SP=0:GOTO 1630  
1560 LOCATE 2,18:PRINT A$  
1570 W$=INKEY$:IF W$="" THEN 1570  
1580 IF W$=" " THEN SP=SP+1:GOTO 1620  
1590 IF W$=CHR$(13) AND LEN(A$)>1 THEN  
1640  
1600 W=ASC(W$) AND 223:W$=CHR$(W)  
1610 IF W$<"A" OR W$>"Z" THEN GOTO 1550  
1620 A$=A$+W$:IF LEN(A$)=15 THEN 1640  
1630 LOCATE 2,18:PRINT STRING$(16," "):  
GOTO 1560  
1640 A=LEN(A$):LOCATE 2+LEN(A$),18:PRIN  
T "?"  
1650 IF A<1 THEN 1550  
1660 RE=FN OK  
1670 RETURN  
1680 REM <<< SIM >>>  
1690 PUT SPRITE 8,(30,10),15,8  
1700 WIN=WIN+1:RETURN  
1710 REM <<< NAO >>>  
1720 PUT SPRITE 8,(30,10),15,9  
1730 WIN=0:RETURN  
1740 REM <<< GANHOU >>>  
1750 SCREEN 1:LOCATE 9,3:PRINT "Parabé  
ns!"  
1760 LOCATE 9,4:PRINT STRING$(10,"-")  
1770 LOCATE 2,10:PRINT "Você d  
escobriu  
o segredo!"  
1780 LOCATE 0,20:PRINT "Aperte 'J' para  
jogar de novo"  
1790 A$=INKEY$  
1800 IF A$="J" OR A$="j" THEN 1200
```


1810 GOTO 1790
1820 GOTO 1820
1830 DATA 0,0,0,0,0,0,0,0,0,0,0,0,1,1,1,2,
,2
1840 DATA 0,0,7C,7,F,1F,3F,7F,41,80,80,
0,0,60,30,60
1850 DATA 0,0,0,C6,FD,FF,FF,FF,7F,20
,0,0,0,0,0
1860 DATA 0,0,0,0,E0,DE,BA,C2,E2,E7,E7,
67,27,2A,2A,2A
1870 DATA 4,4,8,8,7,1,2,2,2,1,2,2,1,0,0
,0
1880 DATA 0,0,0,0,0,0,0,0,0,0,0,0,18,E4,2
,2,2
1890 DATA 0,0,0,0,0,1,1,2,4,4,4,4,4,4,2
,2
1900 DATA 20,20,40,40,80,0,0,0,0,0,0,0,0,
0,0,0,0
1910 DATA 0,0,0,0,0,0,1,6,8,10,20,20,40
,40,40,40
1920 DATA 0,0,0,0,F,70,80,0,0,0,0,0,0,0,0
,0,0
1930 DATA 0,0,0,0,E0,10,8,4,3,0,0,0,0,0,0
,0,0
1940 DATA 0,0,0,0,0,0,0,0,0,0,80,40,20,20
,10,10,10
1950 DATA 80,80,80,80,80,80,40,40,20,20
,10,8,6,1,0,0
1960 DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,80,
60,1F
1970 DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,1,6,3
8,C0
1980 DATA 10,10,10,8,8,8,8,10,10,C,62,9
9,7,0,0,0
1990 DATA 0,0,0,0,0,0,0,0,F8,80,81,F8,9,9
,9,F9,0
2000 DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,3E,2A,2
A,2A,0
2010 DATA 0,0,0,0,0,0,0,0,F8,88,8B,88,8B,
8A,8A,8B,0

```
2020 DATA 0,0,0,0,0,0,0,0,0,C0,0,CF,49,
49,EF,0
2030 PUT SPRITE 0,(X,Y),15,0
2040 PUT SPRITE 1,(X+32,Y),15,1
2050 PUT SPRITE 2,(X,Y+32),15,2
2060 PUT SPRITE 3,(X+32,Y+32),15,3
2070 RETURN
2080 PUT SPRITE 4,(X-64,Y-40),15,4
2090 PUT SPRITE 5,(X-32,Y-40),15,5
2100 PUT SPRITE 6,(X-64,Y-8),15,6
2110 PUT SPRITE 7,(X-32,Y-8),15,7
2120 RETURN
2130 LOCATE 1,15:PRINT "-----"
-----"
2140 FOR Y=16 TO 18:LOCATE1,Y:PRINT "!
!"
2150 NEXT Y
2160 LOCATE 1,19:PRINT "-----"
-----"
2170 RETURN
2180 REM <<< ASSUNTOS >>>
2190 RESTORE 2290
2200 DIM T$(18)
2210 FOR N=1 TO 18
2220 READ T$(N)
2230 NEXT N
2240 DIM T(18,3)
2250 FOR N=1 TO 18
2260 READ T(N,1),T(N,2),T(N,3)
2270 NEXT N
2280 RETURN
2290 DATA " PAIS"," ATOR",ACAPITAL," PL
ANETA"," FILME"," ESTADO"
2300 DATA " RIO",AFRUTA,ACOR,AFLOR,AARV
ORE," ESPORTE"
2310 DATA " ANIMAL"," PASSARO"," INSETO
",AMUSICA," PEIXE"," LIVRO"
2320 DATA 2,66,65,1,69,74,2,70,83,3,2,6
9
```

2330 DATA 5,2,4,4,63,7,4,69,4,3,2,65
2340 DATA 3,3,82,6,73,73,4,19,8,2,65,76
2350 DATA 1,0,67,1,80,80,6,64,84,5,68,6
9
2360 DATA 6,72,68,5,65,65

3

Eu Acuso



Descubra Quem é o Culpado

Este é um jogo para testar a sua capacidade de dedução.

Você é o Inspetor Silva do DIC (Departamento de Investigações Criminais) e está sendo submetido a uma série de testes, que têm por objetivo avaliar sua coragem, persistência e capacidade de usar o computador em investigações criminais. Como não podemos reproduzir aqui as provas de coragem e persistência, vamos nos limitar ao teste de investigação por computador.

O computador tem a capacidade de gerar milhares de diferentes permutações a partir de um certo conjunto de dados. Em um piscar de olhos, ele seleciona um criminoso e uma série de características pessoais que o identificam perfeitamente. Você tem direito de interrogar uma testemunha do crime a respeito de seis dessas características; para isso, basta en-

trar no computador com o número correspondente à característica desejada, de acordo com a lista que aparece do lado esquerdo da tela. O computador então mostra as características dos diferentes suspeitos e a característica do criminoso, de acordo com o depoimento da testemunha. Se você souber escolher bem as perguntas, poderá descobrir facilmente o culpado!

Depois de fazer as seis perguntas, você entra com o número da pessoa que julga ser o criminoso. O computador então revela se você acertou ou não.

```

1000 REM <<< EU ACUSO - JIM/ANDY >>>
1010 REM
1020 KEY OFF
1030 SCREEN 0,0,0:COLOR 15,4,4:WIDTH 36
1040 GOSUB 1350
1050 GOSUB 1560
1060 GOSUB 1700
1070 FOR G=1 TO 6
1080 GOSUB 2090
1090 NEXT G
1100 LOCATE 0,18:PRINT STRING$(39," ")
1110 LOCATE 0,13:PRINT "Agora entre com
      o número da pessoa      que você acha
      que cometeu o crime."
1120 PRINT " A testemunha disse que
      .":PRINT
1130 RESTORE 1330
1140 FOR N=1 TO 10
1150 READ A$
1160 IF W(1)=N OR W(2)=N OR W(3)=N OR W
    (4)=N THEN PRINT "O criminoso";A$;" ";S
    $(S,N)
1170 NEXT N
1180 A$=INKEY$:IF A$<"1" OR A$>"4" THEN
    1180
1190 CR=VAL(A$)
1200 FOR Y=12 TO 21:LOCATE 0,Y:PRINT ST
    RING$(39," "):NEXT Y
1210 IF CR=S THEN 1290

```

```
1220 LOCATE 0,15:PRINT "Que vergonha,  
Inspetor' Gibbs!           Está rebaixado  
a Guarda Gibbs!"  
1230 PLAY "116fedc12c"  
1240 PRINT:PRINT " O criminoso era o su  
spito número ";S;"!"  
1250 LOCATE 0,20:PRINT "Aperte qualquer  
tecla para jogar de novo"  
1260 A$=INKEY$:IF A$="" THEN 1260  
1270 FOR N=0 TO 9:QW(N)=0:NEXT N  
1280 GOTO 1050  
1290 LOCATE 0,15:PRINT "Parabéns, Inspec  
tor Gibbs !"  
1300 PLAY "116cdefcdefcdel2f"  
1310 PRINT:PRINT "O senhor descobriu qu  
em era o criminoso a partir das pistas  
disponíveis!"  
1320 GOTO 1250  
1330 DATA " era"," era"," era"," tinha  
cabelo"," tinha cabelo"," estava de"  
, " calçava"," tinha olhos"," tinha um n  
ariz"," tinha uma"  
1340 REM <<< INICIALIZAÇÃO >>>  
1350 DIM S$(4,10)  
1360 DIM W(4),QW(9)  
1370 DIM L$(10,4)  
1380 RESTORE 1450  
1390 FOR N=1 TO 10  
1400 FOR M=1 TO 4  
1410 READ L$(N,M)  
1420 NEXT M  
1430 NEXT N  
1440 RETURN  
1450 DATA HOMEM,MULHER,HOMEM,MULHER  
1460 DATA GORDO,MAGRO,FORTE,PALIDO  
1470 DATA ALTO,GRANDE,BAIXO,PEQUENO  
1480 DATA LISO,CURTO,CRESPO,LONGO  
1490 DATA BRANCO,PRETO,LOURO,RUIVO  
1500 DATA JEANS,MACACÃO,CAPA,SHORTS
```

```
1510 DATA TAMANCO,BOTA,TENIS,GALOCHA
1520 DATA AZUIS,MARRONS,VERDES,PRETOS
1530 DATA ADUNCO,GRANDE,REGULAR,TORTO
1540 DATA PISTOLA,PINTA,FACA,MALA
1550 REM <<< TELA >>>
1560 NO=1:SCREEN 0:WIDTH 40
1570 FOR N=10 TO 34 STEP 8
1580 LOCATE N,0:PRINT NO
1590 NO=NO+1
1600 NEXT N
1610 RESTORE 1660
1620 FOR N=2 TO 11
1630 LOCATE 0,N
1640 READ A$:PRINT A$
1650 NEXT N
1660 DATA 0-SEXO,1-TIPO,2-ALTURA,3-CABE
LO,4-COR'',5-TRAJE,6-SAPATO,7-OLHOS,8-N
ARIZ,9-OUTROS
1670 LOCATE 0,18:PRINT "Escolha uma pis
ta (0-9). . . ."
1680 RETURN
1690 REM <<< ESCOLHA DOS SUSPEITOS >>>
>OK
1700 S=INT(RND(-TIME)*4)+1
1710 FOR N=1 TO 4
1720 W(N)=0
1730 CH=0
1740 A=INT(RND(1)*10)+1
1750 FOR Q=1 TO 4
1760 IF W(Q)=A THEN CH=1
1770 NEXT Q
1780 IF CH=1 THEN 1730
1790 W(N)=A
1800 NEXT N
1810 FOR N=1 TO 10
1820 FOR Q=1 TO 4
1830 A=INT(RND(1)*4)+1
1840 S$(Q,N)=L$(N,A)
1850 NEXT Q
```


```

1860 NEXT N
1870 N=W(1):GOSUB 1900;N=W(2):GOSUB 190
0
1880 N=W(3):GOSUB 1900:N=W(3):GOSUB 198
0
1890 RETURN
1900 A=INT(RND(1)*4)+1
1910 S$(S,N)=L$(N,A)
1920 T=S+2:IF T>4 THEN T=T-4
1930 S$(T,N)=L$(N,A)
1940 IF A<>1 THEN B=A-1 ELSE B=4
1950 T=T-2:IF T<1 THEN T=T+4
1960 S$(T,N)=L$(N,B)
1970 RETURN
1980 A=INT(RND(1)*4)+1
1990 S$(S,N)=L$(N,A)
2000 IF S<>1 THEN T=S-1 ELSE T=4
2010 S$(T,N)=L$(N,A)
2020 IF A<>4 THEN B=A+1 ELSE B=1
2030 T=T-1:IF T=0 THEN T=4
2040 S$(T,N)=L$(N,B)
2050 T=T-1:IF T=0 THEN T=4
2060 S$(T,N)=L$(N,B)
2070 RETURN
2080 REM <<< ESCOLHA DAS PISTAS >>>
2090 A$=INKEY$:IF A$="" THEN 2090
2100 QW=VAL(A$)
2110 IF QW<0 OR QW>? OR QW(QW)<>0 THEN
2090
2120 PLAY "L8F":QW(QW)=1:QW=QW+1
2130 FOR PE=1 TO 4
2140 LOCATE 8+(PE*8-7),1+QW
2150 PRINT S$(PE,QW);
2160 NEXT PE
2170 RETURN

```

4

Star Trek



Você Agora é o Capitão Kirk

Já foram escritas versões de *Star Trek* para praticamente todos os micros lançados no mercado, o que demonstra a popularidade deste jogo. Agora você pode comandar uma Nave Estelar da classe MSX, com todos os fasores e sensores a que tem direito. Sua missão é eliminar os perigosos klingons antes que sua nave seja vaporizada ou fique sem energia.

Esta versão do jogo *Star Trek* inclui um mapa estelar completo e movimentos em "tempo real" durante os combates com o inimigo e quando a nave atraca em uma base estelar. Segue um extrato do megamanual de 32.544 páginas intitulado: "Aprenda a Comandar uma Nave Estelar em 4.000 Lições":

1. Não perca a calma! Lembre-se de que tudo que tem a perder é: sua

vida, 20 bilhões de créditos galácticos em equipamento e as vidas de todos os tripulantes da nave.

2. Não voe às cegas! Use os sensores da nave para verificar sua posição na galáxia. Apertando a tecla L, você liga o Sensor de Longo Alcance (SLA), que mostra a posição da sua nave, das bases estelares e das naves inimigas. Os setores ocupados por klingons são representados no SLA por um círculo com uma flecha, o símbolo de Marte, deus da guerra. Os setores onde existem bases estelares são representados por um círculo com uma cruz, o símbolo de Vênus, a deusa do amor. Os setores ocupados tanto por inimigos como por bases estelares são indicados por um quadrado. Cada vez que você usa o sensor de longo alcance, você gasta 5 unidades de energia do tipo usado para os motores.

3. Não perca o inimigo de vista! Apertando a tecla C, você liga o Sensor de Curto Alcance (SCA), que indica quantos inimigos e bases estelares existem no setor em que você se encontra no momento. Cada vez que você liga o SCA, são consumidas 5 unidades de energia do tipo usado para os motores.

4. Cuidado com a falta de energia! Para reabastecer a nave, você deve atracar em uma base estelar, apertando a tecla A no painel de controle e usando o controle de direção para posicionar a base estelar no centro da tela. Depois de atracar, você receberá 20 unidades de energia para os motores da nave e mais 20 para os fasores (a energia da blindagem não pode ser repostada). A capacidade máxima é de 95 unidades. Não ultrapasse este limite, caso contrário a nave sofrerá uma explosão.

5. Não perca tempo! Como comandante da Esquadra Estelar, sua missão é destruir os perigosos klingons o mais depressa que puder.

Se houver klingons no setor em que você se encontra, aperte a tecla P para fazer soar o sinal de Postos de Combate e dê início ao ataque. Coloque uma nave inimiga no centro da tela, usando o controle de direção (*joystick* ou teclas de controle do cursor) e aperte o botão de tiro ou a barra de espaço para disparar os fasores.

Se você não agir com rapidez suficiente, as naves dos klingons começarão a responder ao fogo, roubando energia da sua blindagem. Não desista! Procure “limpar” o setor antes de partir para outras batalhas. Caso, porém, a energia dos fasores se esgote, aperte a tecla **RETURN** para fugir do combate e poder se reabastecer em uma base.

6. Como mudar de setor. Para viajar para qualquer setor, é preciso apertar a tecla W, seguida por um número entre 1 e 5, que indica o fator “warp” ou distância que você deseja viajar. Em seguida, é preciso indicar a direção do movimento, escolhendo um número entre 1 e 8. Cada número cor-

responde a uma direção no espaço, começando no topo da tela e continuando no sentido dos ponteiros do relógio.

7. O desafio supremo. No início da missão, você escolhe o grau de dificuldade. No nível mais fácil você tem que destruir 10 inimigos; no nível de dificuldade média, 20; no nível mais difícil, 30. Quando você consegue concluir uma missão com sucesso, a Central de Controle dos klingons é informada e envia uma nova esquadra, com 10 naves a mais que a anterior. O desafio supremo consiste em localizar e destruir 200 naves inimigas. Está preparado, capitão Kirk? Então pode começar...

```

1000 REM <<< STAR TREK - ISSI >>>
1010 MAXFILES=2:OPEN "GRP:S" FOR OUTPUT
    AS 1
1020 SCREEN 1:KEY OFF
1030 RESTORE 3340
1040 S$="" :FOR N=0 TO 7:READ A:S$=S$+CH
R$(A):NEXT N
1050 SPRITE$(1)=S$
1060 S$="" :FOR N=0 TO 7:READ A:S$=S$+CH
R$(A):NEXT N
1070 SPRITE$(2)=S$
1080 DIM S(15,2)
1090 GOSUB 1790
1100 BEEP:PLAY "03GL1604CL8FFF8EL16C.0
3A04D.L4G"
1110 GOSUB 1880
1120 GOSUB 1580
1130 GOTO 1250
1140 A$=INKEY$
1150 IF A$="" THEN GOTO 1140
1160 A$=CHR$(ASC(A$) AND 223)
1170 LINE (184,100)-(248,148),6,BF
1180 IF A$="L" THEN GOSUB 2030
1190 IF A$="C" THEN GOSUB 2170
1200 IF A$="W" THEN GOSUB 2250
1210 IF A$="A" THEN GOSUB 2530
1220 IF A$="P" THEN GOSUB 2820:PUTSPRIT
E 2,(0,0),14,2
1230 FOR N=1 TO 20:NEXT N

```

```
1240 IF DEAD=1 OR ENER<1 OR ENER=100 THEN  
EN GOTO 1370  
1250 COLOR 15,13  
1260 LINE (95,156)-(135,191),13,BF  
1270 PRESET (95,160):PRINT #1,ENER  
1280 PRESET (95,170):PRINT #1,SHIE  
1290 PRESET (95,180):PRINT #1,PHAS  
1300 LINE (215,156)-(275,191),13,BF  
1310 PRESET (215,160):PRINT #1,SEC  
1320 PRESET (215,170):PRINT #1,ALIEN-EM  
E  
1330 PRESET (215,180):PRINT #1,3-BD  
1340 IF EME=ALIEN THEN GOTO 3380  
1350 GOTO 1140  
1360 <<< NAVE DESTRUÍDA >>>  
1370 COLOR 10,6,6:SCREEN 1  
1380 LOCATE 10,2:PRINT "QUE PENA!"  
1390 LOCATE 9,3:PRINT "-----"  
1400 IF ENER<1 THEN M$=" ficou à deriva  
e teve de"  
1410 IF ENER=100 THEN M$=" recebeu uma  
sobrecarga de"  
1420 IF DEAD=1 THEN M$=" foi destruí  
da pelos"  
1430 LOCATE 2,7:PRINT "A Nave Estelar E  
nterprise"  
1440 LOCATE 1,9:PRINT M$  
1450 IF ENER<1 THEN M$=" ser abandonad  
a, mas o"  
1460 IF ENER=100 THEN M$="energia e exp  
lodiu, mas o"  
1470 IF DEAD=1 THEN M$="demoniacos klin  
gons,mas o"  
1480 LOCATE 2,11:PRINT M$  
1490 LOCATE 1,13:PRINT "Almirante Yamas  
hita decidiu"  
1500 LOCATE 2,15:PRINT "confiar-lhe o  
comando de"  
1510 LOCATE 5,17:PRINT "outra Nave Este  
lar."
```

```
1520 LOCATE 3,21:PRINT "APERTE 'C' PARA
    COMEÇAR"
1530 A$=INKEY$: IF A$="" THEN GOTO 1530
1540 A$=CHR$(ASC(A$)AND 223)
1550 IF A$<>"C" THEN GOTO 1530
1560 GOTO 1090
1570 REM <<< TELA >>>
1580 SCREEN 2:COLOR 11,1,1:CLS
1590 LINE (16,16)-(176,148),4,BF
1600 LINE (184,28)-(248,80),6,BF
1610 LINE (184,100)-(248,148),6,BF
1620 LINE (0,2)-(256,14),14,BF
1630 LINE (8,156)-(256,191),13,BF
1640 PRESET (208,18):PRINT #1,"SLA"
1650 PRESET (208,90):PRINT #1,"SCA"
1660 COLOR 1
1670 PRESET (55,4),14:PRINT #1,"*ALERTA
    VERMELHO*"
1680 COLOR 15
1690 PRESET (16,160):PRINT #1,"MOTORES:
    "
1700 PRESET (16,170):PRINT #1,"BLINDAGE
    M:"
1710 PRESET (16,180):PRINT #1,"FASORES:
    "
1720 PRESET (144,160):PRINT #1,"SETOR:"
1730 PRESET (144,170):PRINT #1,"INIMIGO
    S:"
1740 PRESET (144,180):PRINT #1,"BASES:"
1750 PRESET (80,88),4:COLOR 10
1760 PRINT #1,CHR$(1);CHR$(90);CHR$(1);
    CHR$(82);CHR$(1);CHR$(82);CHR$(1);CHR$(91)
1770 RETURN
1780 REM <<< DIFICULDADE >>>
1790 COLOR 15,4,4:SCREEN 1
1800 PRINT "Fácil/Regular/Difícil?"
1810 A$=INKEY$: IF A$="" THEN GOTO 1810
1820 A$=CHR$(ASC(A$)AND 223)
```

```
1830 IF A$=="D" THEN ALIEN=20:RETURN
1840 IF A$=="F" THEN ALIEN=10:RETURN
1850 IF A$=="R" THEN ALIEN=15:RETURN
1860 GOTO 1810
1870 REM <<< INICIALIZAÇÃO >>>
1880 FOR N=1 TO 15:S(N,1)=0:S(N,2)=0:NE
XT N
1890 FOR N=1 TO ALIEN
1900 SEC=INT(RND(-TIME)*15+1)
1910 IF S(SEC,1)>3 THEN GOTO 1900
1920 S(SEC,1)=S(SEC,1)+1
1930 NEXT N
1940 FOR N=1 TO 3
1950 PO=INT(RND(1)*15)+1
1960 IF S(PO,2)=1 THEN GOTO 1950
1970 S(PO,2)=1
1980 NEXT N
1990 PHAS=25:ENER=25:SHIE=25:BAD=0:EME=
0:BD=0
2000 SEC=INT(RND(1)*15)+1
2010 RETURN
2020 REM <<< SENSOR DE LONGO ALCANCE >
>>>
2030 PLAY "L8CDECDE":PO=1
2040 FOR N=1 TO 3
2050 FOR M=1 TO 5
2060 PRESET (188+M*8,18+N*16),6:COLOR 1
0
2070 CH=71:IF PO=SEC THEN CH=79:GOTO 21
10
2080 IF S(PO,1)>0 THEN CH=75
2090 IF S(PO,2)=1 THEN CH=76
2100 IF CH=76 AND S(PO,1)>0 THEN CH=72
2110 PRINT #1,CHR$(1);CHR$(CH)
2120 PO=PO+1:NEXT M
2130 NEXT N
2140 ENER=ENER-5:PLAY "EDCEDC"
2150 RETURN
2160 REM <<< SENSOR DE CURTO ALCANCE >
>>>
```

```

2170 PLAY "L8CDECDE"
2180 COLOR 15:PRESET (204,110),6
2190 PRINT #1,CHR$(1);CHR$(75);";";S(SE
C,1)
2200 PRESET (204,130),6
2210 PRINT #1,CHR$(1);CHR$(76);";";S(SE
C,2)
2220 ENER=ENER-5:PLAY "EDCEDC"
2230 RETURN
2240 REM <<< SALTO WARP >>>
2250 LINE (0,2)-(256,14),12,BF
2260 PRESET (16,4),14:PRINT #1,"Fator W
arp (1 a 5) ?"
2270 A$=INKEY$
2280 IF A$<"1" OR A$>"5" THEN GOTO 2270
2290 DIS=VAL(A$)
2300 LINE (0,2)-(256,14),10,BF
2310 COLOR 1:PRESET(16,4),14:PRINT #1,"
Direção do salto (1 a 8) ?"
2320 A$=INKEY$
2330 IF A$<"1" OR A$>"8" THEN GOTO 2320
2340 LINE (184,28)-(248,80),6,BF
2350 DR=VAL(A$)
2360 COLOR 11
2370 FOR Q=60 TO 16 STEP -4
2380 LINE (Q,Q)-(192-Q,164-Q),INT(RND(1
)*15),B
2390 NEXT Q
2400 RESTORE 2510
2410 FOR N=1 TO DR:READ XY:NEXT N
2420 FOR N=1 TO DIS:SEC=SEC+XY
2430 IF SEC<1 THEN SEC=SEC+15
2440 IF SEC>15 THEN SEC=SEC-15
2450 PLAY "L8"+CHR$(N+66):NEXT N
2460 FOR Q=60 TO 16 STEP -4
2470 LINE (Q,Q)-(192-Q,164-Q),4,B
2480 NEXT Q
2490 ENER=ENER-INT(DIS*1.5)
2500 LINE (0,2)-(256,14),14,BF:RETURN

```

```
2510 DATA -5,-4,1,6,5,4,-1,-6
2520 REM <<< ATRACAR >>>
2530 LINE (0,2)-(256,14),14,BF
2540 PRESET (16,4),14:COLOR 1
2550 IF S(SEC,2)=0 THEN PRINT #1,"Nao h
á bases neste setor...":PLAY "L2GEC":RE
TURN
2560 X=INT(RND(-TIME)*100)+50:Y=18:DIS=
1
2570 PUT SPRITE 1,(X,INT(Y)),11,1
2580 GOSUB 3250
2590 IF RI=1 THEN DIS=DIS-1
2600 IF LE=1 THEN DIS=DIS+1
2610 Y=Y+.5:X=X+DIS
2620 IF X<15 THEN X=15
2630 IF X>170 THEN X=170
2640 IF Y<95 THEN GOTO 2570
2650 IF X>85 AND X<102 THEN GOTO 2720
2660 IF X<65 OR X>125 THEN PRESET (16,4
),14:PRINT #1,"A BASE ESTA FORA DE ALCA
NCE.":GOTO 2750
2670 PRESET (16,4),14:PRINT #1,"VOCE DE
STRUIU A BASE ESTELAR!"
2680 PLAY "S10EDCC"
2690 S(SEC,2)=0
2700 BD=BD+1
2710 GOTO 2750
2720 PRESET (16,4),14:PRINT #1,"VOCE AT
RACOU NA BASE. . ."
2730 ENER=ENER+20
2740 PHAS=PHAS+20
2750 FOR N=1 TO 700:NEXT N:BEEP
2760 IF ENER>95 OR PHAS>95 THEN ENER=10
0
2770 LINE (0,2)-(256,14),14,BF
2780 PUT SPRITE 1,(0,0),11,1
2790 RETURN
2800 REM <<< COMBATE >>>
2810 REM
```

```

2820 LINE (0,2)-(256,14),14,BF
2830 DEAD=0
2840 XD=2:YD=2
2850 COLOR 1
2860 IF S(SEC,1)=0 THEN PRESET (16,4),1
4:PRINT #1,"NAO HA MAIS INIMIGOS AQUI."
:PLAY "L4GEC":RETURN
2870 X=INT(RND(1)*140)+30:Y=INT(RND(1)*
120)+30
2880 GOSUB 3250
2890 PUTSPRITE 2,(X,Y),7,2
2900 IF FI=1 AND PHAS>0 THEN GOTO 3130
2910 IF KIL=1 THEN S(SEC,1)=S(SEC,1)-1:
KIL=0:GOTO 2830
2920 IF LE=1 THEN XD=2
2930 IF RI=1 THEN XD=-2
2940 IF UP=1 THEN YD=2
2950 IF DO=1 THEN YD=-2
2960 IF EX=1 THEN RETURN
2970 X=X+XD:Y=Y+YD
2980 IF X<20 THEN XD=20:XD=2
2990 IF X>165 THEN X=165:XD=-2
3000 IF Y<18 THEN Y=18:YD=2
3010 IF Y>135 THEN Y=135:YD=2
3020 IF RND(1)<.95 THEN GOTO 3060
3030 PLAY "S1402L4C"
3040 FOR M=1 TO 50:NEXT M:BEEP
3050 SHIE=SHIE-1:IF SHIE<0 THEN DEAD=1:
RETURN
3060 LINE (95,170)-(135,190),13,BF:COLO
R 15
3070 PRESET (95,170):PRINT #1,SHIE
3080 PRESET (95,180):PRINT #1,PHAS
3090 LINE (215,170)-(275,179),13,BF
3100 PRESET (215,170):PRINT #1,ALIEN-EM
E
3110 GOTO 2880
3120 REM <<< ATIRAR >>>
3130 COLOR 14

```

```


3140 PSET (16,148):DRAW "E75"
3150 PSET (176,148):DRAW "H75"
3160 PLAY "S14L64CDEFGAB"
3170 FOR N=1 TO 50:NEXT N
3180 PHAS=PHAS-1
3190 IF X>86 AND X<98 AND Y>64 AND Y<70
    THEN EME=EME+1:KIL=1:PLAY "S11L8CDEL2G"
    "
3200 COLOR 4
3210 PSET (16,148):DRAW "E75"
3220 PSET (176,148):DRAW "H75"
3230 BEEP:GOTO 2910
3240 REM <<< ENTRADA DE DADOS >>>
3250 VA=STICK(0) OR STICK(1) OR STICK(2)
)
3260 IF VA=3 THEN RI=1 ELSE RI=0
3270 IF VA=7 THEN LE=1 ELSE LE=0
3280 IF VA=1 THEN UP=1 ELSE UP=0
3290 IF VA=5 THEN DO=1 ELSE DO=0
3300 FI=ABS(STRIG(0) OR STRIG(1) OR STRIG(2))
3310 A$=INKEY$
3320 IF A$=CHR$(13) THEN EX=1 ELSE EX=0
3330 RETURN
3340 DATA 56,68,68,68,56,16,124,16
3350 DATA 15,3,5,121,136,136,136,112
3360 GOTO 1140
3370 REM <<< FIM DA MISSAO.>>>
3380 COLOR 10,6,6:SCREEN 1
3390 PLAY "05CA+.L10AGFEL4D+.", "04CA+.L
10AGFEL4D+."
3400 LOCATE 7,2:PRINT "BOM TRABALHO"
3410 LOCATE 6,3:PRINT "-----"
3420 LOCATE 2,7:PRINT " Capitão James T
. Kirk:"
3430 LOCATE 1,9:PRINT "Agora você deve
enfrentar"
3440 LOCATE 0,11:PRINT "um grupo ainda
mais numeroso"

```

```
3450 LOCATE 1,13:PRINT "dos diabólicos  
klingons..."  
3460 LOCATE 2,21:PRINT "APERTE 'R' PARA  
RECOMEÇAR"  
3470 A$=INKEY$:IF A$="" THEN GOTO 3470  
3480 A$=CHR$(ASC(A$)AND 223)  
3490 IF A$<>"R" THEN GOTO 3470  
3500 ALIEN=ALIEN+10/  
3510 GOTO 1110
```

5

Super Breakout



Um Clássico dos Fliperamas

Este jogo é recomendado para pessoas de todas as idades. Como é o menor programa do livro, talvez os novatos em programação prefiram começar por ele.

No início, o computador pergunta a você se prefere usar uma raquete grande ou pequena. Então, o jogo começa. O objetivo é eliminar todos os blocos coloridos, atingindo-os com uma bola. Para movimentar a raquete, você usa o joystick ou as teclas de controle do cursor. O importante é prever a trajetória da bola e movimentar a raquete de forma a impedir que a bola atinja a base da tela.

Se você perder uma bola, o computador lhe fornecerá mais três antes de o jogo acabar. No final, o computador mostra o número de blocos que você conseguiu destruir.

```

1000 REM <<< SUPER BREAKOUT - BOOTSY >
>>>
1010 SCREEN 1,0,0:COLOR 15,0,0:CLS
1020 KEY OFF
1030 ON SPRITE GOSUB 1490:SPRITE ON
1040 CLS:LOCATE 0,10:PRINT "Raquete Peq
uenha ou Grande ?";:I$=INPUT$(1)
1050 SZ=0:IF I$="g" OR I$="G" THEN SZ=2
1060 SCREEN ,SZ:CLS
1070 LM=240-(16 AND SZ=3):X=120-(8 AND
SZ>0)
1080 RESTORE:SC=0:LI=5:GOSUB 1630
1090 GOSUB 1200
1100 CT=0
1110 B=100
1120 C=INT(RND(-TIME)*2)
1130 C=C+INT(RND(-TIME)*30)+16
1140 BD=5:AD=-5
1150 GOSUB 1270
1160 GOSUB 1350
1170 IF CT=280 THEN GOTO 1090
1180 GOTO 1150
1190 REM <<< TELA >>>
1200 LOCATE 0,0
1210 FOR N=216 TO 248 STEP 8
1220 PRINT STRING$(28,CHR$(N))
1230 PRINT STRING$(28,CHR$(N))
1240 NEXT N
1250 RETURN
1260 REM <<< MOVIMENTO DA RAQUETE >>>
1270 ST=STICK(0) OR STICK(1) OR STICK(2
)
1280 TG=STRIG(0) OR STRIG(1) OR STRIG(2
)
1290 I$=INKEY$
1300 IF ST=7 AND X>4 THEN X=X-4
1310 IF ST=3 AND X<LM THEN X=X+4
1320 PUT SPRITE 0,(X,188),15,0
1330 RETURN

```

```


1340 REM <<< MOVIMENTO DA BOLA >>>
1350 A1=C:B1=B
1360 A1=A1+AD:B1=B1+BD
1370 IF A1<16 THEN A1=16:AD=-AD:PLAY "t
255c16"
1380 IF A1>240 THEN A1=240:AD=-AD:PLAY
"t255c16"
1390 IF B1<0 THEN B1=1
1400 IF B1>190 THEN GOTO 1520
1410 CP=&H1800+(INT(B/B)*32+C/8)
1420 CH=VPEEK(CP)
1430 IF CH<>32 THEN VPOKE CP,32:BEEP:SC
=SC+10:BD=-BD:CT=CT+1
1440 IF B1=0 THEN BD=ABS(BD):PLAY "t255c
16"
1450 C=C+AD:B=B+BD
1460 IF (240-B)/10<BD THEN BD=BD-1
1470 PUT SPRITE 1,(C,B),15,1
1480 RETURN
1490 BEEP:BD=-ABS(BD)
1500 GOTO 1450
1510 REM <<< PERDE VIDA >>>
1520 PLAY "t255116edc"
1530 LI=LI-1:IF LI>0 THEN RETURN 1110
1540 REM <<< FIM DO JOGO >>>
1550 CLS:PUT SPRITE 0,(255,208),0,0
1560 LOCATE 10,7:PRINT "PARABENS"
1570 LOCATE 4,11:PRINT "Você fez";SC;"P
ONTOS"
1580 LOCATE 5,17:PRINT "Quer jogar de n
ovo ?";:I$=INPUT$(1)
1590 IF I$=="S" OR I$=="s" THEN RETURN 10
40
1600 IF I$<>"N" AND I$<>"n" THEN 1580
1610 SCREEN 0:END
1620 REM <<< DEFINIÇÃO DOS SPRITES >>>
1630 GOSUB 1700:SPRITE$(0)=SP$:GOSUB 17
00:SPRITE$(1)=SP$

```

```
1640 FOR N=1728 TO 1984 STEP 64:RESTORE
1800:FOR M=0 TO 7
1650 READ H$:VPOKE N+M,VAL("&h"+H$)
1660 NEXT M,N
1670 FOR N=&H201B TO &H201F
1680 READ H$:VPOKE N,VAL("&h"+H$)
1690 NEXT N:RETURN
1700 SP$="":FOR N=1 TO 32:READ H$:SP$=S
P$+CHR$(VAL("&h"+H$)) :NEXT N:RETURN
1710 REM <<<< DADOS >>>
1720 DATA ff,ff,ff,0,0,0,0,0
1730 DATA 0,0,0,0,0,0,0,0
1740 DATA ff,ff,ff,0,0,0,0,0
1750 DATA 0,0,0,0,0,0,0,0
1760 DATA f0,f0,f0,f0,0,0,0,0
1770 DATA 0,0,0,0,0,0,0,0
1780 DATA 0,0,0,0,0,0,0,0
1790 DATA 0,0,0,0,0,0,0,0
1800 DATA 0,7f,7f,7f,7f,7f,7f,7f
1810 DATA 80,90,a0,20,40
```

6

Labirinto



Um Jogo Tridimensional

Este é um jogo passado em um labirinto tridimensional no qual você tem uma vista em perspectiva do espaço à sua volta. O objetivo é encontrar rapidamente a saída. Se conseguir o seu intento, escapará do louco que habita o labirinto. Se não, ele pegará você!

A figura que aparece na tela mostra apenas uma parte do labirinto. Uma seta indica a direção para a qual você está voltado no momento. Para mudar de direção, use as teclas de controle do cursor da direita e da esquerda. A tecla de cima faz você dar um passo à frente cada vez que é apertada. A tecla de baixo faz você recuar um passo, sem mudar de direção. Os movimentos também podem ser controlados através de um joystick.

Depois que você consegue chegar ao final do labirinto, o limite de tempo é reduzido e o jogo começa de novo.

O jogo utiliza uma técnica sofisticada, baseada nos ponteiros de cores do MSX, para conseguir uma mudança rápida das imagens. Vale a pena estudar esta técnica com atenção, pois você poderá usá-la para conseguir efeitos de alta velocidade em seus próprios programas.

```

1000 REM <<< LABIRINTO - BOOTSY >>>
1010 REM
1020 SCREEN 1,,0:CLEAR 500,&HEEFF
1030 DEFUSR=&HEF00:KEY OFF:WIDTH 32
1040 ON INTERVAL=50 GOSUB 1230
1050 DIM S$(11),C$(13),W%(18,2,1),M%(9,
9),MT%(9,9):TT%=400
1060 LOCATE 13,11:PRINT "ESPERE":GOSUB
2040
1070 GOSUB 1760:GOSUB 1180:INTERVAL ON
1080 GOSUB 1430
1090 I$=INKEY$:FOR I%=0 TO 2
1100 IP%=STICK(I%):IF IP% THEN I%=2
1110 NEXT I%:IF IP%=0 THEN 1090
1120 IF IP%=1 AND M%(X%,Y%)=2 THEN GOTO
1320
1130 IF IP%=1 AND Y%<9 THEN IF M%(X%,Y%
+1)>0 THEN Y%=Y%+1:GOSUB 1180
1140 IF IP%=5 AND Y%>0 THEN IF M%(X%,Y%
-1)>0 THEN Y%=Y%-1:GOSUB 1180
1150 IF IP%=3 THEN SP%=(SP%+1) AND 3:GO
SUB 1180:GOSUB 1560
1160 IF IP%=7 THEN SP%=(SP%-1) AND 3:GO
SUB 1180:GOSUB 1610
1170 GOTO 1080
1180 IF IP%=3 OR IP%=7 THEN 1210
1190 XS%=XS%+(3 AND (IP%=1 AND SP%=1 OR
IP%=5 AND SP%=3))-(3 AND (IP%=5 AND SP
%=1 OR IP%=1 AND SP%=3))
1200 YS%=YS%+(3 AND (IP%=1 AND SP%=2 OR
IP%=5 AND SP%=0))-(3 AND (IP%=5 AND SP
%=2 OR IP%=1 AND SP%=0))
1210 PUT SPRITE 0,(XS%,YS%),4,SP%:RETUR
N

```

```

1220 REM <<< RELOGIO >>>
1230 TM% = TM% + 1 : LOCATE 23, 22 : PRINT USING
    "#####"; TM%
1240 IF TM% < TT% THEN RETURN
1250 REM <<< FIM DO JOGO >>>
1260 INTERVAL OFF
1270 CLS : LOCATE 9, 11
1280 PRINT "TARDE DEMAIS!"
1290 LOCATE 6, 13 : PRINT "O LOUCO PEGOU V
OCE!"
1300 RETURN 1380
1310 REM <<< SAIDA >>>
1320 CLS : INTERVAL OFF : COLOR 15
1330 LOCATE 11, 11 : PRINT "MUITO BEM!" : PR
INT
1340 PRINT " CONSEGUIU SAIR DO LABIRIN
TO!"
1350 LOCATE 5, 15
1360 PRINT "VOCE LEVOU"; TM%; "SEGUNDOS"
1370 IF TM% < TT% THEN TT% = TM%
1380 PUT SPRITE 0, (128, 208), 4, 0
1390 LOCATE 3, 20 : PRINT "QUER JOGAR DE N
OVO (S/N)?"
1400 A$ = INPUT$(1) : IF A$ = "S" OR A$ = "s" T
HEN GOTO 1070
1410 IF A$ = "N" OR A$ = "n" THEN SCREEN 0 :
COLOR 15, 4, 4 : STOP ELSE 1390
1420 REM <<< GERAR PERSPECTIVA >>>
1430 FOR W% = 2 TO 0 STEP -1 : V% = Y% + 3 - W%
1440 IF V% = 10 THEN GOSUB 1690 : K% = W% : W% =
0
1450 IF V% < 10 THEN IF M%(X%, V%) = 0 THEN
GOSUB 1690 : K% = W% : W% = 0
1460 NEXT W% : O% = 0 : IF M%(X%, Y% + 2 - K%) = 2 T
HEN O% = 1 : W% = K% : GOSUB 1690
1470 IF X% = 9 THEN FOR D% = 3 TO 1 + K% STEP
-1 : OC% = 0 : GOSUB 710 : NEXT D% : GOTO 1510
1480 V% = 0 : FOR D% = 3 TO 1 + K% STEP -1 : OC% =
0

```

```

1490 IF MX(X%+1,Y%+V%)>0 THEN OC%=1
1500 GOSUB 1710:V%=V%+1:NEXT DX
1510 IF X%=0 THEN FOR D%=6 TO 4+K% STEP -1:OC%=0:GOSUB 1710:NEXT D%:GOTO 1540
1520 V%=0:FOR D%=6 TO 4+K% STEP -1:OC%=
0:IF MX(X%-1,Y%+V%)>0 THEN OC%=1
1530 GOSUB 1710:V%=V%+1:NEXT DX
1540 RETURN
1550 REM <<<< VIRAR A DIREITA >>>>
1560 FOR I%=0 TO 9:FOR J%=0 TO 9
1570 MT%(J%,I%)=MX(I%,9-J%)
1580 NEXT J%,I%
1590 SWAP X%,Y%:X%=9-X%:GOTO 1650
1600 REM <<<< VIRAR A ESQUERDA >>>>
1610 FOR I%=0 TO 9:FOR J%=0 TO 9
1620 MT%(J%,9-I%)=MX(I%,J%)
1630 NEXT J%,I%
1640 SWAP X%,Y%:Y%=9-Y%
1650 FOR I%=0 TO 9:FOR J%=0 TO 9
1660 MX(J%,I%)=MT%(J%,I%)
1670 NEXT J%,I%:RETURN
1680 REM <<<< DESENHAR PAREDE >>>>
1690 A=USR(VARPTR(W%(0,W%,0%))):RETURN
1700 REM <<<< DESENHAR PORTA >>>>
1710 C1%=0:C2%=128-(32 AND(D%=2 OR D%=5))
1720 I%=8204+D%:J%=8210+D%
1730 IF OC% THEN VPOKE I%,C1%:VPOKE J%,
C2% ELSE C1%=144:C2%=C1%:OC%=1:GOTO 173
0
1740 RETURN
1750 REM <<<< TELA >>>>
1760 RESTORE 2510:O%=0:SP%=1:TM%=0:X$%=
113:YS%=156
1770 FOR I%=0 TO 9:FOR J%=0 TO 9
1780 READ MX(I%,J%)
1790 NEXT J%,I%:X%=0:Y%=0
1800 COLOR 0,0,0:CLS
1810 RESTORE 2620

```


```
1820 GOSUB 2010:FOR I%=0 TO 11
1830 LOCATE 0,I%:FOR J%=1 TO 16
1840 K%=ASC(MID$(S$(I%),J%,1))
1850 PRINT C$(K%);:NEXT J%,I%
1860 GOSUB 2010:FOR I%=0 TO 11
1870 LOCATE 16,I%:FOR J%=16 TO 1 STEP -
1
1880 K%=ASC(MID$(S$(I%),J%,1))
1890 PRINT C$(K%);:NEXT J%,I%
1900 GOSUB 2010:FOR I%=11 TO 0 STEP -1
1910 LOCATE 0,23-I%:FOR J%=1 TO 16
1920 K%=ASC(MID$(S$(I%),J%,1))
1930 PRINT C$(K%);:NEXT J%,I%
1940 GOSUB 2010:FOR I%=11 TO 0 STEP -1
1950 LOCATE 16,23-I%:FOR J%=16 TO 1 STE
P -1
1960 K%=ASC(MID$(S$(I%),J%,1))
1970 PRINT C$(K%);:NEXT J%,I%
1980 LOCATE 5,21:COLOR 15:PRINT "LIMITE"
" TAB(23) "TEMPO"
1990 PRINT TAB(6);TT%
2000 RETURN
2010 FOR I%=1 TO 13:READ K%:C$(I%)=CHR$(
K%):NEXT I%
2020 RETURN
2030 REM <<< DEFINIR FORMAS >>>
2040 FOR I%=832 TO 1984 STEP 64
2050 FOR J%=0 TO 7
2060 K%=256-2^(7-J%):L%=2^(J%+1)-1
2070 M%=2^(8-J%)-1:N%=256-2^J%
2080 VPOKE I%+J%,255:VPOKE I%+J%+8,K%
2090 VPOKE J%+J%+16,L%:VPOKE I%+J%+24,M%
%
2100 VPOKE I%+J%+32,N%:VPOKE I%+J%+40,0
2110 NEXT J%,I%
2120 FOR I%=0 TO 11:FOR J%=1 TO 16
2130 READ K%:S$(I%)=S$(I%)+CHR$(K%)
2140 NEXT J%,I%
2150 FOR I%=0 TO 1:FOR J%=0 TO 2:FOR K%
=0 TO 18
```

```
2160 READ W%(K%,J%,I%):NEXT K%,J%,I%
2170 FOR I%=0 TO 3:S$="":FOR J%=0 TO 7
2180 READ K%:S$=S$+CHR$(K%):NEXT J%
2190 SPRITE$(I%)=S$:NEXT I%
2200 I%=&HEF00
2210 READ H$:IF H$="x" THEN RETURN
2220 POKE I%,VAL("&h"+H$)
2230 I%=I%+1:GOTO 2210
2240 REM <<< DADOS >>>
2250 DATA 8,1,1,1,1,1,1,1,1,1,1,1,1,1,1,1
,1
2260 DATA 9,8,1,1,1,1,1,1,1,1,1,1,1,1,1,1
,1
2270 DATA 9,9,8,1,1,1,1,1,1,1,1,1,1,1,1,1
,1
2280 DATA 9,9,9,8,1,1,1,1,1,1,1,1,1,1,1,1
,1
2290 DATA 5,5,5,5,10,2,2,2,2,2,2,2,2,2,2
,2
2300 DATA 5,5,5,5,11,10,2,2,2,2,2,2,2,2,2
,2
2310 DATA 5,5,5,5,11,11,10,2,2,2,2,2,2,2,2
,2
2320 DATA 5,5,5,5,6,6,6,12,3,3,3,3,3,3,3
,3
2330 DATA 5,5,5,5,6,6,6,13,12,3,3,3,3,3,3
,3
2340 DATA 5,5,5,5,6,6,6,7,7,4,4,4,4,4,4
,4
2350 DATA 5,5,5,5,6,6,6,7,7,4,4,4,4,4,4
,4
2360 DATA 5,5,5,5,6,6,6,7,7,4,4,4,4,4,4
,4
2370 DATA 144,144,144,144,144,144,144,144,1
44,144,144,144,144,0,0,0,0,0,96
2380 DATA 102,144,144,102,144,144,96,14
4,144,96,144,144,6,0,0,6,0,0,96
2390 DATA 102,102,144,102,102,144,96,96
,144,96,96,144,6,6,0,6,6,0,96
```

```
2400 DATA 144,144,144,144,144,144,144,144,1  
44,144,144,144,144,0,0,0,0,0,0  
2410 DATA 0,144,144,0,144,144,0,144,144  
,0,144,144,0,0,0,0,0,0  
2420 DATA 0,0,144,0,0,144,0,0,144,0,0,1  
44,0,0,0,0,0,0  
2430 REM <<< SETA >>>  
2440 DATA 16,56,68,130,16,16,16,16  
2450 DATA 8,4,2,255,2,4,8,0  
2460 DATA 16,16,16,16,130,68,56,16  
2470 DATA 16,32,64,255,64,32,16,0  
2480 REM <<< PROGRAMA >>>  
2490 DATA 23,23,5e,23,56,21,d,20,cd,53,  
0,e3,e3,eb,6,13,7e,d3,98,23,23,10,f9,c9  
,x  
2500 REM <<< LABIRINTO >>>  
2510 DATA 1,1,1,0,1,1,1,0,1,1  
2520 DATA 1,0,1,1,1,0,1,1,1,0  
2530 DATA 1,1,0,1,0,1,1,0,1,1  
2540 DATA 0,1,0,1,1,1,0,0,0,1  
2550 DATA 1,1,1,0,0,0,1,1,1,0  
2560 DATA 0,0,1,0,1,1,1,0,1,0  
2570 DATA 1,1,1,0,0,1,0,1,1,1  
2580 DATA 1,0,0,1,1,1,0,1,0,1  
2590 DATA 1,1,1,0,1,0,1,1,1,0  
2600 DATA 0,0,1,1,1,0,1,0,1,2  
2610 REM <<< CARACTERES GRAFICOS >>>  
2620 DATA 245,237,229,248,192,184,176,1  
45,144,137,136,129,128  
2630 DATA 245,237,229,248,168,160,152,1  
22,120,114,112,106,104  
2640 DATA 221,213,205,248,192,184,176,1  
48,144,140,136,132,128  
2650 DATA 221,213,205,248,168,160,152,1  
23,120,115,112,107,104
```

7

Damas



Um Tabuleiro Eletrônico para Jogar Damas

A idéia é muito simples: você deve acabar com as peças do adversário saltando sobre elas com suas próprias peças. Para poder executar o salto, é preciso que haja uma casa vazia do outro lado da peça adversária.

Para movimentar uma peça, basta entrar com uma letra para indicar a coluna em que a peça se encontra e um número para indicar a linha. Em seguida, a casa para onde a peça vai deve ser indicada da mesma forma.

Quando não é possível saltar sobre nenhuma peça adversária, uma das peças deve ser movimentada para uma casa vizinha da mesma cor. Quando uma peça atinge a extremidade oposta do tabuleiro, ela se transforma em uma *dama*. As damas podem andar para trás e são, portanto, peças mais importantes que as peças comuns (nesta versão do jogo, ao

contrário da versão mais conhecida, as damas não podem andar mais de uma casa de cada vez).

Algumas pessoas usam uma regra um pouco diferente. Em vez de exigirem que o jogador tome uma peça do adversário sempre que isso for possível, estabelecem que o jogador perca essa peça caso execute uma jogada normal de deslocamento com outra de suas peças. Esta penalidade é chamada de “soprar” a peça adversária.

Nesta versão para o MSX, o programa se recusa a aceitar uma jogada normal quando há uma peça do adversário que possa ser tomada. O computador continua a mostrar que é a vez do mesmo jogador até que ele execute uma jogada de salto.

```

1000 REM <<< DAMAS - ISSI >>>
1010 SCREEN 1,0,0
1020 CLEAR 2000
1030 KEY OFF
1040 RESTORE
1050 FOR CH=182 TO 199
1060 FOR N=0 TO 7
1070 READ B
1080 IF CH<190 THEN B=B XOR 255
1090 VPOKE 8*CH+N,B
1100 NEXT N
1110 NEXT CH
1120 VPOKE &H2016,31:VPOKE &H2017,31
1130 GOTO 1350
1140 REM -----
1150 DATA 255,255,255,255,240,143,127,3
1
1160 DATA 255,255,255,255,15,241,254,20
6
1170 DATA 66,80,85,85,133,240,255,255
1180 DATA 194,30,226,30,241,15,255,255
1190 DATA 255,255,255,255,240,128,0,0
1200 DATA 255,255,255,255,15,1,0,48
1210 DATA 67,80,85,85,133,240,255,255
1220 DATA 194,28,224,0,1,15,255,255
1230 REM
1240 DATA 0,1,9,87,99,118,125,255
1250 DATA 0,128,144,234,198,110,58,187

```

```
1260 DATA 0,1,9,86,98,86,68,128
1270 DATA 0,128,144,106,70,106,26,1
1280 DATA 0,0,15,16,32,35,36,36
1290 DATA 0,0,240,8,4,196,36,36
1300 DATA 36,36,35,32,16,15,0,0
1310 DATA 36,36,196,4,8,240,0,0
1320 DATA 0,0,255,0,0,255,0,0
1330 DATA 36,36,36,36,36,36,36,36
1340 REM -----
1350 CLS:GOSUB 2080
1360 FOR N=1 TO 100
1370 Z$(N,1)=""
1380 Z$(N,2)=""
1390 NEXT N
1400 DIM C(8,8)
1410 GOSUB 1740
1420 GOSUB 2120:GOSUB 2280
1430 LOCATE 5,19:PRINT STRING$(14," ")
1440 LOCATE 5,20:PRINT STRING$(14," ")
1450 COU=1
1460 GOTO 1600
1470 GAM=0
1480 FOR N=1 TO 8
1490 FOR M=1 TO 8
1500 IF M$(N,M)="B" OR M$(M,N)="D" THEN
    GAM=1
1510 NEXT M:NEXT N
1520 IF GAM=0 THEN GOTO 1630
1530 GAM=0
1540 FOR N=1 TO 8
1550 FOR M=1 TO 8
1560 IF M$(N,M)="A" OR M$(M,N)="C" THEN
    GAM=1
1570 NEXT M:NEXT N
1580 IF GAM=0 THEN GOTO 1670
1590 RETURN
1600 GOSUB 3550
1610 GOTO 1600
1620 REM <<< PRETAS VENDEM >>>
```

```
1630 CLS
1640 LOCATE 4,9:PRINT "AS PRETAS VENCER
AM"
1650 GOTO 1690
1660 REM <<< BRANCAS VENCEM >>>
1670 CLS
1680 LOCATE 4,9:PRINT "AS BRANCAS VENCE
RAM"
1690 PLAY "L8CDEFGCDEFG"
1700 LOCATE 1,23:PRINT "Quer jogar de n
ovo (S/N) ?"
1710 A$=INPUT$(1):A$=CHR$(ASC(A$) AND 2
23)
1720 IF A$="S" THEN RUN ELSE SCREEN 0:S
TOP
1730 REM <<< TELA >>>
1740 COLOR 15,4,4
1750 FOR A=0 TO 7:VPOKE 8*214+A,255:NEX
T A
1760 FOR Y=2 TO 17 STEP 4
1770 FOR X=2 TO 17 STEP 4
1780 LOCATE X,Y:PRINT CHR$(214);CHR$(21
4);"
1790 LOCATE X,Y+1:PRINT CHR$(214);CHR$(214);"
1800 LOCATE X,Y+2:PRINT " ";CHR$(214);
CHR$(214)
1810 LOCATE X,Y+3:PRINT " ";CHR$(214);
CHR$(214)
1820 C(X/2,Y/2)=1:C(X/2+1,Y/2+1)=1
1830 C(X/2,Y/2+1)=0:C(X/2+1,Y/2+1)=1
1840 NEXT X
1850 NEXT Y
1860 FOR X=2 TO 17
1870 LOCATE X,1:PRINT CHR$(198)
1880 LOCATE X,18:PRINT CHR$(198)
1890 NEXT X
1900 FOR Y=2 TO 17
1910 LOCATE 1,Y:PRINT CHR$(199)
```

```
1920 LOCATE 18,Y:PRINT CHR$(199)
1930 NEXT Y
1940 LOCATE 1,1:PRINT CHR$(194)
1950 LOCATE 18,1:PRINT CHR$(195)
1960 LOCATE 1,18:PRINT CHR$(196)
1970 LOCATE 18,18:PRINT CHR$(197)
1980 FOR Y=2 TO 16 STEP 2
1990 LOCATE Y,0:PRINT CHR$((Y/2)+96)
2000 LOCATE 19,Y:PRINT CHR$(48+9-Y/2)
2010 NEXT Y
2020 LOCATE 21,0:PRINT " DAMAS "
2030 LOCATE 21,1:PRINT "===== "
2040 LOCATE 21,6:PRINT "De :"
2050 LOCATE 21,8:PRINT "Para:"
2060 RETURN
2070 REM <<< INICIALIZAÇÃO >>>
2080 DIM M$(8,8)
2090 DIM Z$(100,2)
2100 RETURN
2110 REM <<< INICIO DO JOGO >>>
2120 FOR N=1 TO 8
2130 FOR M=1 TO 8
2140 M$(N,M)="0"
2150 X$(N,M)="0"
2160 NEXT M
2170 NEXT N
2180 FOR N=1 TO 8 STEP 2
2190 M$(1,N+1)="A"
2200 M$(2,N)="A"
2210 M$(3,N+1)="A"
2220 M$(8,N)="B"
2230 M$(7,N+1)="B"
2240 M$(6,N)="B"
2250 NEXT N
2260 RETURN
2270 REM <<< DESENHAR PEÇAS >>>
2280 X=2:Y=2
2290 FOR N=1 TO 8
2300 FOR M=1 TO 8
```

```
2310 C$=M$(N,M)
2320 IF C$="" THEN 2350
2330 LOCATE X,Y
2340 GOSUB 2410
2350 X=X+2
2360 NEXT M
2370 X=2:Y=Y+2
2380 NEXT N
2390 RETURN
2400 REM <<< PEÇA >>>
2410 LOCATE X,Y
2420 D$=CHR$(182)+CHR$(183):D1$=CHR$(184)+CHR$(185)
2430 IF C$="B" THEN D$=CHR$(186)+CHR$(187):D1$=CHR$(188)+CHR$(189)
2440 IF C$="C" THEN D$=CHR$(192)+CHR$(193)
2450 IF C$="D" THEN D$=CHR$(190)+CHR$(191):D1$=CHR$(188)+CHR$(189)
2460 IF C$="" THEN D$="":D1$=""
2470 PRINT D$;
2480 LOCATE X,Y+1
2490 PRINT D1$;
2500 RETURN
2510 REM <<< ENTRADA DE DADOS >>>
2520 LOCATE 21,3:IF PLA=1 THEN PRINT "BRANCAS" ELSE PRINT "PRETAS"
2530 LOCATE 26,6:PRINT " "
2540 LOCATE 26,8:PRINT " "
2550 LOCATE 26,6
2560 GOSUB 2620
2570 F$=N$
2580 LOCATE 26,8
2590 GOSUB 2620
2600 S$=N$
2610 RETURN
2620 N$=""
2630 A$=INKEY$:IF A$="" THEN GOTO 2630
2640 A$=CHR$(ASC(A$) AND 223)
```

```

2650 IF A$<"A" OR A$>"H" THEN 2630
2660 PLAY "L8C":N$=A$ :PRINT N$;
2670 A$=INKEY$
2680 IF A$<"1" OR A$>"9" THEN 2670
2690 PLAY "L8F":N$=N$+A$ :PRINT A$;
2700 LOCATE 2,22:PRINT STRING$(16,32)
2710 RETURN
2720 REM <<< MOVIMENTO >>>
2730 GOSUB 2950
2740 R0=S1:NC=S:GOSUB 2820
2750 M$(S1,S)=M$(F1,F)
2760 IF S1=1 AND M$(1,S)="B" THEN M$(1,
S)="D":HUFF=0
2770 IF S1=8 AND M$(8,S)="A" THEN M$(8,
S)="C":HUFF=0
2780 R0=S1:NC=S:GOSUB 2820
2790 M$(F1,F)="0"
2800 R0=F1:NC=F:GOSUB 2820
2810 RETURN
2820 X=2:Y=2
2830 FOR N=1 TO 8
2840 FOR M=1 TO 8
2850 IF N=R0 AND M=NC THEN GOSUB 2910
2860 X=X+2
2870 NEXT M
2880 X=2:Y=Y+2
2890 NEXT N
2900 RETURN
2910 C$=M$(N,M)
2920 GOSUB 2410
2930 RETURN
2940 REM <<< CONVERTER >>>
2950 F=ASC(LEFT$(F$,1))
2960 F=F-64
2970 F1=VAL(RIGHT$(F$,1))
2980 F1=9-F1
2990 S=ASC(LEFT$(S$,1))
3000 S=S-64
3010 S1=VAL(RIGHT$(S$,1))

```

```
3020 S1=9-S1
3030 RETURN
3040 REM <<< RECONVERTER >>>
3050 R$=CHR$(NC+64)
3060 R$=R$+STR$(9-LIN)
3070 RETURN
3080 REM <<< VALIDAR >>>
3090 GOSUB 2950:HUFF=0
3100 P$="B":P1$="D"
3110 IF PLA=2 THEN P$="A":P1$="C"
3120 FOR Y=1 TO 8
3130 FOR X=1 TO 8
3140 IF HUFF=1 THEN GOTO 3170
3150 IF M$(Y,X)<>P$ AND M$(Y,X)<>P1$ THEN GOTO 3170
3160 GOSUB 3630
3170 NEXT X
3180 NEXT Y
3190 CHEC=1
3200 C$=M$(F1,F):D$=M$(S1,S)
3210 IF C$="0" THEN RETURN
3220 IF (C$="A" OR C$="C") AND PLA=1 THEN RETURN
3230 IF (C$="B" OR C$="D") AND PLA=2 THEN RETURN
3240 IF D$<>"0" THEN RETURN
3250 IF C$="A" AND S1<F1 THEN RETURN
3260 IF C$="B" AND S1>F1 THEN RETURN
3270 XD=S-F:YD=S1-F1
3280 IF SGN(XD)*(XD)=1 AND SGN(YD)*(YD)=1 THEN GOTO 3390
3290 IF SGN(XD)*(XD)<>SGN(YD)*(YD) THEN RETURN
3300 X$=M$(F1+(YD/2),F+(XD/2))
3310 IF (X$="A" OR X$="C") AND PLA=2 THEN RETURN
3320 IF (X$="B" OR X$="D") AND PLA=1 THEN RETURN
3330 IF X$="0" THEN RETURN
```


```

3340 RO=F1+(YD/2):NC=F+(XD/2)
3350 M$(RO,NC)="0":GOSUB 2820
3360 GOSUB 3050
3370 Z$(COU,1)=R$:Z$(COU,2)=R$
3380 COU=COU+1:GOTO 3400
3390 IF HUFF=1 THEN RETURN
3400 CHEC=0:RETURN
3410 REM <<< JOGADA >>>
3420 GOSUB 2520:HUFF=0
3430 GOSUB 3090
3440 IF CHEC=0 THEN GOTO 3470
3450 LOCATE 2,22:PRINT "MOVIMENTO ILEGA
L"
3460 PLAY "L4GEC":GOTO 3420
3470 GOSUB 2730
3480 Z$(COU,1)=F$:Z$(COU,2)=S$
3490 COU=COU+1:IF COU=101 THEN COU=1
3500 IF HUFF=0 THEN GOTO 3530
3510 Y=S1:X=S:GOSUB 3630
3520 IF HUFF=1 THEN GOTO 3780
3530 HUFF=0:RETURN
3540 REM <<< ALTERNAR JOGADORES >>>
3550 PLA=1
3560 GOSUB 3420
3570 GOSUB 1470
3580 PLA=2
3590 GOSUB 3420
3600 GOSUB 1470
3610 RETURN
3620 REM <<< COMER >>>
3630 HUFF=0
3640 H1$="A":H2$="C"
3650 IF PLA=2 THEN H1$="B":H2$="D"
3660 IF Y<3 OR M$(Y,X)="A" THEN GOTO 37
10
3670 IF X<3 THEN GOTO 3690
3680 IF (M$(Y-1,X-1)=H1$ OR M$(Y-1,X-1)
=H2$) AND M$(Y-2,X-2)="0" THEN HUFF=1
3690 IF X>6 THEN GOTO 3710

```

```
3700 IF (M$(Y-1,X+1)=H1$ OR M$(Y-1,X-1)
=H2$) AND M$(Y-2,X+2)="0" THEN HUFF=1
3710 IF Y>6 OR M$(Y,X)="B" THEN GOTO 37
60
3720 IF X<3 THEN GOTO 3740
3730 IF (M$(Y+1,X-1)=H1$ OR M$(Y+1,X-1)
=H2$) AND M$(Y+2,X-2)="0" THEN HUFF=1
3740 IF X>6 THEN GOTO 3760
3750 IF (M$(Y+1,X+1)=H1$ OR M$(Y+1,X+1)
=H2$) AND M$(Y+2,X+2)="0" THEN HUFF=1
3760 RETURN
3770 REM <<< COMER DE NOVO >>>
3780 LOCATE 26,6:PRINT S$
3790 F$=S$
3800 LOCATE 26,8:PRINT " "
3810 LOCATE 26,8
3820 GOSUB 2620
3830 S$=N$
3840 LOCATE 2,22:PRINT STRING$(16,32)
3850 GOSUB 2950:GOSUB 3190
3860 IF CHEC=0 THEN GOTO 3890
3870 LOCATE 2,22:PRINT "MOVIMENTO ILEGA
L"
3880 PLAY "L8GECGEC":GOTO 3800
3890 GOSUB 2730
3900 Z$(COU,1)=F$:Z$(COU,2)=S$
3910 COU=COU+1:IF COU=101 THEN COU=1
3920 Y=S1:X=S:GOSUB 3630
3930 IF HUFF=1 THEN GOTO 3780
3940 HUFF=0:GOTO 3530
```

Xadrez



Um Tabuleiro Eletrônico para Jogar Xadrez

Você provavelmente já ouviu falar de computadores que jogam xadrez. Um programa inteligente para jogar xadrez, porém, seria grande demais para caber neste livro, de modo que preferimos criar um programa que fosse útil como “tabuleiro eletrônico”. O programa pode ser usado para substituir um tabuleiro comum, com a vantagem de que não é necessário anotar as jogadas. Os dois jogadores entram com os lances usando a notação normal do xadrez, isto é, letras para indicar as colunas e números para indicar as linhas.

O computador examina cada lance para verificar se é possível dentro das regras do xadrez. Apenas são permitidos os movimentos de peças da cor correta, de uma casa ocupada para uma casa que possa ser atingida por essa peça. As peças tomadas são automaticamente removidas do jogo.

O programa não indica xeque nem xeque-mate e permite jogadas que deixem o rei em xeque; cabe aos jogadores tomar cuidado para evitar que isso aconteça. O programa também não prevê a tomada de peões *en passant*; a opção “remover peça” deve ser usada para retirar um peão do tabuleiro se ele for tomado *en passant*.

O programa executa corretamente os dois tipos de roque, verificando primeiro se o rei e a torre estão nas suas posições iniciais e se não existe nenhuma peça entre eles. Entretanto, o computador aceita o roque se uma das peças já foi movimentada e voltou à posição inicial ou se uma das casas entre o rei e a torre estiverem em xeque. Cabe aos jogadores evitar uma jogada de roque nesses últimos dois casos.

O programa também permite fixar uma posição inicial qualquer para as peças, o que é útil para o estudo de problemas de xadrez ou variantes de aberturas e finais. Além disso, jogos de até 100 lances podem ser salvos em fita e analisados jogada por jogada. Os peões são automaticamente promovidos a damas quando chegam à última linha.

Resumo dos comandos e opções

1. Para mudar a posição inicial das peças: Quando o computador perguntar “Você quer montar o tabuleiro?”, entre com “S” e indique a posição das peças, da esquerda para a direita e de cima para baixo, usando o seguinte código:

Peças brancas: P = peão, T = torre, C = cavalo, R = rei, D = dama, B = bispo.

Peças pretas: o mesmo código, mas usando letras minúsculas.

Casas vazias: aperte a barra de espaço.

As brancas sempre começam na parte de baixo da tela.

2. Para remover uma peça: Entre com **CODE-R** e a posição da peça. A vez passa para o adversário.
3. Para fazer o roque: Entre com **R** e depois com **E** ou **D**, conforme o caso.
4. Para salvar uma partida: Entre com **GRAPH-S**, depois de apertar as teclas **RECORD** e **PLAY** do gravador, se estiver usando fita. Se estiver usando disquete, indique a unidade de disco em que se encontra o seu disquete de dados (se quiser, você também pode salvar o jogo no disquete de programas).

5. Para carregar uma partida: Entre com **GRAPH-C** e depois aperte a tecla **PLAY** do gravador. O computador carregará o primeiro arquivo que encontrar na fita. Se estiver usando disquete, o computador carregará a última partida que foi salva, que estará guardada no arquivo de dados chamado *Xadrez*. Depois de carregar a partida, o computador entra automaticamente no modo de repetição (vide item seguinte).

6. Para repetir uma partida: Para entrar no modo de repetição, entre com **GRAPH-J**. Cada vez que a barra de espaço for apertada, o computador mostrará uma nova jogada, a partir da inicial. Para voltar ao modo normal, basta entrar de novo com **GRAPH-J**.

7. Para abandonar uma partida: Entre com **GRAPH-A**. O computador lhe oferece a oportunidade de salvar a partida em fita ou disquete antes de começar um novo jogo.

```

1000 REM <<<< XADREZ - ISSI/BOOTSY/ANDY
      >>>>
1010 SCREEN 1,0,0:CLEAR 2000
1020 COLOR 15,4,4:KEY OFF
1030 DIM C(8,8),M$(8,8),X$(8,8),Z$(100,
2)
1040 LOCATE 10,10:PRINT "ESPERE":LOCATE
8,12:PRINT "15 SEGUNDOS"
1050 GOSUB 5740:GOSUB 5190
1060 GOSUB 4510
1070 LOCATE 1,19:PRINT "Você quer monta
r"
1080 LOCATE 1,20:PRINT "o tabuleiro (S/
N)?""
1090 A$=INPUT$(1):BEEP
1100 LOCATE 1,19:PRINT STRING$(16," ")
1110 LOCATE 1,20:PRINT STRING$(18," ")
1120 IF A$=="s" OR A$=="S" THEN GOSUB 386
0:IF CHE=0 THEN 1150 ELSE 1060
1130 IF A$=="n" OR A$=="N" THEN 1140 ELSE
1070
1140 GOSUB 1200:GOSUB 1420
1150 COU=1
1160 GOSUB 3800
1170 GOTO 1160
1180 END

```

```

1190 REM <<< INICIO DO JOGO >>>
1200 FOR N=1 TO 8
1210 FOR M=1 TO 8
1220 M$(N,M)="0"
1230 X$(N,M)="0"
1240 NEXT M
1250 NEXT N
1260 FOR N=1 TO 8
1270 M$(2,N)="P":X$(2,N)="P"
1280 M$(7,N)="P":X$(7,N)="P"
1290 NEXT N
1300 RESTORE 6340
1310 FOR N=1 TO 8
1320 READ A$
1330 M$(8,N)=A$
1340 X$(8,N)=A$
1350 C$=CHR$(ASC(A$)+32):A$=C$
1360 M$(1,N)=A$
1370 X$(1,N)=A$
1380 NEXT N
1390 GOSUB 5690
1400 RETURN
1410 REM <<< DESENHAR PEÇAS >>>
1420 X=2:Y=2
1430 FOR N=1 TO 8
1440 FOR M=1 TO 8
1450 C$=M$(N,M)
1460 IF C$="0" THEN 1490
1470 LOCATE X,Y
1480 GOSUB 1550
1490 X=X+2
1500 NEXT M
1510 X=2:Y=Y+2
1520 NEXT N
1530 RETURN
1540 <<< PEÇA >>>
1550 LOCATE X,Y:XX=X:X=X/2:YY=Y:Y=Y/2:I
F C$="K" AND C(X,Y)=1 THEN D$=K$:D1$=K1
$: ELSE IF C$="K" THEN D$=K2$:D1$=K3$

```

```

1560 IF C$="k" AND C(X,Y)=1 THEN D$=K4$:  

:D1$=K5$ ELSE IF C$="k" THEN D$=K6$:D1$=  

=K7$  

1570 IF C$="Q" AND C(X,Y)=1 THEN D$=Q$:D1$=  

Q5$ ELSE IF C$="Q" THEN D$=Q2$:D1$=  

Q3$  

1580 IF C$="q" AND C(X,Y)=1 THEN D$=Q4$:  

:D1$=Q5$ ELSE IF C$="q" THEN D$=Q6$:D1$=  

=Q7$  

1590 IF C$="B" AND C(X,Y)=1 THEN D$=B$:D1$=  

B1$ ELSE IF C$="B" THEN D$=B2$:D1$=  

B3$  

1600 IF C$="b" AND C(X,Y)=1 THEN D$=B4$:  

:D1$=B5$ ELSE IF C$="b" THEN D$=B6$:D1$=  

=B7$  

1610 IF C$="T" AND C(X,Y)=1 THEN D$=T$:D1$=  

T1$ ELSE IF C$="T" THEN D$=T2$:D1$=  

T3$  

1620 IF C$="t" AND C(X,Y)=1 THEN D$=T4$:  

:D1$=T5$ ELSE IF C$="t" THEN D$=T6$:D1$=  

=T7$  

1630 IF C$="R" AND C(X,Y)=1 THEN D$=R$:D1$=  

R1$ ELSE IF C$="R" THEN D$=R2$:D1$=  

R3$  

1640 IF C$="r" AND C(X,Y)=1 THEN D$=R4$:  

:D1$=R5$ ELSE IF C$="r" THEN D$=R6$:D1$=  

=R7$  

1650 IF C$="P" AND C(X,Y)=1 THEN D$=P$:D1$=  

P1$ ELSE IF C$="P" THEN D$=P2$:D1$=  

P3$  

1660 IF C$="p" AND C(X,Y)=1 THEN D$=P4$:  

:D1$=P5$ ELSE IF C$="p" THEN D$=P6$:D1$=  

=P7$  

1670 IF C$="Ø" AND C(X,Y)=1 THEN D$=CHR$  

(214)+CHR$(214):D1$=D$ ELSE IF C$="Ø"  

THEN D$=" " :D1$=D$  

1680 PRINT D$;  

1690 X=XX:Y=YY  

1700 LOCATE X,Y+1

```

```
1710 PRINT D1$;
1720 RETURN
1730 REM <<< ENTRADA DE DADOS >>>
1740 LOCATE 21,3:IF PLA=1 THEN PRINT "BRANCAS" ELSE PRINT "PRETAS"
1750 LOCATE 26,6:PRINT " "
1760 LOCATE 26,8:PRINT " "
1770 LOCATE 26,6
1780 GOSUB 1850
1790 F$=N$
1800 IF CAS=1 THEN RETURN
1810 LOCATE 26,8
1820 GOSUB 1850
1830 S$=N$
1840 RETURN
1850 N$="" :CAS=0
1860 A$=INPUT$(1)
1870 IF A$=CHR$(196) THEN GOSUB 2030
1880 IF A$="R" OR A$="r" THEN GOSUB 3260
1890 IF CAS=1 THEN RETURN
1900 IF A$=CHR$(188) THEN GOSUB 4910:LOCATE 26,6
1910 IF A$=CHR$(210) THEN GOSUB 4730:LOCATE 26,6
1920 IF A$=CHR$(156) THEN GOTO 2320
1930 A$=CHR$(ASC(A$) AND 223)
1940 IF A$=CHR$(198) AND COU>1 THEN Z$(COU,1)="FF":GOTO 4240
1950 IF A$<"A" OR A$>"H" THEN GOTO 1860
1960 PLAY "L8C":N$=A$ :PRINT N$:
1970 A$=INPUT$(1)
1980 IF A$=CHR$(8) THEN PRINT CHR$(127):GOTO 1850
1990 IF A$<"1" OR A$>"9" THEN 1970
2000 PLAY "L8G":N$=N$+A$ :PRINT A$;
2010 RETURN
2020 REM <<< FIM DO JOGO >>>
2030 LOCATE 0,20:PRINT "Quer salvar a partida (S/N)? "
```

```

2040 SS$=INPUT$(1):IF SS$="S" OR SS$="s"
" THEN A$=CHR$(210):RETURN
2050 IF SS$="N" OR SS$="n" THEN LOCATE
0,20:PRINT " Quer jogar de novo (S/N)?
"
2060 RR$=INPUT$(1):IF RR$="S" OR RR$="s"
" THEN RUN ELSE SCREEN 0:STOP
2070 REM <<< MOVIMENTO >>>
2080 GOSUB 2490
2090 CH$=M$(F1,F)
2100 M$(S1,S)=M$(F1,F)
2110 IF PLA=1 AND F1=1 AND M$(F1,F)="p"
THEN M$(F1,F)="q"
2120 IF PLA=2 AND F1=8 AND M$(F1,F)="P"
THEN M$(F1,F)="Q"
2130 C$=CH$
2140 R0=S1:CR=S:GOSUB 2190
2150 M$(F1,F)="0"
2160 R0=F1:CR=F:GOSUB 2190
2170 M$(F1,F)="0"
2180 RETURN
2190 X=2:Y=2
2200 FOR N=1 TO 8
2210 FOR M=1 TO 8
2220 IF N=R0 AND M=CR THEN GOSUB 2280
2230 X=X+2
2240 NEXT M
2250 X=2:Y=Y+2
2260 NEXT N
2270 RETURN
2280 C$=M$(F1,F):LOCATE X,Y:GOSUB 1550
2290 RETURN
2300 RETURN
2310 REM <<< REMOVER PEÇA >>>
2320 LOCATE 20,19:PRINT "Peça"
2330 LOCATE 20,20:F$=INPUT$(1):A$=CHR$(ASC(F$) AND 223)
2340 IF A$<"A" OR A$>"H" THEN GOTO 2330
ELSE PRINT A$

```

```

2350 LOCATE 21,20:G$=INPUT$(1)
2360 IF G$<"1" OR G$>"9" THEN GOTO 2350
    ELSE PRINT G$
2370 F$=A$+G$
2380 LOCATE 20,19:PRINT "      "
2390 LOCATE 20,20:PRINT "      "
2400 S$="A1":GOSUB 2490
2410 M$(F1,F)="0"
2420 R0=F1:CR=F:GOSUB 2190
2430 Z$(COU,1)=F$:Z$(COU,2)=F$
2440 COU=COU+1
2450 IF COU=101 THEN COU=1
2460 IF PLA=1 THEN PLA=2:GOSUB 3650
2470 RETURN 1160
2480 REM <<< CONVERTER >>>
2490 F=ASC(LEFT$(F$,1))
2500 F=F-64
2510 F1=9-VAL(RIGHT$(F$,1))
2520 S=ASC(LEFT$(S$,1))
2530 S=S-64
2540 S1=9-VAL(RIGHT$(S$,1))
2550 RETURN
2560 F$=CHR$(F+64)+STR$(9-F1)
2570 S$=CHR$(S+64)+STR$(9-S1)
2580 Z$(COU,1)=F$
2590 Z$(COU,2)=S$
2600 COU=COU+1:RETURN
2610 REM <<< VALIDAR >>>
2620 GOSUB 2490
2630 CHE=1
2640 C$=M$(F1,F):D$=M$(S1,S)
2650 IF C$="0" THEN RETURN
2660 IF C$>"Z" AND PLA=1 THEN RETURN
2670 IF C$<"a" AND PLA=2 THEN RETURN
2680 IF D$="0" THEN 2710
2690 IF D$>"Z" AND PLA=2 THEN RETURN
2700 IF D$<"a" AND PLA=1 THEN RETURN
2710 IF ASC(C$)>ASC("Z") THEN C$=CHR$(ASC(C$)-32)

```

```
2720 IF ASC(D$)>ASC("Z") THEN D$=CHR$(ASC(D$)-32)
2730 IF D$="K" THEN RETURN
2740 IF C$="P" THEN 2850
2750 IF C$="R" THEN 2980
2760 IF C$="T" THEN 3010
2770 IF C$="B" THEN 3120
2780 IF C$="Q" THEN 3170
2790 IF C$="K" THEN 3230
2800 CHE=0
2810 IF C$="K" OR C$="P" OR C$="T" THEN
    RETURN
2820 GOSUB 3540
2830 RETURN
2840 REM <<< FEAO >>>
2850 IF PLA=2 AND F1>S1 THEN RETURN
2860 IF PLA=1 AND S1>F1 THEN RETURN
2870 DIS=SGN(S1-F1)*(S1-F1):IF DIS=1 TH
EN 2930
2880 IF DIS>2 THEN RETURN
2890 IF S<>F THEN RETURN
2900 IF PLA=2 AND F1=2 AND M$(3,F)="0"
THEN 2960
2910 IF PLA=1 AND F1=7 AND M$(6,F)="0"
THEN 2960
2920 RETURN
2930 IF S=F AND D$="0" THEN 2960
2940 IF SGN(S-F)*(S-F)=1 AND D$<>"0" TH
EN 2960
2950 RETURN
2960 GOTO 2800
2970 REM <<< TORRE >>>
2980 IF F1<>S1 AND F<>S THEN RETURN
2990 GOTO 2800
3000 REM <<< CAVALO >>>
3010 IF S1=F1+2 AND S=F-1 THEN 3100
3020 IF S1=F1+2 AND S=F+1 THEN 3100
3030 IF S1=F1-2 AND S=F-1 THEN 3100
3040 IF S1=F1-2 AND S=F+1 THEN 3100
```

```

3050 IF S1=F1+1 AND S=F-2 THEN 3100
3060 IF S1=F1+1 AND S=F+2 THEN 3100
3070 IF S1=F1-1 AND S=F-2 THEN 3100
3080 IF S1=F1-1 AND S=F+2 THEN 3100
3090 RETURN
3100 GOTO 2800
3110 REM <<< BISPO >>>
3120 XD=F-S:XD=SGN(XD)*XD
3130 YD=F1-S1:YD=SGN(YD)*YD
3140 IF XD<>YD THEN RETURN
3150 GOTO 2800
3160 REM <<< DAMA >>>
3170 XD=F-S:XD=SGN(XD)*XD
3180 YD=F1-S1:YD=SGN(YD)*YD
3190 IF XD=YD THEN 3210
3200 IF F1<>S1 AND F<>S THEN RETURN
3210 GOTO 2800
3220 REM <<< REI >>>
3230 IF SGN(F1-S1)*(F1-S1)<>1 AND SGN(F-S)*(F-S)<>1 THEN RETURN
3240 GOTO 2800
3250 REM <<< ROQUE >>>
3260 PRINT "C";
3270 A=ASC(INPUT$(1)):A=A AND 223
3280 IF A<>68 AND A<>69 THEN 3270
3290 A$=CHR$(A):PRINT A$;
3300 N$="C"+A$:CAS=1:RETURN
3310 REM <<< MOVIMENTO DE ROQUE >>>
3320 S$="A8":A$=RIGHT$(F$,1):GOSUB 2490
:CHEC=1
3330 IF (PLA=1 AND A$="E" AND M$(8,1)<>"R") OR (PLA=1 AND A$="D" AND M$(8,8)<>"R") THEN RETURN
3340 IF (PLA=2 AND A$="E" AND M$(1,1)<>"r") OR (PLA=2 AND A$="D" AND M$(1,8)<>"r") THEN RETURN
3350 IF PLA=2 AND M$(1,5)<>"k" THEN RETURN
3360 IF PLA=1 AND M$(8,5)<>"K" THEN RETURN

```

```

3370 AU=1:EN=7:IF PLA=1 THEN LIN=8 ELSE
    LIN=1
3380 IF A$="E" THEN AU=-1:EN=2
3390 FOR N=5+AU TO EN STEP AU
3400 IF M$(LIN,N)<>"0" THEN N=EN:RETURN
3410 NEXT N
3420 PLAY "LBCDEGFECFG"
3430 F1=LIN:F=5+AU:S1=F1:S=EN
3440 IF AU=-1 THEN S=S+1
3450 M$(F1,F)="R":M$(S1,S)="K"
3460 M$(F1,5)="0":M$(S1,EN+AU)="0"
3470 IF LIN=1 THEN M$(F1,F)="r":M$(S1,S)
    ="k"
3480 Z$(COU,1)="CC":Z$(COU,2)="CC":COU=
    COU+1
3490 FOR F=1 TO 8
3500 R0=F1:CR=F:GOSUB 2190
3510 NEXT F
3520 CHEC=0:RETURN
3530 REM <<< *VERIFICAR CAMINHO >>>
3540 ND=0:MD=0
3550 IF F1>S1 THEN ND=-1
3560 IF S1>F1 THEN ND=1
3570 IF F>S THEN MD=-1
3580 IF S>F THEN MD=1
3590 CN=F1:CM=F
3600 CN=CN+ND:CM=CM+MD
3610 IF CN=S1 AND CM=S THEN RETURN
3620 IF M$(CN,CM)<>"0" THEN CHE=1:RETUR
N
3630 GOTO 3600
3640 REM <<< VEZ >>>
3650 GOSUB 1740
3660 LOCATE 20,11:PRINT "
3670 LOCATE 21,12:PRINT "
3680 IF CAS=1 THEN GOSUB 3320 ELSE GOSU
B 2620
3690 IF CHE=0 THEN 3730
3700 LOCATE 20,11:PRINT "MOVIMENTO"

```

```
3710 LOCATE 21,12:PRINT "ILEGAL"
3720 GOTO 3650
3730 IF CAS=1 THEN 3780
3740 Z$(COU,1)=F$
3750 Z$(COU,2)=S$
3760 COU=COU+1:IF COU=101 THEN COU=1
3770 GOSUB 2080
3780 RETURN
3790 REM <<< JOGADA COMPLETA >>>
3800 PLA=1
3810 GOSUB 3650
3820 PLA=2
3830 GOSUB 3650
3840 RETURN
3850 REM <<< MONTAR TABULEIRO >>>
3860 X=2:Y=2
3870 FOR N=1 TO 8
3880 FOR M=1' TO 8
3890 M$(N,M)="0":X$(N,M)="0"
3900 GOSUB 4040
3910 X=X+2
3920 NEXT M
3930 X=2:Y=Y+2
3940 NEXT N
3950 LOCATE 20,18:PRINT "TUDO"
3960 LOCATE 20,19:PRINT "CERTO?"
3970 LOCATE 20,20:PRINT " "
3980 A$=INKEY$:IF A$="" THEN 3980
3990 IF A$>"Z" THEN A$=CHR$(ASC(A$)-32)
4000 IF A$="S" THEN CHE=0:LOCATE 20,18:
PRINT "      ":LOCATE 20,19:PRINT "
":RETURN
4010 IF A$="N" THEN 3860
4020 GOTO 3980
4030 REM <<< ENTRADA DE DADOS >>>
4040 LOCATE X,Y:PRINT "?";
4050 D$=INKEY$:IF D$="" THEN GOTO 4050
4060 IF D$=" " THEN D$="0":GOTO 4170
4070 IF D$>"Z" THEN C$=CHR$(ASC(D$)-32)
ELSE C$=D$
```

```

4080 IF C$<>"T" AND C$<>"C" AND C$<>"B"
    AND C$<>"D" AND C$<>"R" AND C$<>"P" TH
EN GOTO 4050
4090 IF D$="r" THEN D$="k"
4100 IF D$="R" THEN D$="K"
4110 IF D$="t" THEN D$="r"
4120 IF D$="T" THEN D$="R"
4130 IF D$="d" THEN D$="q"
4140 IF D$="D" THEN D$="Q"
4150 IF D$="c" THEN D$="t"
4160 IF D$="C" THEN D$="T"
4170 PLAY "L8C":M$(N,M)=D$
4180 X$(N,M)=D$
4190 C$=D$
4200 LOCATE X,Y
4210 GOSUB 1550
4220 RETURN
4230 REM <<< REPETIR >>>
4240 PLAY "L8CEGCEG":COU=0:PLA=1
4250 FOR N=1 TO 8:FOR M=1 TO 8
4260 M$(N,M)=X$(N,M)
4270 NEXT M:NEXT N
4280 GOSUB 4510:GOSUB 1420
4290 COU=COU+1
4300 IF Z$(COU,1)="FF" THEN 4480
4310 F$=Z$(COU,1):S$=Z$(COU,2)
4320 GOSUB 2490:GOSUB 2080
4330 COU=COU+1
4340 IF Z$(COU,1)="CC" THEN 4290
4350 A$=INKEY$
4360 IF A$="" THEN 4350
4370 PLA=2:PLAY "L8DE"
4380 IF A$=CHR$(198) THEN 4480
4390 IF Z$(COU,1)="FF" THEN 4480
4400 F$=Z$(COU,1):S$=Z$(COU,2)
4410 GOSUB 2490:GOSUB 2080
4420 IF Z$(COU+1,1)="CC" THEN COU=COU+2
    :GOTO 4390
4430 A$=INKEY$

```

```

4440 IF A$="" THEN 4430
4450 PLA=1:PLAY "L8DE"
4460 IF A$=CHR$(198) THEN 4480
4470 GOTO 4290
4480 IF PLA=1 THEN 1160
4490 GOSUB 3650:GOTO 1160
4500 REM <<< TELA >>>
4510 COLOR 15,4,4
4520 FOR Y=2 TO 16 STEP 2
4530 LOCATE 1,Y:PRINT CHR$(220);BD$;CHR$((220))
4540 LOCATE 1,Y+1:PRINT CHR$(220);BD$;CHR$((220))
4550 BD$=RIGHT$(BD$,14)+LEFT$(BD$,2)
4560 NEXT Y
4570 LOCATE 1,1:PRINT CHR$(215);STRING$(16,CHR$(219));CHR$(216)
4580 LOCATE 1,18:PRINT CHR$(217);STRING$(16,CHR$(219));CHR$(218)
4590 FOR Y=2 TO 16 STEP 2
4600 LOCATE Y,0:PRINT CHR$((Y/2)+96)
4610 LOCATE 19,Y:PRINT CHR$(48+9-Y/2)
4620 NEXT Y
4630 LOCATE 21,0:PRINT "XADREZ"
4640 LOCATE 21,1:PRINT "===== "
4650 LOCATE 21,6:PRINT "De :"
4660 LOCATE 21,8:PRINT "Para:"
4670 FOR X=1 TO 7 STEP 2:FOR Y=1 TO 7 STEP 2
4680 C(X,Y)=1:C(X+1,Y)=0
4690 C(X,Y+1)=0:C(X+1,Y+1)=1
4700 NEXT Y,X
4710 RETURN
4720 REM <<< SALVAR >>>
4730 Z$(COU,1)="FF":G$="":H$="":I$=""
4740 LOCATE 0,20:PRINT "      ESPERE 7 M
INUTOS      "
4750 FOR N=1 TO 8:FOR M=1 TO 8
4760 G$=G$+M$(N,M)+X$(N,M)

```

```
4770 NEXT M,N
4780 FOR N=1 TO 100
4790 H$=H$+Z$(N,1):I$=I$+Z$(N,2)
4800 NEXT N
4810 H$=H$+STR$(COU):BEEP
4820 LOCATE 2,20:PRINT "CASSETE OU DISQUETE (C/D)?
4830 SS$=INPUT$(1):IF SS$="C" OR SS$="c"
" THEN 4840 ELSE 4850
4840 OPEN "CAS:XADREZ" FOR OUTPUT AS 1:
GOTO 4880
4850 LOCATE 2,20:PRINT "DRIVE A OU DRIVE B (A/B)?"
4860 DD$=INPUT$(1):IF DD$="A" OR DD$="a"
" THEN OPEN "A:XADREZ" FOR OUTPUT AS 1:
GOTO 4880
4870 IF DD$="B" OR DD$="b" THEN OPEN "B
:XADREZ" FOR OUTPUT AS 1 ELSE GOTO 4820
4880 PRINT#1,G$:PRINT#1,H$:PRINT#1,I$
4890 CLOSE:BEEP:BEEP:LOCATE 0,20:PRINT
SPACE$(30):RETURN
4900 REM <<< CARREGAR >>>
4910 LOCATE 0,20:PRINT " CASSETE OU DISQUETE (C/D)? "
4920 SS$=INPUT$(1):IF SS$="C" OR SS$="c"
" THEN 4930 ELSE 4940
4930 OPEN "CAS:XADREZ" FOR INPUT AS 1:G
OTO 4970
4940 LOCATE 2,20:PRINT "DRIVE A OU DRIVE B (A/B)?"
4950 DD$=INPUT$(1):IF DD$="A" OR DD$="a"
" THEN OPEN "A:XADREZ" FOR INPUT AS 1:
GOTO 4970
4960 IF DD$="B" OR DD$="b" THEN OPEN "B
:XADREZ" FOR INPUT AS 1 ELSE GOTO 4940
4970 IF EOF(1) THEN LOCATE 2,20:PRINT "
NAO HA NENHUM JOGO SALVO ":"CLOSE:FOR
KK=1 TO 200:NEXT KK:LOCATE 2,20:PRINT S
PACE$(27):RETURN
```

```

4980 G$="" : H$="" : I$=""
4990 LINE INPUT#1, G$
5000 LINE INPUT#1, H$
5010 LINE INPUT#1, I$
5020 CLOSE
5030 LOCATE 2,20:PRINT "      ESPERE 7 MI
NUTOS "
5040 FOR N=1 TO 8:FOR M=1 TO 8
5050 M$(N,M)=LEFT$(G$,1):X$(N,M)=MID$(G
$,2,1)
5060 G$=RIGHT$(G$,LEN(G$)-2)
5070 NEXT M,N
5080 FOR N=1 TO 100
5090 Z$(N,1)=LEFT$(H$,2)
5100 H$=RIGHT$(H$,LEN(H$)-2)
5110 Z$(N,2)=LEFT$(I$,2)
5120 I$=RIGHT$(I$,LEN(I$)-2)
5130 NEXT N
5140 COU=VAL(H$)
5150 A$=CHR$(198)
5160 LOCATE 2,20:PRINT SPACE$(30)
5170 RETURN
5180 REM <<< INICIALIZAÇÃO >>>
5190 K$=CHR$(182)+CHR$(183)
5200 K1$=CHR$(192)+CHR$(193)
5210 K2$=CHR$(150)+CHR$(151)
5220 K3$=CHR$(160)+CHR$(161)
5230 K4$=CHR$(166)+CHR$(167)
5240 K5$=CHR$(176)+CHR$(177)
5250 K6$=CHR$(198)+CHR$(199)
5260 K7$=CHR$(208)+CHR$(209)
5270 Q$=CHR$(184)+CHR$(185)
5280 Q1$=CHR$(192)+CHR$(193)
5290 Q2$=CHR$(152)+CHR$(153)
5300 Q3$=CHR$(160)+CHR$(161)
5310 Q4$=CHR$(168)+CHR$(169)
5320 Q5$=CHR$(176)+CHR$(177)
5330 Q6$=CHR$(200)+CHR$(201)
5340 Q7$=CHR$(208)+CHR$(209)

```

```

5350 B$=CHR$(186)+CHR$(187)
5360 B1$=CHR$(192)+CHR$(193)
5370 B2$=CHR$(154)+CHR$(155)
5380 B3$=CHR$(160)+CHR$(161)
5390 B4$=CHR$(170)+CHR$(171)
5400 B5$=CHR$(176)+CHR$(177)
5410 B6$=CHR$(202)+CHR$(203)
5420 B7$=CHR$(208)+CHR$(209)
5430 T$=CHR$(188)+CHR$(189)
5440 T1$=CHR$(192)+CHR$(193)
5450 T2$=CHR$(156)+CHR$(157)
5460 T3$=CHR$(160)+CHR$(161)
5470 T4$=CHR$(172)+CHR$(173)
5480 T5$=CHR$(176)+CHR$(177)
5490 T6$=CHR$(204)+CHR$(205)
5500 T7$=CHR$(208)+CHR$(209)
5510 R$=CHR$(190)+CHR$(191)
5520 R1$=CHR$(192)+CHR$(193)
5530 R2$=CHR$(158)+CHR$(159)
5540 R3$=CHR$(160)+CHR$(161)
5550 R4$=CHR$(174)+CHR$(175)
5560 R5$=CHR$(176)+CHR$(177)
5570 R6$=CHR$(206)+CHR$(207)
5580 R7$=CHR$(208)+CHR$(209)
5590 P$=CHR$(194)+CHR$(195)
5600 P1$=CHR$(196)+CHR$(197)
5610 P2$=CHR$(162)+CHR$(163)
5620 P3$=CHR$(164)+CHR$(165)
5630 P4$=CHR$(178)+CHR$(179)
5640 P5$=CHR$(180)+CHR$(181)
5650 P6$=CHR$(210)+CHR$(211)
5660 P7$=CHR$(212)+CHR$(213)
5670 BD$=STRING$(2,CHR$(214))+SPACE$(2)
5680 BD$=BD$+BD$:BD$=BD$+BD$
5690 FOR N=1 TO 100
5700 Z$(N,1)=SPACE$(2):Z$(N,2)=SPACE$(2)
)
5710 NEXT N
5720 RETURN


```

```
5730 REM <<< DEFINIÇÃO DAS FORMAS >>>
5740 FOR CH=150 TO 181
5750 FOR N=0 TO 7
5760 READ B
5770 IF CH>=166 THEN B=B XOR 255
5780 VPOKE 8*CH+N,B
5790 NEXT N
5800 IF CH=165 THEN RESTORE
5810 NEXT CH:RESTORE 6120
5820 FOR CH=182 TO 213
5830 FOR N=0 TO 7
5840 READ B
5850 IF CH>=198 THEN B=B XOR 255
5860 VPOKE 8*CH+N,B
5870 NEXT N
5880 IF CH=197 THEN RESTORE 6120
5890 NEXT CH
5900 FOR CH=215 TO 220
5910 FOR N=0 TO 7
5920 READ B:VPOKE 8*CH+N,B
5930 NEXT N,CH
5940 FOR A=0 TO 7:VPOKE 8*214+A,255:NEX
T A
5950 RETURN
5960 DATA 1,3,29,63,63,57,28,7
5970 DATA 0,128,56,252,252,156,56,224
5980 DATA 1,19,81,91,109,55,27,15
5990 DATA 128,200,138,218,182,236,216,2
40
6000 DATA 1,1,7,14,28,28,14,7
6010 DATA 128,128,224,240,120,120,240,2
24
6020 DATA 0,0,3,7,15,31,9,3
6030 DATA 0,192,64,224,176,208,232,120
6040 DATA 0,0,25,25,25,31,9,3
6050 DATA 0,0,152,152,152,248,240,240
6060 DATA 7,7,4,31,16,127,64,127
6070 DATA 224,224,224,248,120,254,30,25
4
```

6080 DATA 0,0,0,0,0,3,7,7
6090 DATA 0,0,0,0,0,192,224,224
6100 DATA 5,4,3,15,8,31,16,31
6110 DATA 224,224,192,240,112,248,56,24
8
6120 DATA 253,227,221,191,191,187,222,2
31
6130 DATA 127,131,59,253,253,157,59,231
6140 DATA 237,147,85,91,109,173,219,239
6150 DATA 183,201,170,218,182,239,219,2
47
6160 DATA 253,249,247,238,220,220,238,2
47
6170 DATA 191,159,239,247,251,251,247,2
39
6180 DATA 255,254,249,247,239,223,233,2
43
6190 DATA 63,223,95,239,183,215,235,123
6200 DATA 255,230,217,217,217,223,236,2
39
6210 DATA 255,103,155,155,155,251,247,2
47
6220 DATA 247,247,228,223,144,127,64,12
7
6230 DATA 239,239,231,251,121,254,30,25
4
6240 DATA 255,255,255,255,252,251,247,2
47
6250 DATA 255,255,255,255,63,223,239,23
9
6260 DATA 245,250,243,239,232,239,208,2
23
6270 DATA 239,239,207,247,119,251,59,25
1
6280 DATA 0,0,63,32,47,40,40,40
6290 DATA 0,0,248,8,224,40,40,40
6300 DATA 40,40,47,32,63,0,0,0
6310 DATA 40,40,224,8,248,0,0,0
6320 DATA 0,0,255,0,255,0,0,0

```
6330 DATA 40,40,40,40,40,40,40,40  
6340 DATA R,T,B,Q,K,B,T,R
```

Gustavus



Um Curioso Jogo de Estratégia

Pelo que sei, esta é a primeira adaptação para computador de *Simmons Simples*, um conhecido jogo de lápis e papel.

Simmons Simples ou *Sim* foi inventado por um sujeito chamado Gustavus Simmons. Decidimos chamar esta versão de *Gustavus* porque o nome parece mais pomposo que o do seu primo pobre.

Neste jogo, cada jogador tem que traçar uma reta ligando dois pontos. Perde o primeiro que formar um triângulo de sua própria cor. O segredo é forçar o adversário a formar primeiro o triângulo.

O computador se encarrega de mudar a cor da reta de acordo com o jogador e de indicar o vencedor assim que é formado um triângulo.

Para traçar uma reta, basta movimentar o cursor até a posição desejada, usando o joystick ou as teclas de controle do cursor da direita e da esquerda. Quando o cursor estiver sobre o primeiro ponto da reta a ser

traçada, aperte o botão de tiro ou a barra de espaço. Em seguida, desloque o cursor para o segundo ponto da reta e torne a apertar o botão de tiro ou a barra de espaço. Se a reta não for permitida, o computador se recusará a traçá-la e emitirá um ruído característico. Nesse caso, será preciso entrar de novo com os pontos inicial e final.

O jogo não pode terminar empatado e aparentemente não há nenhuma vantagem em ser o primeiro a jogar.

```

1000 REM <<< GUSTAVUS - ISSI >>>
1010 SCREEN 2,0,0
1020 MAXFILES=2:OPEN "GRP:S" FOR OUTPUT
    AS 1
1030 GOSUB 1290
1040 GOSUB 1190
1050 GOSUB 1370
1060 GOSUB 1800
1070 FOR N=1 TO 1000:NEXT N
1080 IF PLY=1 THEN WIN=2 ELSE WIN=1
1090 PRESET (50,160)
1100 IF WIN=2 THEN CO=7 ELSE CO=9
1110 COLOR CO,0:PRINT #1,"O JOGADOR";WI
N;"VENCEU!"
1120 PRESET (50,175)
1130 PRINT #1,"QUER JOGAR DE NOVO?"
1140 A$=INKEY$:IF A$="S" OR A$="s" THEN
    1040
1150 IF A$="N" OR A$="n" THEN SCREEN 0:
    COLOR 15,4,4:STOP
1160 GOTO 1140
1170 GOTO 1060
1180 REM <<< TELA >>>
1190 KEY OFF
1200 COLOR 11,4,4:CLS
1210 PRESET (100,0):PRINT #1,"GUSTAVUS"
    :PRESET (92,8):PRINT #1,"-----"
1220 FOR N=1 TO 6
1230 PUT SPRITE N,(P(N,1),P(N,2)),3,1
1240 NEXT N
1250 PRESET (8,32):COLOR 9:PRINT #1,"JO
GADOR 1"

```

```

1260 PRESET (182,32):COLOR 7:PRINT #1,"  

JOGADOR 2"  

1270 RETURN  

1280 REM <<< INITIALIZAÇÃO >>>  

1290 DIM P(6,2)  

1300 RESTORE  

1310 FOR N=1 TO 6:READ P(N,1):READ P(N,  

2):NEXT N  

1320 S$="":FOR N=0 TO 7:READ A$:S$=S$+C  

HR$(VAL("&H"+A$)) :NEXT:SPRITE$(1)=S$  

1330 S$="":FOR N=0 TO 7:READ A$:S$=S$+C  

HR$(VAL("&H"+A$)) :NEXT:SPRITE$(2)=S$  

1340 DIM D(6,6)  

1350 RETURN  

1360 REM <<< INICIO DO JOGO >>>  

1370 FOR N=1 TO 6:FOR M=1 TO 6  

1380 D(N,M)=0  

1390 NEXT M:NEXT N  

1400 RETURN  

1410 REM <<< ENTRADA DE DADOS >>>  

1420 LE=0:RI=0:FI=0  

1430 A=STICK(0) OR STICK (1) OR STICK(2  

):FI=STRIG(0) OR STRIG(1) OR STRIG(2):A  

$=INKEY$  

1440 IF A=3 THEN RI=1  

1450 IF A=7 THEN LE=1  

1460 FI=ABS(FI)  

1470 FOR W=1 TO 70:NEXT W  

1480 RETURN  

1490 REM <<< JOGADA >>>  

1500 PNT=1:IF PLY=1 THEN CO=9 ELSE CO=7  

1510 COLOR CO  

1520 PUTSPRITE 7,(P(PNT,1),P(PNT,2)),CO  

,2  

1530 GOSUB 1420  

1540 IF FI=1 THEN GOTO 1600  

1550 IF LE=1 THEN PNT=PNT-1  

1560 IF RI=1 THEN PNT=PNT+1  

1570 IF PNT=7 THEN PNT=1

```

```
1580 IF PNT=0 THEN PNT=6
1590 GOTO 1510
1600 PLAY "L20C"
1610 FOR W=1 TO 300:NEXT W
1620 RETURN
1630 PUT SPRITE 7,(P(DES,1),P(DES,2)),C
0,2
1640 RETURN
1650 REM <<< VEZ >>>
1660 GOSUB 1500
1670 FRO=PNT:PUT SPRITE 8,(P(FRO,1),P(F
RO,2)),13,2
1680 GOSUB 1500
1690 DES=PNT:PUT SPRITE 8,(0,0),13,2
1700 GOSUB 1630
1710 IF FRO=DES OR D(FRO,DES)<>0 THEN G
OTO 1780
1720 LINE (P(FRO,1)+4,P(FRO,2)+4)-(P(DE
S,1)+4,P(DES,2)+4),CO
1730 D(DES,FRO)=PLY:D(FRO,DES)=PLY
1740 FOR N=1 TO 6
1750 IF D(DES,N)=PLY AND D(N,FRO)=PLY T
HEN DD=1:GOSUB 1880
1760 NEXT N
1770 RETURN
1780 PLAY "CDE":GOTO 1660
1790 REM <<< JOGADA COMPLETA >>>
1800 DD=0
1810 PLY=1:GOSUB 1660
1820 IF DD=1 THEN RETURN
1830 PLY=2:GOSUB 1660
1840 IF DD=1 THEN RETURN
1850 GOTO 1800
1860 REM <<< FIM DO JOGO >>>
1870 LINE (P(FRO,1)+4,P(FRO,2)+4)-(P(N,
1)+4,P(N,2)+4),15
1880 LINE (P(N,1)+4,P(N,2)+4)-(P(DES,1)
+4,P(DES,2)+4),15
1890 LINE (P(DES,1)+4,P(DES,2)+4)-(P(FR
O,1)+4,P(FRO,2)+4),15
```

```
1900 PLAY "T140L4CL8EFGFEL4CL8CEFGFE"
1910 FOR W=1 TO 200:NEXT W
1920 RETURN
1930 DATA 127,148,190,117,190,55
1940 DATA 127,24,64,55,64,117
1950 DATA 00,7E,42,5A,5A,42,7E,00
1960 DATA FF,81,BD,A5,A5,BD,81,FF
```

10

Corrida de Dados



Foi Dada a Partida!

Este é o primeiro de uma série de três jogos de dados. Todos usam a mesma rotina central; os jogos são escolhidos através de um menu.

Corrida de Dados é um jogo de corrida no qual o resultado dos dados determina a distância percorrida pelos cavalos. Pode ser jogado por duas a quatro pessoas. O botão de tiro (ou barra de espaço) deve ser apertado duas vezes, uma para sacudir os dados e outra para jogá-los na mesa. Isso quer dizer que o jogador pode sacudir os dados durante o tempo que quiser.

Os cavalos disputam para ver qual conseguirá chegar primeiro ao lado direito da tela. Entretanto, para que o último jogador não seja prejudicado, o resultado só é anunciado depois que termina uma rodada.

O jogo pode ser interrompido a qualquer momento, bastando para

isso entrar com **D** (de "Desisto"). Nesse caso, o programa volta ao menu.
Nota: Este jogo e os dois que se seguem usam uma rotina chamada *Dados*, que deve ser carregada primeiro e salva em fita. Os jogos podem ser então acrescentados um por um. Não será possível escolher um jogo a partir do menu até que todos os três tenham sido carregados.

```

1000 REM <<< DADOS - ANDY/JIM >>>
1010 CLEAR 500
1020 ON INTERVAL=10 GOSUB 1160:GOSUB 14
70
1030 SCREEN 1,3:WIDTH 30:KEY OFF
1040 GOSUB 1190:COLOR 15,0,1
1050 GOSUB 3220
1060 LOCATE 4,3:PRINT "1> Corrida de Da
dos"
1070 LOCATE 4,6:PRINT "2> Olhos de Cobr
a"
1080 LOCATE 4,9:PRINT "3> Craps"
1090 INTERVAL ON
1100 A=VAL(INKEY$):IF A<1 OR A>3 THEN 1
100
1110 PLAY "o4a64"
1120 ON A GOTO 1710,2250,2670
1130 LOCATE 0,7:PRINT "Aperte qualquer
tecla para ver o menu ou 'P' para par
ar"
1140 A$=INKEY$:IF A$="" THEN 1140 ELSE
IF A$="P" OR A$="p" THEN 2070
1150 GOSUB 1500:GOTO 1060
1160 INTERVAL OFF:A$=INKEY$
1170 IF A$="d" OR A$="D" THEN SOUND 8,0
:GOTO 1150 ELSE INTERVAL ON:RETURN
1180 REM <<< TELA >>>
1190 COLOR 14:FOR X=0 TO 31:LOCATE X,11
:PRINT "[";LOCATE X,20:PRINT "]":NEXT X
1200 FOR Y=12 TO 19:LOCATE 0,Y:PRINT "["
:LOCATE 31,Y:PRINT "]";:NEXT Y
1210 RETURN
1220 REM <<< SORTEIO >>>
1230 SOUND 0,0:SOUND 1,3:SOUND 8,10

```

```


1240 PUT SPRITE 1,(25,90),15,11
1250 PUT SPRITE 2,(57,90),15,12
1260 PUT SPRITE 3,(25,122),15,13
1270 PUT SPRITE 4,(57,122),15,14
1280 R%=RND(-TIME)*6
1290 D1=INT(RND(R%)*6)+1
1300 R%=RND(-TIME)*6
1310 D2=INT(RND(R%)*6)+1
1320 SOUND 8,0
1330 A$=INKEY$: IF A$=CHR$(32) THEN 1390
1340 PUT SPRITE 1,(35,90),15,11
1350 PUT SPRITE 2,(67,90),15,12
1360 PUT SPRITE 3,(35,122),15,13
1370 PUT SPRITE 4,(67,122),15,14
1380 A$=INKEY$: IF A$<>CHR$(32) THEN 123
0
1390 PUT SPRITE 1,(76,95),15,7
1400 PUT SPRITE 2,(108,95),15,8
1410 PUT SPRITE 3,(76,127),15,9
1420 PUT SPRITE 4,(108,127),15,10
1430 PUT SPRITE 5,(12,127),15,D1
1440 PUT SPRITE 6,(44,127),15,D2
1450 RETURN
1460 REM <<< INICIALIZAÇÃO >>>
1470 DIM S$(14),P(4,3),O(5)
1480 RETURN
1490 REM <<< APAGAR O MENU >>>
1500 FOR Y=0 TO 10:LOCATE 0,Y
1510 PRINT STRING$(30," "):NEXT Y
1520 PLAY "o3a64":RETURN
1530 REM <<< JOGADORES >>>
1540 GOSUB 1500
1550 LOCATE 0,0:PRINT "NO. DE JOGADORES
(2-4)"
1560 A$=INKEY$:PL=VAL(A$)
1570 IF PL<2 OR PL>4 THEN 1560
1580 PLAY "o5g64"
1590 GOSUB 1500
1600 FOR N=1 TO 4

```

```
1610 P(N,1)=0
1620 NEXT N
1630 FOR N=1 TO PL
1640 O(N)=1
1650 NEXT N
1660 FOR Q=N TO 4
1670 O(Q)=0
1680 NEXT Q
1690 RETURN
1700 REM <<< CORRIDA DE DADOS >>>
1710 COLOR , ,13:GOSUB 1540
1720 FOR N=1 TO 4
1730 P(N,1)=0:P(N,2)=2*N-1
1740 NEXT N
1750 RESTORE 1810
1760 FOR N=0 TO 8
1770 READ A
1780 LOCATE 31,N
1790 PRINT CHR$(A)
1800 NEXT N
1810 DATA 188,67,72,69,71,65,68,65,188
1820 GOSUB 2200
1830 FOR W=1 TO PL
1840 LOCATE 18,13:PRINT "JOGADOR";
1850 LOCATE 20,15:PRINT W
1860 PLAY "b3f64"
1870 A$=INKEY$: IF A$="" THEN 1870
1880 GOSUB 2080
1890 GOSUB 1230
1900 FOR D=1 TO (D1+D2)*2
1910 GOSUB 2160
1920 P(W,1)=P(W,1)+1
1930 GOSUB 2180
1940 NEXT D
1950 NEXT W
1960 WI=0
1970 FOR N=1 TO 4
1980 IF WI>0 THEN 2000
1990 IF P(N,1)>220 THEN WI=N
```

```
2000 NEXT N
2010 IF WI=0 THEN 1830
2020 GOSUB 1500
2030 GOSUB 2080
2040 LOCATE 1,0:IF WI<>0 THEN PRINT "O
jogador";WI;"ganhou!" ELSE PRINT "Ningu
ém ganhou!"
2050 REM <<<< FIM DO JOGO >>>>
2060 GOTO 1130
2070 SCREEN 0:COLOR 15,4,4:END
2080 PUT SPRITE 5,(100,100),1,0
2090 PUT SPRITE 6,(100,100),1,0
2100 RETURN
2110 END
2120 FOR N=1 TO 4
2130 LOCATE INT(P(N,1)/8),P(N,2)
2140 PRINT " ":NEXT N
2150 RETURN
2160 LOCATE INT(P(W,1)/8),P(W,2)
2170 PRINT " ":RETURN
2180 LOCATE INT(P(W,1)/8),P(W,2)
2190 PRINT CHR$(201):RETURN
2200 FOR N=1 TO 4
2210 LOCATE INT(P(N,1)/8),P(N,2)
2220 PRINT CHR$(201):NEXT N
2230 RETURN
```

Olhos de Cobra



Façam Suas Apostas!

Este é o segundo dos três jogos de dados. *Corrida de Dados* deve ser carregado em primeiro lugar; em seguida, carregue *Olhos de Cobra*. As instruções para jogar os dados são as mesmas.

O jogo é para quatro jogadores e as apostas aparecem na tela. No início, é estabelecida uma meta em cruzados, que pode variar entre Cz\$ 40,00 e Cz\$ 400,00. A banca começa com Cz\$ 5,00 se a meta é menor que Cz\$ 200,00 e Cz\$ 10,00 se a meta é igual ou maior que Cz\$ 200,00. Os jogadores então jogam os dados, um de cada vez. Se um jogador tira uma dobradinha, recebe o dinheiro da banca. Se um jogador tira “olhos de cobra” (duplo um), recebe o dobro do dinheiro da banca. Cada vez que a banca paga a um jogador, seu dinheiro volta ao valor inicial.

Depois de cada rodada, a banca recebe uma quantia igual ao valor

inicial. O jogo continua até que um jogador consiga atingir a meta. Para interromper o jogo antes do final e voltar ao menu, entre com **D**.

```


2240 REM <<< OLHOS DE COBRA >>>
2250 COLOR,,6:GOSUB 1540
2260 LOCATE 0,22:PRINT "Qual é a meta "
;
2270 INPUT LI$:LI=VAL(LI$)
2280 LOCATE 0,22:PRINT STRING$(28," ")
2290 IF LI>400 OR LI<40 THEN 2260
2300 UA=5:GOSUB 1500:IF LI>200 THEN UA=
10
2310 LOCATE 5,0:PRINT "1"
2320 LOCATE 11,0:PRINT "2"
2330 LOCATE 17,0:PRINT "3"
2340 LOCATE 23,0:PRINT "4"
2350 LOCATE 0,1:PRINT STRING$(30,"-");
2360 LOCATE 0,10:PRINT STRING$(30,"-");
2370 LOCATE 0,7:PRINT STRING$(30,"-");
2380 KI=0
2390 KI=KI+UA
2400 PE=1
2410 FOR N=1 TO 4:LOCATE N*6-2,3
2420 PRINT P(N,1):NEXT N
2430 LOCATE 18,13:PRINT "Jogador"
2440 LOCATE 20,15:PRINT PE
2450 GOSUB 2480:IF P(PE,1)>LI THEN 2640
2460 PE=PE+1:IF PE>PL THEN 2390
2470 GOTO 2410
2480 A$=INKEY$
2490 IF A$="" THEN 2480
2500 LOCATE 0,8:PRINT "A banca tem";KI
2510 GOSUB 2080
2520 GOSUB 1230
2530 IF D1=1 AND D2=1 THEN GOSUB 2560
2540 IF D1=D2 AND D1<>1 THEN GOSUB 2600
2550 RETURN
2560 PLAY "o4a64b64c64"
2570 P(PE,1)=P(PE,1)+KI*2
2580 KI=UA

```

```
2590 RETURN
2600 PLAY "o4g64e64d64"
2610 P(PE,1)=P(PE,1)+KI
2620 KI=UA
2630 RETURN
2640 WI=PE
2650 GOTO 2020
```

12

Craps



Igualzinho a Las Vegas!

Aqui está a versão para o seu microcomputador do jogo de dados que os cassinos de Las Vegas tornaram popular.

Dois a quatro jogadores agora podem fazer de conta que estão em Las Vegas, por cortesia do MSX.

No início, cada jogador dispõe de Cz\$ 100,00 (você pode mudar para dólares, para tornar o jogo mais emocionante!). Na sua vez, o jogador pode apostar qualquer quantia até o máximo de que dispõe na ocasião. Em seguida, está na hora de jogar os dados.

Se da primeira vez o jogador obtém 7 ou 11, recebe uma quantia igual à que apostou e passa a vez. Se obtém 2, 3 ou 12, perde tudo que apostou e passa a vez. Se obtém qualquer outro número, este número passa a ser o *ponto*, isto é, o valor que ele vai tentar obter novamente. O jogador continua a lançar os dados até obter o ponto, caso em que recebe uma

quantia igual à que apostou, ou 7, caso em que perde a aposta. Nos dois casos, o jogador passa a vez.

Se um jogador perde todo o dinheiro que tem, é imediatamente excluído do jogo. O jogo prossegue até que reste apenas um jogador.

Para interromper o jogo antes do final e voltar ao menu, entre com D.

Antes de carregar este jogo, carregue *Corrida de Dados e Olhos de Cobra*.

```

2660 REM <<< CRAPS >>>
2670 COLOR , , 4:GOSUB 1540
2680 LOCATE 5,0:PRINT "1"
2690 LOCATE 11,0:PRINT "2"
2700 LOCATE 17,0:PRINT "3"
2710 LOCATE 23,0:PRINT "4"
2720 LOCATE 0,1:PRINT STRING$(30,"-");
2730 LOCATE 0,10:PRINT STRING$(30,"-");
2740 LOCATE 0,7:PRINT STRING$(30,"-");
2750 FOR N=1 TO 4
2760 P(N,1)=100
2770 NEXT N
2780 FOR W=1 TO PL
2790 FOR N=1 TO 4
2800 LOCATE N*6,3:PRINT " " :LOCATE N*6
-2,3:PRINT P(N,1)
2810 NEXT N
2820 IF P(W,1)<1 THEN 2990
2830 LOCATE 0,22:PRINT "Jogador";W;"apo
sta";
2840 INPUT B$
2850 LOCATE 0,22:PRINT STRING$(23," ")
2860 BE=VAL(B$):IF BE<1 OR BE>P(W,1) TH
EN PLAY "o4a64":GOTO 2830
2870 GOSUB 2080:GOSUB 1230
2880 T=D1+D2
2890 IF T=7 OR T=11 THEN 3130
2900 IF T=2 OR T=3 OR T=12 THEN 3170
2910 P0=T
2920 LOCATE 0,8:PRINT "Jogue de novo .
"
2930 A$=INKEY$:IF A$="" THEN 2930

```

```

2940 GOSUB 2080:GOSUB 1230
2950 T=D1+D2
2960 IF T=PO THEN 3130
2970 IF T=7 THEN 3170
2980 GOTO 2920
2990 NEXT W
3000 FL=0:CO=0
3010 FOR N=1 TO PL
3020 IF CO=1 THEN 3050
3030 IF P(N,1)>0 AND FL=1 THEN CO=1
3040 IF P(N,1)>0 THEN FL=1
3050 NEXT N
3060 IF CO=1 THEN 2780
3070 WI=0
3080 FOR N=1 TO PL
3090 IF WI<>0 THEN 3110
3100 IF P(N,1)>0 THEN WI=N
3110 NEXT N
3120 GOTO 2020
3130 PLAY "o4g64f64e64"
3140 LOCATE 0,B:PRINT "        Você ganh
ou!"
3150 P(W,1)=P(W,1)+BE
3160 GOTO 2990
3170 PLAY "o4e64g64"
3180 LOCATE 0,B:PRINT "        Você perde
u!"
3190 P(W,1)=P(W,1)-BE
3200 GOTO 2990
3210 REM <<< DEFINIR SPRITES/CARACTERE
S >>>
3220 RESTORE 3330
3230 FOR S=1 TO 14:S$(S)=""
3240 FOR DD=1 TO 32
3250 READ DA:S$(S)=S$(S)+CHR$(DA)
3260 NEXT DD:SPRITE$(S)=S$(S)
3270 NEXT S
3280 FOR CH=200 TO 202
3290 FOR DD=0 TO 7


```



```
3490 DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,1,1
,1,0,0,0,0,15,16,32,32,64,64,128,128,12
8,0,0
3500 DATA 0,0,0,0,240,8,4,4,2,2,2,2,1,1
,1,1,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,12
8
3510 DATA 1,2,2,2,2,2,4,4,4,4,4,4,2,1,0,0
,0,0,0,0,0,0,0,0,0,0,128,112,15,0
3520 DATA 0,0,0,0,0,0,0,0,0,0,0,0,0,0,1,14,
240,0,128,128,64,64,64,64,32,32,32,3
2,64,128,0,0,0
3530 DATA 0,124,198,198,198,108,238,0
3540 DATA 15,7,13,120,204,204,204,120
3550 DATA 24,24,24,24,24,24,24,24,24
```

13

Vinte-e-Um



Um 21 Real Pode Quebrar a Banca

Nesta versão para o MSX do conhecido jogo de cartas, a banca começa com Cz\$ 500,00 e cada jogador tem Cz\$ 100,00. Um ou dois jogadores jogam contra a banca, tentando chegar o mais perto possível de vinte e um pontos.

Inicialmente, cada jogador recebe uma carta aberta e a banca recebe uma carta fechada. Os jogadores podem então apostar qualquer quantia até o máximo de que dispõem na ocasião. Uma segunda carta é distribuída. Ao chegar a sua vez, cada jogador pode fazer as seguintes opções:

Segurar. Se o total das cartas é 16 ou mais e o jogador aperta a tecla S, ele passa a vez.

Pedir. O jogador aperta a tecla P para pedir mais uma carta.

Dobrar. Se o jogador acha que está com um jogo muito bom, aperta a tecla **D** para dobrar a aposta. Nesse caso, sempre recebe mais uma carta.

No vinte-e-um, o ás pode valer 1 ou 11, de acordo com o que for mais conveniente para o jogador. As figuras (valete, dama e rei) valem 10 pontos. As cartas restantes têm um valor igual ao dos seus números (o 10 vale 10, o 9 vale 9 etc.). Se a soma dos pontos de um jogador ultrapassa 21, ele perde a aposta e passa a vez. Depois que todos os jogadores terminam, a banca joga sua mão e faz os pagamentos. É preciso não esquecer que os jogadores jogam contra a banca e não um contra o outro.

O resultado de cada mão é calculado da seguinte forma: o jogo mais alto é o *21 real* (21 pontos com duas cartas, ás e dama, por exemplo). Em seguida vem a *vaza de cinco cartas*, qualquer conjunto de cinco cartas cujo conjunto não excede 21. O terceiro maior jogo é o *vinte-e-um* com qualquer número de cartas. Se nenhum desses jogos estiver presente, prevalece a mão com maior número de pontos. Em caso de empate, ganha a banca.

O jogo continua até todos os jogadores irem à falência ou a banca quebrar. Para recomeçar o jogo, basta apertar a tecla **R**.

```

1000 REM <<< VINTE-E-UM - ANDY/JIM >>>
>
1010 SCREEN 0:COLOR 15,4:KEY OFF
1020 DEFINT A-Z
1030 DIM PA(52),AV(52),PAC$(5,13),CC(3)
1040 DIM CX(4),CY(4)
1050 DIM NOC(4),P(2)
1060 DIM PLC(3,5),BC(5)
1070 DIM MO(2),P$(2)
1080 GOSUB 3950
1090 GOSUB 3060
1100 MO(1)=100:MO(2)=100:MC=500
1110 CC(1)=20:CC(2)=33:CC(3)=6
1120 SC(1)=0:SC(2)=0
1130 GOSUB 2520
1140 GOSUB 2640
1150 CLS:IF NP=1 THEN P(2)=0
1160 IF P(1)=0 AND P(2)=0 THEN 4640
1170 IF P(1)=0 THEN BE(1)=0
1180 IF P(2)=0 THEN BE(2)=0
1190 GOSUB 3140

```

```

1200 CX(1)=15:CX(2)=28
1210 CY(1)=2:CY(2)=2
1220 BX=2:BY=2
1230 GOSUB 3340
1240 FOR Q=1 TO NP
1250 IF P(Q)=0 THEN 1280
1260 GOSUB 3410
1270 GOSUB 4380
1280 NEXT Q
1290 CX=BX:CY=BY:FOR Q=1 TO NC
1300 GOSUB 2980:CY=CY+2:CX=CX+1:NEXT Q
1310 GOSUB 3740
1320 FOR Q=1 TO NP:IF P(Q)=0 THEN 1380
1330 CY=CY(Q):CX=CX(Q)
1340 CY=CY+2:CX=CX+1
1350 GOSUB 3880
1360 GOSUB 4380
1370 CX(Q)=CX
1380 NEXT Q
1390 CX=BX+1:CY=BY+2:GOSUB 2940
1400 GOSUB 2980
1410 GOSUB 3490
1420 REM <<< CONTAGEM DE PONTOS >>>
1430 Q=3
1440 IF NP=2 THEN 1450 ELSE IF SC(1)>21
  THEN P$(1)="L":GOTO 1610
1450 IF SC(1)>21 AND SC(2)>21 THEN P$(1)
  ="L":P$(2)="L":GOTO 1610
1460 FOR W=1 TO NC
1470 CX=BX:CY=(BY-2)+W*2
1480 PLC(3,W)=BC(W)
1490 GOSUB 3270:GOSUB 2760
1500 BX=BX+1
1510 NEXT W
1520 BX=CX:BY=CY
1530 FOR W=1 TO NC
1540 PLC(3,W)=BC(W):NEXT W
1550 Q=3:NOC(3)=NC:CX(3)=BX:GOSUB 4380
1560 IF SC(3)<16 THEN GOSUB 2940 ELSE G

```

```

OTO 1610
1570 BX=BX+1:BY=BY+2:PLC(3,NC)=BC(NC):Q
=3:CX=BX:CY=BY:W=NC:GOSUB 3270:GOSUB 27
60
1580 NOC(3)=NC:GOSUB 4380
1590 IF NC=5 THEN 1610 ELSE GOTO 1560
1600 F=(RND(1)*2)+17:IF SC(3)>F THEN GO
SUB 2940:GOTO 1570
1610 Q=1
1620 IF P(Q)=0 THEN 1770
1630 IF SC(Q)>21 THEN P$(Q)="L":GOTO 17
70
1640 IF SC(3)>21 THEN P$(Q)="W":GOTO 17
70
1650 IF SC(3)=21 AND NC=2 THEN P$(Q)="L
":GOTO 1770
1660 IF SC(Q)=21 AND NOC(Q)=2 THEN P$(Q)
)="W":GOTO 1770
1670 IF NC=5 AND NOC(Q)=5 THEN P$(Q)="L
":GOTO 1770
1680 IF NOC(Q)=5 THEN P$(Q)="W":GOTO 17
70
1690 IF NC=5 THEN P$(Q)="L":GOTO 1770
1700 IF SC(3)=21 AND NC=2 AND SC(Q)=21
AND NOC(Q)=2 THEN P$(Q)="L":GOTO 1770
1710 IF NOC(Q)=5 AND NC<5 THEN P$(Q)="W
":GOTO 1770
1720 IF SC(Q)=21 AND NOC(Q)=2 THEN P$(Q)
)=""":GOTO 1770
1730 IF SC(Q)>SC(3) THEN P$(Q)="W":GOTO
1770
1740 IF SC(Q)=SC(3) THEN P$(Q)="L":GOTO
1770
1750 IF SC(3)>SC(Q) THEN P$(Q)="L"
1760 IF SC(Q)>SC(3) THEN P$(Q)="W"
1770 IF NP=2 AND Q=1 THEN Q=2:GOTO 1620
1780 IF NP=1 THEN 1830
1790 IF P$(1)="L" AND P$(2)="L" THEN 21
QQ

```

```

1800 IF P$(1)="W" AND P$(2)="W" THEN 20
20
1810 IF P$(1)="W" AND P$(2)="L" THEN 22
10
1820 IF P$(1)="L" AND P$(2)="W" THEN 23
10
1830 IF P$(1)="W" THEN 1870
1840 IF P$(1)="L" THEN 1950
1850 END
1860 REM <<< GANHA O JOGADOR >>>
1870 MC=MC-BE(1)
1880 MO(1)=MO(1)+BE(1)*2
1890 IF MC<1 THEN 2410
1900 GOSUB 4500
1910 PLAY "01L64ERC"
1920 LOCATE 1,20:PRINT "O jogador ganhou":GOSUB 4510
1930 GOTO 1140
1940 REM <<< GANHA A BANCA >>>
1950 MC=MC+BE(1)
1960 GOSUB 4500
1970 PLAY "04L64CEG"
1980 LOCATE 1,20:PRINT "Ganhou a banca":GOSUB 4510
1990 IF MO(1)<1 THEN P(1)=0
2000 GOTO 1140
2010 REM <<< GANHAM OS DOIS JOGADORES >>>
2020 MO(1)=MO(1)+BE(1)*2
2030 MO(2)=MO(2)+BE(2)*2
2040 MC=MC-BE(1)-BE(2)
2050 IF MC<1 THEN 2410
2060 GOSUB 4500
2070 LOCATE 1,20:PRINT "Os dois jogadores ganharam"
2080 GOTO 1140
2090 REM <<< PERDEM OS DOIS JOGADORES >>>
2100 MC=MC+BE(1)+BE(2)

```

```
2110 GOSUB 4500
2120 IF MO(1)<1 THEN P(1)=0
2130 IF MO(2)<1 THEN P(2)=0
2140 IF P(1)=0 AND P(2)=1 THEN LOCATE 1
,20:PRINT "O jogador 2 perdeu":PLAY "01
L8C"
2150 IF P(1)=1 AND P(2)=0 THEN LOCATE 1
,20:PRINT "O jogador 1 perdeu":PLAY "01
LBC"
2160 IF MO(1)<1 THEN P(1)=0
2170 IF P(1)=1 AND P(2)=1 THEN LOCATE 1
,20:PRINT "Os dois jogadores perderam":PLAY
"01L8EDC"
2180 GOSUB 4510
2190 GOTO 1140
2200 REM <<< GANHA O JOGADOR 1 >>>
2210 MC=MC+BE(2)
2220 MO(1)=MO(1)+BE(1)*2
2230 MC=MC-BE(1)
2240 IF MO(2)<1 THEN P(2)=0
2250 IF MC<1 THEN 2410
2260 GOSUB 4500
2270 PLAY "04L64CDEFGABAGFEDC"
2280 LOCATE 1,20:PRINT "O jogador 1 gan
hou":GOSUB 4510
2290 GOTO 1140
2300 REM <<< GANHA O JOGADOR 2 >>>
2310 MC=MC+BE(1)
2320 MO(2)=MO(2)+BE(2)*2
2330 MC=MC-BE(2)
2340 IF MO(1)<1 THEN P(1)=0
2350 IF MC<1 THEN 2410
2360 GOSUB 4500
2370 PLAY "04L64CDEFGABÁGFEDC"
2380 LOCATE 1,20:PRINT "O jogador 2 gan
hou":GOSUB 4510
2390 GOTO 1140
2400 REM <<< A BANCA QUEBRA >>>
2410 GOSUB 4500
```

```
2420 LOCATE 1,20
2430 PLAY "04L64CDEFGABAGFEDCCDEFGAB"
2440 PRINT "Parabéns! Você quebrou a ba-
nca!"
2450 GOSUB 4510:GOSUB 4510
2460 CLS
2470 PRINT "Jogador 1:";MO(1)
2480 IF NP=2 THEN PRINT "Jogador 2:";MO
(2)
2490 GOSUB 4510:GOSUB 4510:GOSUB 4510
2500 GOTO 1090
2510 REM <<< GERAR BARALHO >>>
2520 FOR SUIT=1 TO 4
2530 FOR CAR=1 TO 13
2540 PAC$(SUIT,CAR)=CHR$(SUIT+217)
2550 IF CAR<11 AND CAR>1 THEN PAC$(SUIT
,CAR)=PAC$(SUIT,CAR)+CHR$(CAR+220)
2560 IF CAR=1 THEN PAC$(SUIT,CAR)=PAC$(SUIT
,CAR)+CHR$(231)
2570 IF CAR=11 THEN PAC$(SUIT,CAR)=PAC$(SUIT
,CAR)+CHR$(232)
2580 IF CAR=12 THEN PAC$(SUIT,CAR)=PAC$(SUIT
,CAR)+CHR$(233)
2590 IF CAR=13 THEN PAC$(SUIT,CAR)=PAC$(SUIT
,CAR)+CHR$(234)
2600 NEXT CAR
2610 NEXT SUIT
2620 RETURN
2630 REM <<< EMBARALHAR >>>
2640 FOR N=1 TO 52
2650 PA(N)=N:AV(N)=1
2660 NEXT N
2670 FOR A=1 TO 30
2680 N1=INT(RND(1)*51)+1
2690 N2=INT(RND(1)*51)+1
2700 CC=PA(N1)
2710 PA(N1)=PA(N2)
2720 PA(N2)=CC
2730 NEXT A
```

```
2740 RETURN
2750 REM <<<< DESENHAR UMA CARTA >>>>
2760 LOCATE CX-1,CY
2770 PRINT CHR$(200);:FOR Z=1 TO 5:PRINT
T CHR$(204);:NEXT Z:PRINT CHR$(203)
2780 LOCATE CX-1,CY+1
2790 PRINT CHR$(206);RIGHT$(PAC$(SUIT,C
AR),1);
2800 PRINT LEFT$(PAC$(SUIT,CAR),1);:PRI
NT STRING$(3,CHR$(208));CHR$(207)
2810 LOCATE CX-1,CY+2
2820 PRINT CHR$(206);STRING$(5,CHR$(208
));CHR$(207)
2830 LOCATE CX-1,CY+3
2840 PRINT CHR$(206);CHR$(208);CHR$(208
);
2850 PRINT LEFT$(PAC$(SUIT,CAR),1);CHR$(
208);CHR$(208);CHR$(207)
2860 LOCATE CX-1,CY+4
2870 PRINT CHR$(206);STRING$(5,CHR$(208
));CHR$(207)
2880 LOCATE CX-1,CY+5
2890 PRINT CHR$(206);CHR$(208);CHR$(208
);CHR$(208);CHR$(208);RIGHT$(PAC$(SUIT,
CAR),1);CHR$(207)
2900 LOCATE CX-1,CY+6
2910 PRINT CHR$(201);STRING$(5,CHR$(205
));CHR$(202)
2920 RETURN
2930 REM <<<< ESCOLHER CARTA DO COMPUTA
DOR >>>>
2940 C=(RND(1)*51)+1:IF AV(C)=0 THEN 29
40
2950 NC=NC+1:BC(NC)=C
2960 RETURN
2970 REM <<<< DESENHAR AS COSTAS DE UMA
CARTA >>>>
2980 LOCATE CX-1,CY
2990 PRINT CHR$(209);STRING$(5,CHR$(213
));CHR$(210)
```

```

3000 FOR NN=1 TO 5:LOCATE CX-1,CY+NN
3010 PRINT CHR$(215);STRING$(5,CHR$(217)
  );CHR$(216)
3020 NEXT NN
3030 LOCATE CX-1,CY+6:PRINT CHR$(211);S
  TRING$(5,CHR$(214));CHR$(212)
3040 RETURN
3050 REM <<< INICIO DO JOGO >>>
3060 CLS
3070 LOCATE 13,1:PRINT "VINTE-E-UM"
3080 PRINT:PRINT "1 ou 2 jogadores?"
3090 NP=VAL(INPUT$(1)):IF NP<1 OR NP>2
  THEN 3090
3100 BEEP
3110 FOR Q=1 TO NP:P(Q)=1:NEXT Q
3120 RETURN
3130 REM <<< DAR AS CARTAS >>>
3140 FOR Q=1 TO NP
3150 C1=(RND(-TIME)*51)+1
3160 IF AV(C1)=0 THEN 3150
3170 AV(C1)=0
3180 PLC(Q,1)=C1
3190 NEXT Q
3200 C1=(RND(-TIME)*51)+1
3210 IF AV(C1)=0 THEN 3200
3220 AV(C1)=0
3230 BC(1)=C1
3240 NOC(1)=1:NOC(2)=1:NC=1
3250 RETURN
3260 REM <<< INTERPRETAR UMA CARTA >>>
  >
3270 S=PLC(Q,W)/13:CAR=PLC(Q,W) MOD 13:
  IF CAR=0 THEN CAR=13
3280 IF S>=0 THEN SUIT=1
3290 IF S>=1 THEN SUIT=2
3300 IF S>=2 THEN SUIT=3
3310 IF S>=3 THEN SUIT=4
3320 RETURN
3330 REM <<< DESENHAR A TELA >>>

```

```

3340 LOCATE 1,0:PRINT "BANCA", "JOGADOR
1", :IF NP=2 THEN LOCATE 27,0:PRINT "JOG
ADOR 2"
3350 LOCATE 1,19:PRINT "$":IF P(1)<>0 T
HEN LOCATE 14,19:PRINT "$"
3360 IF NP=2 AND P(2)<>0 THEN LOCATE 27
,19:PRINT "$"
3370 LOCATE 1,18:PRINT "RESERVAS":LOCAT
E 1,17:PRINT "Total":IF P(1)<> 0 THEN
LOCATE 14,17:PRINT "Total:"
3380 IF NP=2 AND P(2)<>0 THEN LOCATE 27
,17:PRINT "Total:"
3390 RETURN
3400 REM <<<< DESENHAR UMA MAO >>>>
3410 CX=CX(Q):CY=CY(Q)
3420 FOR W=1 TO NOC(Q)
3430 GOSUB 3270
3440 GOSUB 2760
3450 CY=CY+2:CX=CX+1:NEXT W
3460 LOCATE CC(Q)-5,19:PRINT MO(Q)
3470 RETURN
3480 REM <<<< SEGURAR >>>>
3490 FOR Q=1 TO NP:IF P(Q)=0 THEN 3650
3500 CX(Q)=CX(Q)+1
3510 LOCATE 1,20:PRINT STRING$(39," ")
3520 LOCATE 1,20:PRINT "Jogador";Q
3530 LOCATE 1,21:PRINT "Segurar:Pedir:D
obrar:Recomeçar"
3540 A$=INKEY$:IF A$="" THEN GOTO 3540
3550 A$=CHR$(ASC(A$) AND 223)
3560 IF A$="S" OR A$="P" OR A$="D" OR A
$="R" THEN 3570 ELSE 3540
3570 CX=CX(Q):CY=CY(Q)+(2*NOC(Q))
3580 IF A$="D" THEN 4530
3590 IF A$="R" THEN 1090
3600 IF A$="S" AND SC(Q)>15 THEN GOSUB
3670:CX=CX+1:GOTO 3650
3610 IF A$="P" THEN GOSUB 3690:CX=CX+1:
CX(Q)=CX(Q)+1

```

```

3620 GOSUB 4380
3630 IF SC(Q)>21 THEN LOCATE CC(Q)-6,20
:PRINT "ESTOUROU":GOSUB 4510:GOTO 3650
3640 IF NOC(Q)<>5 THEN 3540
3650 NEXT Q
3660 LOCATE 1,20:PRINT STRING$(39," "):
RETURN
3670 GOSUB 4380:LOCATE CC(Q),17:PRINT "
":LOCATE CC(Q),17:PRINT SC(Q):RETURN
3680 REM <<< PEDIR >>>
3690 C=(RND(1)*51)+1:IF AV(C)=0 THEN 36
90
3700 AV(C)=0:NOC(Q)=NOC(Q)+1
3710 PLC(Q,NOC(Q))=C:W=NOC(Q):GOSUB 327
0:GOSUB 2760
3720 RETURN
3730 REM <<< APOSTAR >>>
3740 LOCATE 2,19:PRINT MC:FOR Q=1 TO NP
3750 IF P(Q)=0 THEN 3850
3760 LOCATE 1,20:PRINT STRING$(39," ")
3770 LOCATE 1,20:PRINT "Jogador ";Q;" S
ua aposta";
3780 LOCATE CC(Q)-6,18:PRINT "Aposta";
3790 INPUT BE$:BE(Q)=VAL(BE$)
3800 IF BE(Q)<1 THEN BE(Q)=MO(Q)+1
3810 IF BE(Q)>MO(Q) THEN LOCATE CC(Q)+2
,18:PRINT "      ":GOTO 3760
3820 LOCATE CC(Q),19:PRINT "      "
3830 MO(Q)=MO(Q)-BE(Q)
3840 LOCATE CC(Q)-4,19:PRINT "      ":LOCA
TE CC(Q)-5,19:PRINT MO(Q)
3850 NEXT Q
3860 RETURN
3870 REM <<< DAR OUTRA CARTA >>>
3880 C=(RND(1)*51)+1
3890 IF AV(C)=0 THEN 3880
3900 PLC(Q,NOC(Q)+1)=C:AV(C)=0
3910 NOC(Q)=NOC(Q)+1
3920 W=NOC(Q):GOSUB 3270:GOSUB 2760

```

```
3930 RETURN
3940 REM <<< CARACTERES >>>
3950 RESTORE 4020
3960 FOR C=200 TO 234
3970 FOR B=0 TO 7
3980 READ A:VPOKE 2048+C*B+B,A
3990 NEXT B
4000 NEXT C
4010 RETURN
4020 DATA 0,60,124,124,124,124,124,124,124,124
4030 DATA 124,124,124,124,124,124,124,60,0
4040 DATA 248,248,248,248,248,248,248,248,240,0
4050 DATA 0,240,248,248,248,248,248,248,248,248
4060 DATA 0,252,252,252,252,252,252,252,252,252
4070 DATA 252,252,252,252,252,252,252,252,252,0
4080 DATA 124,124,124,124,124,124,124,124,124,1
24
4090 DATA 248,248,248,248,248,248,248,248,248,2
48
4100 DATA 252,252,252,252,252,252,252,252,252,2
52
4110 DATA 0,60,84,104,84,104,84,104,84,104
4120 DATA 0,240,88,168,88,168,88,168,88,168
4130 DATA 84,104,84,104,84,104,84,104,60,0
4140 DATA 88,168,88,168,88,168,240,0
4150 DATA 0,252,84,168,84,168,84,168,84,168
4160 DATA 84,168,84,168,84,168,252,0
4170 DATA 85,106,85,106,85,106,85,106,85,106
4180 DATA 88,168,88,168,88,168,88,168,88,168
4190 DATA 84,168,84,168,84,168,84,168,84,168
4200 DATA 252,140,140,84,4,84,220,140
4210 DATA 220,140,4,4,4,140,220,252
4220 DATA 252,180,0,0,0,132,204,252
4230 DATA 220,140,4,4,4,4,220,140
4240 DATA 140,116,244,204,188,124,4,252
4250 DATA 4,244,236,204,244,116,140,252
4260 DATA 236,204,172,108,4,236,236,252
4270 DATA 4,124,12,244,244,116,140,252
4280 DATA 204,188,124,12,116,116,140,25
2
```

```


4290 DATA 4,244,236,220,188,188,188,252
4300 DATA 140,116,116,140,116,116,140,2
52
4310 DATA 140,116,116,132,244,236,156,2
52
4320 DATA 100,88,88,88,88,88,100,252
4330 DATA 220,172,116,116,4,116,116,252
4340 DATA 196,236,236,236,236,108,156,2
52
4350 DATA 140,116,116,116,84,108,148,25
2
4360 DATA 116,108,92,60,92,108,116,252
4370 REM <<< PLAYER'S SCORE >>>
4380 SC(Q)=0:AC=0:FOR Z=1 TO NOC(Q):W=Z
4390 GOSUB 3270:IF RIGHT$(PAC$(SUIT,CAR)
),1)=CHR$(232) OR RIGHT$(PAC$(SUIT,CAR)
),1)=CHR$(233) OR RIGHT$(PAC$(SUIT,CAR),
1)=CHR$(234) THEN SC(Q)=SC(Q)+10:GOTO 4
420
4400 IF RIGHT$(PAC$(SUIT,CAR),1)=CHR$(2
31) THEN AC=AC+1:GOTO 4420
4410 PU=ASC(RIGHT$(PAC$(SUIT,CAR),1))-2
20:SC(Q)=SC(Q)+PU
4420 NEXT Z:IF AC>0 THEN 4460
4430 LOCATE CC(Q),17:PRINT " " :LOCATE
CC(Q),17:PRINT SC(Q)
4440 RETURN
4450 REM <<< AS NA MAO >>>
4460 FOR R=1 TO AC
4470 IF SC(Q)+11>21 THEN SC(Q)=SC(Q)+1
ELSE SC(Q)=SC(Q)+11
4480 NEXT R
4490 GOTO 4430
4500 LOCATE 1,20:PRINT STRING$(39," "):
RETURN
4510 FOR T=1 TO 1400:NEXT T:RETURN
4520 REM <<< DOBRAR >>>
4530 IF MO(Q)<BE(Q) THEN 3620
4540 BE(Q)=BE(Q)*2

```

```
4550 MO(Q)=MO(Q)-BE(Q)/2
4560 GOSUB 3690
4570 LOCATE CC(Q)+1,18:PRINT "    "
4580 LOCATE CC(Q)+1,18:PRINT BE(Q)
4590 LOCATE CC(Q)-5,19:PRINT "    "
4600 LOCATE CC(Q)-5,19:PRINT MO(Q)
4610 CX=CX+1:CX(Q)=CX(Q)+1
4620 GOTO 3590
4630 REM <<< FIM DO JOGO >>>
4640 GOSUB 4500:LOCATE 0,20:PRINT "Acab
ou o dinheiro de todos os jogadores"
4650 GOSUB 4510:GOTO 1090
```

14

Maior-Menor



O Dobro ou Nada!

Aqui está a versão computadorizada de outro conhecido jogo de cartas.

Cada jogador começa com Cz\$ 150,00 e recebe onze cartas. A primeira carta é virada e o jogador faz uma aposta, que pode ser qualquer quantia até o máximo de que dispõe na ocasião. O computador pede então que o jogador diga se acha que a carta seguinte vai ser maior ou menor que a última carta que foi mostrada. (Para escolher maior, entre com M; para escolher menor, entre com m). Se o jogador acertar, recebe uma quantia igual à que apostou; se escolher a alternativa incorreta ou se as duas cartas tiverem o mesmo valor, perde a aposta. O jogo continua até todas as cartas terem sido viradas.

```
1000 REM <<< MAIOR OU MENOR - ANDY/JIM
    >>>
1010 SCREEN 0:COLOR 15,4:KEY OFF
1020 DEFINT A-Z
1030 DIM PA(52),AV(52),PAC$(5,13),BC(11)
1040 DIM CX(11),CY(11)
1050 HI=0
1060 SCREEN 0:WIDTH 40
1070 GOSUB 2470
1080 CLS
1090 LOCATE 15,8:PRINT "M A I O R"
1100 LOCATE 18,10:PRINT "O U"
1110 LOCATE 15,12:PRINT "M E N O R"
1120 GOSUB 1400
1130 GOSUB 1520
1140 GOSUB 1900
1150 LOCATE 11,16:PRINT "Aposta:"
1160 LOCATE 11,17:PRINT "Total:"
1170 LOCATE 27,15:PRINT "Recorde:"
1180 TT=50
1190 BE=0
1200 GOSUB 2070
1210 GOSUB 2220
1220 C=1
1230 CX=CX(C):CY=CY(C)
1240 GOSUB 2070
1250 IF C=11 THEN 2910
1260 PLC(1,1)=BC(C):Q=1:W=1
1270 GOSUB 2000:GOSUB 1640
1280 GOSUB 2150
1290 GOSUB 2300
1300 GOSUB 2070
1310 CX=CX(C+1):CY=CY(C+1)
1320 PLC(1,1)=BC(C+1):CB=CAR
1330 Q=1:W=1:GOSUB 2000
1340 PLAY "02L64CRERF"
1350 GOSUB 1640
1360 IF CAR>CB AND A$="M" THEN 2420:GOT
    O 1230
```

```

1370 IF CAR<CB AND A$="m" THEN 2420:GOT
0 1230
1380 GOTO 2380
1390 REM <<< GERAR BARALHO >>>
1400 FOR SUIT=1 TO 4
1410 FOR CAR=1 TO 13
1420 PAC$(SUIT,CAR)=CHR$(SUIT+217)
1430 IF CAR<11 AND CAR>1 THEN PAC$(SUIT,
,CAR)=PAC$(SUIT,CAR)+CHR$(CAR+220)
1440 IF CAR=1 THEN PAC$(SUIT,CAR)=PAC$(SUIT,
,CAR)+CHR$(231)
1450 IF CAR=11 THEN PAC$(SUIT,CAR)=PAC$(SUIT,
,CAR)+CHR$(232)
1460 IF CAR=12 THEN PAC$(SUIT,CAR)=PAC$(SUIT,
,CAR)+CHR$(233)
1470 IF CAR=13 THEN PAC$(SUIT,CAR)=PAC$(SUIT,
,CAR)+CHR$(234)
1480 NEXT CAR
1490 NEXT SUIT
1500 RETURN
1510 REM <<< EMBARALHAR >>>
1520 FOR N=1 TO 52
1530 PA(N)=N:AV(N)=1
1540 NEXT N
1550 FOR A=1 TO 30
1560 N1=INT(RND(-TIME)*51)+1
1570 N2=INT(RND(-TIME)*51)+1
1580 CC=PA(N1)
1590 PA(N1)=PA(N2)
1600 PA(N2)=CC
1610 NEXT A
1620 RETURN
1630 REM <<< DESENHAR UMA CARTA >>>
1640 LOCATE CX-1,CY
1650 PRINT CHR$(200);:FOR Z=1 TO 5:PRIN
T CHR$(204);:NEXT Z:PRINT CHR$(203)
1660 LOCATE CX-1,CY+1
1670 PRINT CHR$(206);RIGHT$(PAC$(SUIT,C
AR),1);

```

```
1680 PRINT LEFT$(PAC$(SUIT,CAR),1);:PRINT STRING$(3,CHR$(208));CHR$(207)
1690 LOCATE CX-1,CY+2
1700 PRINT CHR$(206);STRING$(5,CHR$(208))
1710 LOCATE CX-1,CY+3
1720 PRINT CHR$(206);CHR$(208);CHR$(208)
1730 PRINT LEFT$(PAC$(SUIT,CAR),1);CHR$(208);CHR$(208);CHR$(207)
1740 LOCATE CX-1,CY+4
1750 PRINT CHR$(206);STRING$(5,CHR$(208))
1760 LOCATE CX-1,CY+5
1770 PRINT CHR$(206);CHR$(208);CHR$(208)
1780 LOCATE CX-1,CY+6
1790 PRINT CHR$(201);STRING$(5,CHR$(205))
1800 RETURN
1810 REM <<< DESENHAR AS COSTAS DE UMA
     CARTA >>>
1820 LOCATE CX-1,CY
1830 PRINT CHR$(209);STRING$(5,CHR$(213))
1840 FOR NN=1 TO 5:LOCATE CX-1,CY+NN
1850 PRINT CHR$(215);STRING$(5,CHR$(217))
1860 NEXT NN
1870 LOCATE CX-1,CY+6:PRINT CHR$(211);$TRING$(5,CHR$(214));CHR$(212)
1880 RETURN
1890 REM <<< DESENHAR AS CARTAS >>>
1900 RESTORE 1960:FOR A=1 TO 11
1910 READ CX(A),CY(A)
1920 CX=CX(A):CY=CY(A)
1930 GOSUB 1820
1940 NEXT A
```

```

1950 RETURN
1960 DATA 2,0,10,0,18,0,26,0,34,0
1970 DATA 2,7,10,7,18,7,26,7,34,7
1980 DATA 2,14
1990 REM <<< INTERPRETAR UMA CARTA >>>
>
2000 S=PLC(Q,W)/13:CAR=PLC(Q,W) MOD 13:
IF CAR=0 THEN CAR=13
2010 IF S>=0 THEN SUIT=1
2020 IF S>=1 THEN SUIT=2
2030 IF S>=2 THEN SUIT=3
2040 IF S>=3 THEN SUIT=4
2050 RETURN
2060 REM <<< RECORDE, TOTAL, APOSTA >>>
2070 LOCATE 18,16:PRINT "      "
2080 LOCATE 18,16:PRINT BE
2090 LOCATE 17,17:PRINT "      "
2100 LOCATE 17,17:PRINT TT
2110 LOCATE 28,16:PRINT "      "
2120 LOCATE 28,16:PRINT HI
2130 RETURN
2140 REM <<< RECEBER APOSTA >>>
2150 LOCATE 15,19:PRINT "Você tem $";TT
2160 LOCATE 15,20:PRINT "Quanto apostar"
2170 LOCATE 15,21:INPUT BE$:BE=VAL(BE$)
2180 IF BE<1 OR BE>TT THEN LOCATE 15,21
:PRINT STRING$(10," "):GOTO 2170
2190 TT=TT-BE:GOSUB 2070
2200 RETURN
2210 REM <<< GERAR AS CARTAS >>>
2220 FOR A=1 TO 11
2230 C=INT(RND(-TIME)*51)+1
2240 IF AV(C)=0 THEN 2230
2250 BC(A)=PA(C)
2260 AV(C)=0
2270 NEXT A
2280 RETURN
2290 REM <<< MAIOR / MENOR >>>
2300 LOCATE 15,20:PRINT STRING$(25," ")

```


```
2310 LOCATE 15,19:PRINT STRING$(20," ")
2320 LOCATE 15,21:PRINT STRING$(20," ")
2330 LOCATE 15,19:PRINT "MAIOR/menor"
2340 A$=INKEY$
2350 IF A$<>"m" AND A$<>"M" THEN 2340
2360 RETURN
2370 REM <<<< PERDE >>>>
2380 PLAY "02L64BAGFEDC01BAGFEDC"
2390 IF TT<1 THEN 2950
2400 C=C+1:GOTO 1230
2410 REM <<<< GANHA >>>>
2420 TT=TT+BE*2
2430 C=C+1
2440 PLAY "04L64CDEFGAB05CDEFGAB"
2450 GOTO 1230
2460 REM <<<< CARACTERES >>>>
2470 RESTORE 2540
2480 FOR C=200 TO 234
2490 FOR B=0 TO 7
2500 READ A:VPOKE 2048+C*8+B,A
2510 NEXT B
2520 NEXT C
2530 RETURN
2540 DATA 0,60,124,124,124,124,124,124,124
2550 DATA 124,124,124,124,124,124,124,60,0
2560 DATA 248,248,248,248,248,248,248,240,0
2570 DATA 0,240,248,248,248,248,248,248,248
2580 DATA 0,252,252,252,252,252,252,252,252
2590 DATA 252,252,252,252,252,252,252,252,0
2600 DATA 124,124,124,124,124,124,124,124,1
24
2610 DATA 248,248,248,248,248,248,248,248,2
48
2620 DATA 252,252,252,252,252,252,252,252,2
52
2630 DATA 0,60,84,104,84,104,84,104
2640 DATA 0,240,88,168,88,168,88,168
2650 DATA 84,104,84,104,84,104,60,0
2660 DATA 88,168,88,168,88,168,240,0
```

```
2670 DATA 0,252,84,168,84,168,84,168
2680 DATA 84,168,84,168,84,168,252,0
2690 DATA 85,106,85,106,85,106,85,106
2700 DATA 88,168,88,168,88,168,88,168
2710 DATA 84,168,84,168,84,168,84,168
2720 DATA 252,140,140,84,4,84,220,140
2730 DATA 220,140,4,4,4,140,220,252
2740 DATA 252,180,0,0,0,132,204,252
2750 DATA 220,140,4,4,4,4,220,140
2760 DATA 140,116,244,204,188,124,4,252
2770 DATA 4,244,236,204,244,116,140,252
2780 DATA 236,204,172,108,4,236,236,252
2790 DATA 4,124,12,244,244,116,140,252
2800 DATA 204,188,124,12,116,116,140,25
2
2810 DATA 4,244,236,220,188,188,188,252
2820 DATA 140,116,116,140,116,116,140,2
52
2830 DATA 140,116,116,132,244,236,156,2
52
2840 DATA 100,88,88,88,88,88,100,252
2850 DATA 220,172,116,116,4,116,116,252
2860 DATA 196,236,236,236,236,108,156,2
52
2870 DATA 140,116,116,116,84,108,148,25
2
2880 DATA 116,108,92,60,92,108,116,252
2890 FOR T=1 TO 5000:NEXT T:RETURN
2900 LOCATE 15,20
2910 LOCATE 15,20
2920 IF TT>HI THEN HI=TT:PRINT "Você ba
teu o recorde"
2930 LOCATE 15,21:PRINT "Você acabou co
m $";TT
2940 GOSUB 2890:GOTO 1080
2950 LOCATE 15,20:PRINT STRING$(25," ")
2960 LOCATE 15,20:PRINT "Seu dinheiro a
cabou"
2970 GOSUB 2890
```

```
2980 GOTO 1080
2990 DEFINT A
3000 A=50
3010 PRINT A
3020 A=A*2
3030 GOTO 3010
```

15

Memória



Encontre os Pares

Este jogo aparentemente fácil exige boa memória e um grande poder de concentração.

O jogo começa com todas as cartas com a face virada para baixo. O primeiro jogador vira duas cartas. Se elas formam um par, o jogador faz um ponto e as cartas são retiradas; caso contrário, as cartas são viradas de novo com a face para baixo. Nos dois casos, a vez passa para o segundo jogador. O jogo prossegue até acabarem as cartas; o vencedor é aquele que conseguir encontrar o maior número de pares.

Em algumas versões deste jogo, a pessoa que descobre um par tem o direito de virar mais duas cartas. No nosso caso, porém, em que são usadas apenas 12 cartas, o jogo fica mais interessante se só for permitido virar um par a cada jogada.

```
1000 REM <<< MEMORIA - ANDY/JIM >>>
1010 SCREEN 0:KEY OFF
1020 WIDTH 40
1030 COLOR 15,4
1040 DEFINT A-Z
1050 DIM PAC$(5,13),CC(3),SC(2)
1060 DIM PA(52),AV(52)
1070 DIM CX(12),CY(12)
1080 DIM CA(12),BC(12)
1090 DIM PLC(1,2),AC(12)
1100 GOSUB 2820
1110 CLS:GOSUB 1250
1120 SC(1)=0:SC(2)=0
1130 GOSUB 1250
1140 GOSUB 1370
1150 GOSUB 1810
1160 GOSUB 2060
1170 FOR A=1 TO 12:AC(A)=1:NEXT A
1180 P=1
1190 GOSUB 2550
1200 GOSUB 2240
1210 GOTO 2430
1220 IF P=1 THEN P=2 ELSE P=1
1230 GOTO 1190
1240 REM <<< GERAR BARALHO >>>
1250 FOR SUIT=1 TO 4
1260 FOR CAR=1 TO 13
1270 PAC$(SUIT,CAR)=CHR$(SUIT+217)
1280 IF CAR<11 AND CAR>1 THEN PAC$(SUIT,CAR)=PAC$(SUIT,CAR)+CHR$(CAR+220)
1290 IF CAR=1 THEN PAC$(SUIT,CAR)=PAC$(SUIT,CAR)+CHR$(231)
1300 IF CAR=11 THEN PAC$(SUIT,CAR)=PAC$(SUIT,CAR)+CHR$(232)
1310 IF CAR=12 THEN PAC$(SUIT,CAR)=PAC$(SUIT,CAR)+CHR$(233)
1320 IF CAR=13 THEN PAC$(SUIT,CAR)=PAC$(SUIT,CAR)+CHR$(234)
1330 NEXT CAR
```

```

1340 NEXT SUIT
1350 RETURN
1360 REM <<< EMBARALHAR >>>
1370 FOR N=1 TO 52
1380 PA(N)=N:AV(N)=1
1390 NEXT N
1400 FOR A=1 TO 30
1410 N1=INT(RND(-TIME)*51)+1
1420 N2=INT(RND(-TIME)*51)+1
1430 CC=PA(N1)
1440 PA(N1)=PA(N2)
1450 PA(N2)=CC
1460 NEXT A
1470 RETURN
1480 REM <<< DESENHAR UMA CARTA >>>
1490 LOCATE CX-1,CY
1500 PRINT CHR$(200);:FOR Z=1 TO 5:PRINT
    CHR$(204);:NEXT Z:PRINT CHR$(203)
1510 LOCATE CX-1,CY+1
1520 PRINT CHR$(206);RIGHT$(PAC$(SUIT,C
AR),1);
1530 PRINT LEFT$(PAC$(SUIT,CAR),1);:PRI
NT STRING$(3,CHR$(208));CHR$(207)
1540 LOCATE CX-1,CY+2
1550 PRINT CHR$(206);STRING$(5,CHR$(208
));CHR$(207)
1560 LOCATE CX-1,CY+3
1570 PRINT CHR$(206);CHR$(208);CHR$(208
);
1580 PRINT LEFT$(PAC$(SUIT,CAR),1);CHR$(
208);CHR$(208);CHR$(207)
1590 LOCATE CX-1,CY+4
1600 PRINT CHR$(206);STRING$(5,CHR$(208
));CHR$(207)
1610 LOCATE CX-1,CY+5
1620 PRINT CHR$(206);CHR$(208);CHR$(208
);CHR$(208);CHR$(208);RIGHT$(PAC$(SUIT,
CAR),1);CHR$(207)
1630 LOCATE CX-1,CY+6

```

```
1640 PRINT CHR$(201);STRING$(5,CHR$(205
));CHR$(202)
1650 RETURN
1660 REM <<< DESENHAR AS COSTAS DE UMA
CARTA >>>
1670 LOCATE CX-1,CY
1680 PRINT CHR$(209);STRING$(5,CHR$(213
));CHR$(210)
1690 FOR NN=1 TO 5:LOCATE CX-1,CY+NN
1700 PRINT CHR$(215);STRING$(5,CHR$(217
));CHR$(216)
1710 NEXT NN
1720 LOCATE CX-1,CY+6:PRINT CHR$(211);S
TRING$(5,CHR$(214));CHR$(212)
1730 RETURN
1740 REM <<< APAGAR UMA CARTA >>>
1750 LOCATE CX-1,CY
1760 FOR Y=1 TO 7
1770 LOCATE CX-1,CY+Y-1:PRINT "
1780 NEXT Y
1790 RETURN
1800 REM <<< GERAR CARTAS >>>
1810 RESTORE 2010:FOR A=1 TO 12
1820 CA(A)=1:NEXT A
1830 FOR A=1 TO 10 STEP 2
1840 READ C1,C2
1850 C=INT(RND(-TIME)*11)+1
1860 IF CA(C)=0 THEN 1850
1870 BC(C)=C1:CA(C)=0
1880 C=INT(RND(-TIME)*11)+1
1890 IF CA(C)=0 THEN 1880
1900 BC(C)=C2:CA(C)=0
1910 NEXT A
1920 A=0
1930 A=A+1
1940 IF A<13 THEN IF CA(A)=1 THEN READ
BC(A):CA(A)=0 ELSE 1930
1950 A=A+1
1960 IF A<13 THEN IF CA(A)=1 THEN READ
BC(A):CA(A)=0 ELSE 1950
```

```

1970 FOR A=1 TO 12
1980 READ CX(A),CY(A)
1990 NEXT A
2000 RETURN
2010 DATA 1,14,2,15,3,16,4,17,5,18,6,19
2020 DATA 2,0,10,0,18,0,26,0
2030 DATA 2,7,10,7,18,7,26,7
2040 DATA 2,14,10,14,18,14,26,14
2050 REM <<< TELA >>>
2060 CLS
2070 FOR A=1 TO 12
2080 CX=CX(A):CY=CY(A)
2090 GOSUB 1670
2100 NEXT A
2110 LOCATE 33,3:PRINT "Pontos1"
2120 LOCATE 33,6:PRINT "Pontos2"
2130 LOCATE 32,15:PRINT "Jogador1"
2140 LOCATE 32,17:PRINT "Jogador2"
2150 RETURN
2160 REM <<< INTERPRETAR CARTA >>>
2170 S=PLC(Q,W)/13:CAR=PLC(Q,W) MOD 13:
IF CAR=0 THEN CAR=13
2180 IF S>=0 THEN SUIT=1
2190 IF S>=1 THEN SUIT=2
2200 IF S>=2 THEN SUIT=3
2210 IF S>=3 THEN SUIT=4
2220 RETURN
2230 REM <<< ENTRADA DE DADOS >>>
2240 LOCATE 1,21:PRINT STRING$(35," ")
2250 LOCATE 2,21:PRINT "Primeiro número
";
2260 INPUT N1
2270 PLAY "03L64CRE"
2280 IF N1<1 OR N1>12 THEN 2250
2290 IF AC(N1)=0 THEN 2250
2300 PLC(1,1)=BC(N1):Q=1:W=1
2310 GOSUB 2170:CX=CX(N1):CY=CY(N1)
2320 GOSUB 1490
2330 LOCATE 2,21:PRINT "Segundo número
";;
```

```

2340 INPUT N2
2350 PLAY "03L64CRE"
2360 IF N2<1 OR N2>12 OR N1=N2 THEN 233
0
2370 IF AC(N2)=0 THEN 2330
2380 PLC(1,2)=BC(N2):Q=1:W=2
2390 GOSUB 2170:CX=CX(N2):CY=CY(N2)
2400 GOSUB 1490
2410 RETURN
2420 REM <<< VERIFICAR SE E UM PAR >>>
>
2430 Q=1:W=1:GOSUB 2170
2440 A1=CAR:Q=1:W=2:GOSUB 2170
2450 A2=CAR
2460 IF A1=A2 AND N1<>N2 THEN 2610
2470 LOCATE 1,21:PRINT STRING$(35," ")
2480 PLAY "02L64BAGFEDC01BAGFEDC"
2490 LOCATE 1,21:PRINT "As cartas são d
iferentes"
2500 GOSUB 3240
2510 CX=CX(N1):CY=CY(N1)
2520 GOSUB 1670:CX=CX(N2):CY=CY(N2)
2530 GOSUB 1670:GOTO 1220
2540 REM <<< IMPRIMIR A CONTAGEM >>>
2550 LOCATE 33,4:PRINT SC(1)
2560 LOCATE 33,7:PRINT SC(2)
2570 IF P=1 THEN LOCATE 32,15:PRINT "JO
GADOR1":LOCATE 32,17:PRINT "jogador2"
2580 IF P=2 THEN LOCATE 32,15:PRINT "JO
GADOR2":LOCATE 32,17:PRINT "jogador1"
2590 RETURN
2600 REM <<< EXISTE UM PAR >>>
2610 LOCATE 1,21:PRINT STRING$(35," ")
2620 PLAY "05L64CDEFGAB06CDEFGAB"
2630 LOCATE 1,21:PRINT "As cartas são i
guais"
2640 AC(N1)=0:AC(N2)=0
2650 GOSUB 3240
2660 CX=CX(N1):CY=CY(N1)


```

```
2670 GOSUB 1750:CX=CX(N2):CY=CY(N2)
2680 GOSUB 1750:SC(P)=SC(P)+5
2690 IF SC(1)+SC(2)=30 THEN 2720
2700 GOTO 1220
2710 REM <<< FIM DO JOGO >>>
2720 CLS
2730 PRINT:PRINT "Total do jogador 1:";SC(1)
2740 PRINT:PRINT "Total do jogador 2:";SC(2)
2750 PRINT:PRINT:LOCATE 3,10:PRINT "Ape
rte 'ESPAÇO' para jogar de novo"
2760 LOCATE 11,12:PRINT "ou 'P' para pa
rar."
2770 A$=INKEY$
2780 IF A$=" " THEN 1110
2790 IF A$="P" OR A$="p" THEN END
2800 GOTO 2770
2810 REM <<< CARACTERES >>>
2820 RESTORE 2890
2830 FOR C=200 TO 234
2840 FOR B=0 TO 7
2850 READ A:VPOKE 2048+C*B+B,A
2860 NEXT B
2870 NEXT C
2880 RETURN
2890 DATA 0,60,124,124,124,124,124,124,124
2900 DATA 124,124,124,124,124,124,124,60,0
2910 DATA 248,248,248,248,248,248,248,240,0
2920 DATA 0,240,248,248,248,248,248,248,248
2930 DATA 0,252,252,252,252,252,252,252,252
2940 DATA 252,252,252,252,252,252,252,252,0
2950 DATA 124,124,124,124,124,124,124,124,1
24
2960 DATA 248,248,248,248,248,248,248,248,2
48
2970 DATA 252,252,252,252,252,252,252,252,2
52
2980 DATA 0,60,84,104,84,104,84,104
```

```
2990 DATA 0,240,88,168,88,168,88,168
3000 DATA 84,104,84,104,84,104,60,0
3010 DATA 88,168,88,168,88,168,240,0
3020 DATA 0,252,84,168,84,168,84,168
3030 DATA 84,168,84,168,84,168,252,0
3040 DATA 85,106,85,106,85,106,85,106
3050 DATA 88,168,88,168,88,168,88,168
3060 DATA 84,168,84,168,84,168,84,168
3070 DATA 252,140,140,84,4,84,220,140
3080 DATA 220,140,4,4,4,140,220,252
3090 DATA 252,180,0,0,0,132,204,252
3100 DATA 220,140,4,4,4,4,220,140
3110 DATA 140,116,244,204,188,124,4,252
3120 DATA 4,244,236,204,244,116,140,252
3130 DATA 236,204,172,108,4,236,236,252
3140 DATA 4,124,12,244,244,116,140,252
3150 DATA 204,188,124,12,116,116,140,25
2
3160 DATA 4,244,236,220,188,188,188,252
3170 DATA 140,116,116,140,116,116,140,2
52
3180 DATA 140,116,116,132,244,236,156,2
52
3190 DATA 100,88,88,88,88,88,100,252
3200 DATA 220,172,116,116,4,116,116,252
3210 DATA 196,236,236,236,236,108,156,2
52
3220 DATA 140,116,116,116,84,108,148,25
2
3230 DATA 116,108,92,60,92,108,116,252
3240 FOR T=1 TO 2000:NEXT T:RETURN
```

16

Senha



Um Jogo de Raciocínio

O objetivo deste jogo é descobrir uma combinação secreta de quatro cores (entre seis cores possíveis), escolhendo quatro cores em uma certa ordem. Depois de cada tentativa, o computador fornece as seguintes indicações:

Um ponto de exclamação para cada cor certa no lugar certo.
Um ponto de interrogação para cada cor certa no lugar errado.

Para tornar o jogo mais difícil, a ordem em que as informações são mostradas pelo computador não tem relação com a ordem com que as cores foram escolhidas; primeiro são mostrados todos os pontos de exclamação e depois todos os pontos de interrogação.

As cores são escolhidas usando-se o *joystick* ou as teclas de controle do cursor para colocar uma seta sobre a cor desejada. O botão de tiro ou a tecla de espaço é então usado para “fixar” a cor. O processo é repetido para cada uma das quatro posições.

O jogador tem dez oportunidades para descobrir a seqüência correta. Depois da décima tentativa, o computador mostra a combinação secreta na parte de cima da tela. Para jogar de novo, basta apertar a tecla S.

No início do jogo, o computador oferece a opção de usar símbolos em lugar de cores. Esta opção é particularmente recomendada no caso de receptores de TV em preto e branco.

```

1000 REM <<< SENHA - ISSI >>>
1010 SCREEN 2,0,0:COLOR 10,1,0:CLS
1020 MAXFILES=2:OPEN "GRP:S" FOR OUTPUT
    AS 1
1030 PRESET (30,175):PRINT #1,"CORES OU
    SIMBOLOS (C/S) ?"
1040 A$=INPUT$(1):A=ASC(A$) AND 223
1050 IF A=67 OR A = 99 THEN GR=0:GOTO 1
080
1060 IF A=83 OR A=115 THEN GR=1:GOTO 10
80
1070 GOTO 1040
1080 CLS
1090 CX=2:CY=180
1100 PRESET (8,4):PRINT #1, "      S E N
    H A"
1110 W$="###"
1120 PRESET (8,15):PRINT #1,W$;W$;W$;W$
1130 W$=" "+CHR$(1)+CHR$(79)+CHR$(1)+CH
    R$(79)+" "
1140 FOR N=40 TO 150 STEP 12:COLOR 9
1150 PRESET (8,N):PRINT #1,W$;W$;W$;W$
1160 NEXT N
1170 GOSUB 1510
1180 GOSUB 1240
1190 GOSUB 1310
1200 GOSUB 1600
1210 IF A$="!!!!" THEN GOTO 2080 ELSE A
    $=" "

```

```

1220 TRY=TRY+1:IF TRY=10 THEN GOTO 1970
    ELSE GOTO 1190
1230 REM <<< SORTEIO DAS CORES >>>
1240 A$="123456":FOR PLA=1 TO 4
1250 N=INT(RND(-TIME)*LEN(A$))+1
1260 NN(PLA)=VAL(MID$(A$,N,1))
1270 A$=LEFT$(A$,N-1)+RIGHT$(A$,LEN(A$)-N)
1280 NEXT PLA
1290 RETURN
1300 REM <<< CORES/SIMBOLOS >>>
1310 RESTORE:PRESET (40,170):FOR N=1 TO
    6:READ C,R:COLOR C:C$=CHR$(1)+CHR$(R)
1320 IF GR=0 THEN PRINT #1,CHR$(1)+CHR$(72)+CHR$(1)+CHR$(72); ELSE PRINT #1,C$;
    ;" ";
1330 NEXT N
1340 TX=1:CY=180:W$=CHR$(1)+CHR$(75)
1350 COLOR 9:PRESET (CX*16+8,CY):PRINT
    #1,W$;:FOR W=1 TO 10:NEXT W
1360 COLOR 1:PRESET (CX*16+8,CY):PRINT
    #1,W$;
1370 GOSUB 2240
1380 IF LE=1 THEN CX=CX-1
1390 IF RI=1 THEN CX=CX+1
1400 IF CX=8 THEN CX=2
1410 IF CX=1 THEN CX=7
1420 IF FI=1 THEN PLAY "L64CDE":GOTO 14
40
1430 GOTO 1350
1440 PRESET (TX*32-16,TRY*12+40):COLOR
    1:PRINT #1,CHR$(1)+CHR$(79)+CHR$(1)+CHR
    $(79)
1450 RESTORE:FOR W=1 TO CX-1:READ C,R:N
    EXT W:COLOR C,1:C$=CHR$(1)+CHR$(R)
1460 PRESET (TX*32-16,TRY*12+40)
1470 IF GR=0 THEN PRINT #1,CHR$(1)+CHR$(72)+CHR$(1)+CHR$(72) ELSE PRINT #1,C$;
    C$

```

```

1480 PE(TX)=CX-1:TX=TX+1
1490 IF TX=5 THEN RETURN ELSE GOTO 1350
1500 REM <<< CABEÇA >>>
1510 COLOR 7:PSET (250,160)
1520 RESTORE 2310
1530 FOR N=1 TO 25:READ A,B:A=A/2.5:B=1
92-B/2
1540 IF A<180 THEN A=A+25
1550 D$="M"+STR$(INT(A))+","+STR$(INT(B))
)) 
1560 DRAW D$:NEXT N
1570 PSET (220,106):DRAW "U2BU4U20E10U1
0H10L10G10"
1580 RETURN
1590 REM <<< VERIFICAÇÃO >>>
1600 RJ=0
1610 FOR X=1 TO 4:L(X)=0:NEXT X
1620 FOR X=1 TO 4:FOR Y=1 TO 4
1630 IF PE(X)=PE(Y) AND X<>Y THEN L(X)=
1:RJ=RJ+1
1640 NEXT Y,X
1650 RJ=RJ/2:IF RJ>3 THEN RJ=3
1660 GOSUB 1690
1670 IF RJ=0 THEN GOSUB 1770:RETURN
1680 IF RJ<>0 THEN GOSUB 1840:E=E-CE:GO
SUB 1770:RETURN
1690 D=0:E=0
1700 FOR X=1 TO 4:L1(X)=0:L2(X)=0:NEXT
X
1710 FOR Y=1 TO 4:FOR Z=1 TO 4
1720 IF PE(Y)=NN(Z) THEN D=D-(Y=Z):E=E-
(Y<>Z):L1(Y)==-(Y=Z):L2(Y)==-(Y<>Z)
1730 NEXT Z,Y
1740 PP=D
1750 QQ=E
1760 RETURN
1770 A$=STRING$(D,CHR$(33))+STRING$(E,C
HR$(63))
1780 FOR X=1 TO 4

```

```


1790 IF MID$(A$,X,1)="!" THEN PLAY "E"
1800 IF MID$(A$,X,1)="?" THEN PLAY "F"
ELSE PLAY "A"
1810 NEXT X
1820 COLOR 4:PRESET (136,TRY*12+40):PRINT #1,A$
1830 RETURN
1840 CC=0:CE=0
1850 FOR X=1 TO 4
1860 IF L2(X)=1 AND L(X)= 1 THEN CE=CE+1
1870 IF L1(X)=1 AND L(X)=1 THEN CC=CC+1
1880 NEXT X
1890 IF CC=0 AND PP=0 AND QQ<>0 THEN CE=CE-1
1900 IF D=0 AND E=0 THEN CE=0
1910 IF PP=1 AND RJ=2 THEN CE=CE-1
1920 IF QQ=1 AND RJ=2 THEN CE=CE-1
1930 IF QQ=4 AND RJ=2 THEN CE=CE-1
1940 IF QQ=1 AND RJ=3 THEN CE=CE+1
1950 RETURN
1960 REM <<< PERDEU >>>
1970 LINE (8,15)-(170,25),1,BF
1980 FOR N=1 TO 4
1990 RESTORE:FOR W=1 TO NN(N):READ C,R:NEXT W:C$=CHR$(1)+CHR$(R)
2000 IF GR=0 THEN C$=CHR$(1)+CHR$(72)
2010 PRESET (N*40-24,15):COLOR C:PRINT #1,C$;C$;C$;C$
2020 NEXT N
2030 PRESET (190,20):PRINT #1,"ERROU!"
2040 PLAY "L8GFL4EL8FEL4DL8EDL4CD"
2050 FOR N=1 TO 4000:NEXT N
2060 GOTO 2180
2070 REM <<< GANHOU >>>
2080 FOR N=1 TO 4
2090 RESTORE:FOR W=1 TO NN(N):READ C,R:C$=CHR$(1)+CHR$(R):NEXT W
2100 LINE(N*40-24,15)-(N*40+16,25),1,BF

```

```
2110 IF GR=0 THEN C$=CHR$(1)+CHR$(72)
2120 PRESET (N*40-24,15):COLOR C:PRINT
#1,STRING$(4,CHR$(1)+C$)
2130 NEXT N
2140 PLAY "LBD#F#G#L4A#LBR8A#G#F#G#FD#L
4C#"
2150 CLS
2160 PRESET (90,50):PRINT #1,"MUITO BEM
!"
2170 PRESET (10,70):PRINT #1,"VOCE PREC
ISOU DE";TRY+1;"TENTATIVAS!"
2180 PRESET (25,160):PRINT #1,"QUER JOG
AR DE NOVO (S/N)?"
2190 A$=INPUT$(1):A=ASC(A$) AND 223
2200 IF A=83 THEN TRY=0:GOTO 1080
2210 IF A=78 THEN END
2220 GOTO 2190
2230 REM <<< ENTRADA DE DADOS >>>
2240 LE=0:RI=0:FI=0
2250 DI=STICK(0) OR STICK(1) OR STICK(2
)
2260 IF DI=7 THEN LE=1
2270 IF DI=3 THEN RI=1
2280 FI=ABS(STRIG(0) OR STRIG(1) OR STR
IG(2))
2290 RETURN
2300 DATA 7,77,2,67,10,69,15,70,8,68,13
,78
2310 DATA 560,110,605,180,615,220,615,2
50,605,290,555,320,500,320
2320 DATA 415,290,395,250,365,200,375,1
95,390,190,400,170,395,165
2330 DATA 400,160,395,155,400,150,400,1
45,395,140,395,130,400,125
2340 DATA 415,120,425,110,430,95,405,50
```

17

Alberto Diz



Alberto Diz: Carregue Este Programa

Se não me engano, esta é a primeira versão para computador do jogo para crianças conhecido como Simão Diz.

O programa pode ser usado para ensinar os conceitos de esquerda e direita, mas é preciso não esquecer que, da forma como foi escrito, direita e esquerda se referem ao boneco que aparece na tela e não à pessoa que está sentada diante do computador!

Você começa o jogo com apenas um ponto. A mensagem do computador informa o que você tem a fazer e também quem está dando a ordem. Se é Alberto que está mandando, obedeça prontamente, mas se a ordem partiu de Roberto, deve ser ignorada! O bonequinho na tela obedece a todos os seus comandos.

Você pode receber quatro ordens diferentes. Para executar as ordens

“Para Cima” e “Para Baixo”, use o *joystick* ou as teclas de controle do cursor da forma normal. As ordens “Esquerda” e “Direita” são cumpridas usando o *joystick* ou as teclas de controle do cursor, mas em relação ao boneco que aparece na tela. Em outras palavras, se você mover o *joystick* para a esquerda ou apertar a tecla de controle do cursor da esquerda, o boneco levantarão o braço direito. Esta complicação foi introduzida para tornar o jogo mais difícil. Você tem que decidir rapidamente se vai apertar alguma tecla e qual a tecla que deve ser apertada, a da esquerda ou a da direita. Jogue algumas vezes e verá que não é tão fácil como parece.

```

1000 REM <<< ALBERTO DIZ - JIM/ANDY >>
>>
1010 SCREEN 1,1:KEY OFF
1020 FOR A=1 TO 9
1030 FOR B=1 TO 8
1040 READ A$
1050 S$(A)=S$(A)+CHR$(VAL("8H"+A$))
1060 NEXT B
1070 SPRITE$(A)=S$(A)
1080 NEXT A
1090 DATA 18,24,42,A5,81,5A,24,18
1100 DATA 31,9,5,3,1,1,1,1
1110 DATA 1,1,1,3,5,9,11,21
1120 DATA 8C,90,A0,C0,80,80,80,80
1130 DATA 80,80,80,C0,A0,90,88,84
1140 DATA 41,7F,0,0,0,0,0,0
1150 DATA 1,1,2,4,8,10,20,60
1160 DATA 82,FE,0,0,0,0,0,0
1170 DATA 80,80,40,20,10,8,4,6
1180 REM <<< INICIO DO JOGO >>>
1190 GOSUB 1600
1200 SC=1:GOSUB 1910
1210 FOR AA=1 TO 500:NEXT AA
1220 GOSUB 1790
1230 GOSUB 1670
1240 IF GO=PO AND OG=1 THEN MO=MO+1:SC=
SC+1:GOSUB 1910:GOTO 1210
1250 IF TI=200 AND OG=2 THEN 1290
1260 IF TI=200 THEN 1280
1270 IF OG=2 OR FO<>GO THEN SC=SC-1:MO=
MO-1:GOSUB 2020

```

```
1280 IF SC<1 THEN 2100
1290 GOSUB 1910:GOTO 1210
1300 GOTO 1300
1310 REM <<< FIGURA 'PARA BAIXO' >>>
1320 PUT SPRITE M,(X,Y),15,1
1330 PUT SPRITE M+1,(X-8,Y+16),15,3
1340 PUT SPRITE M+2,(X+8,Y+16),15,5
1350 PUT SPRITE M+3,(X-8,Y+32),15,7
1360 PUT SPRITE M+4,(X+8,Y+32),15,9
1370 RETURN
1380 REM <<< FIGURA 'PARA CIMA' >>>
1390 PUT SPRITE M,(X,Y),15,1
1400 PUT SPRITE M+1,(X-8,Y+16),15,2
1410 PUT SPRITE M+2,(X+8,Y+16),15,4
1420 PUT SPRITE M+3,(X-8,Y+32),15,6
1430 PUT SPRITE M+4,(X+8,Y+32),15,8
1440 RETURN
1450 REM <<< FIGURA 'ESQUERDA' >>>
1460 PUT SPRITE M,(X,Y),15,1
1470 PUT SPRITE M+1,(X-8,Y+16),15,3
1480 PUT SPRITE M+2,(X+8,Y+16),15,4
1490 PUT SPRITE M+3,(X-8,Y+32),15,7
1500 PUT SPRITE M+4,(X+8,Y+32),15,8
1510 RETURN
1520 REM <<< FIGURA 'DIREITA' >>>
1530 PUT SPRITE M,(X,Y),15,1
1540 PUT SPRITE M+1,(X-8,Y+16),15,2
1550 PUT SPRITE M+2,(X+8,Y+16),15,4
1560 PUT SPRITE M+3,(X-8,Y+32),15,6
1570 PUT SPRITE M+4,(X+8,Y+32),15,9
1580 RETURN
1590 REM <<< TELA >>>
1600 LOCATE 10,0:PRINT "ALBERTO DIZ"
1610 LOCATE 10,1:PRINT "-----"
1620 LOCATE 2,18:PRINT "PONTOS"
1630 X=180:Y=30:M=1:GOSUB 1320
1640 X=180:Y=100:M=6:GOSUB 1320
1650 RETURN
1660 REM <<< ENTRADA DE DADOS >>>
```

```


1670 TI=0
1680 AA=STICK(0) OR STICK(1)
1690 IF AA=0 THEN 1740
1700 IF AA=3 THEN GO=1:GOTO 1760
1710 IF AA=7 THEN GO=2:GOTO 1760
1720 IF AA=1 THEN GO=3:GOTO 1760
1730 IF AA=5 THEN GO=4:GOTO 1760
1740 TI=TI+1:IF TI=200 THEN 1970
1750 GOTO 1680
1760 X=180:Y=100:M=6
1770 ON GO GOTO 1460,1530,1390,1320
1780 REM <<< ESCOLHER POSIÇÃO >>>
1790 OG=INT(RND(1)*5)+1:IF OG<3 THEN OG
=2:GOTO 1810
1800 OG=1
1810 PO=INT(RND(1)*4)+1
1820 X=180:Y=30:M=1
1830 IF OG=1 THEN LOCATE 3,5:PRINT "ALB
ERTO DIZ"
1840 IF OG=2 THEN LOCATE 3,5:PRINT "ROB
ERTO DIZ"
1850 IF PO=1 THEN LOCATE 3,7:PRINT "ESQ
UERDA"
1860 IF PO=2 THEN LOCATE 3,7:PRINT "DIR
EITA"
1870 IF PO=3 THEN LOCATE 3,7:PRINT "PAR
A CIMA"
1880 IF PO=4 THEN LOCATE 3,7:PRINT "PAR
A BAIXO"
1890 ON PO GOTO 1460,1530,1390,1320
1900 REM <<< CONTAGEM >>>
1910 LOCATE 3,5:PRINT " "
1920 LOCATE 3,7:PRINT " "
1930 LOCATE 3,9:PRINT " "
1940 LOCATE 3,19:PRINT SC
1950 RETURN
1960 REM <<< TEMPO ESGOTADO >>>
1970 LOCATE 3,5:PRINT " "
1980 LOCATE 3,7:PRINT " "

```

```
1990 IF OG=2 THEN LOCATE 3,5:PRINT "MUI  
TO BEM!":LOCATE 3,7:PRINT "FOI ROBERTO"  
:LOCATE 3,9:PRINT "QUE DISSE":SC=SC+1:FOR T=1 TO 800:NEXT T:LOCATE 3,7:PRINT "  
":RETURN  
2000 LOCATE 3,7:PRINT "TEMPO ESGOTADO":  
SC=SC-1:FOR T=1 TO 800:NEXT T:RETURN  
2010 REM <<< MOVIMENTO ERRADO >>>  
2020 LOCATE 3,5:PRINT " " " "  
2030 LOCATE 3,7:PRINT " " " "  
2040 LOCATE 3,5:PRINT "ERRADO!" "  
2050 IF OG=2 THEN LOCATE 3,7:PRINT "FOI  
ROBERTO":LOCATE 3,9:PRINT "QUE DISSE"  
2060 FOR T=1 TO 800:NEXT T  
2070 LOCATE 3,7:PRINT "  
"  
2080 RETURN  
2090 REM <<< FIM DO JOGO >>>  
2100 LOCATE 3,5:PRINT " " " "  
2110 LOCATE 3,7:PRINT " " " "  
2120 LOCATE 3,7:PRINT "JOGO ENCERRADO"  
2130 FOR T=1 TO 800:NEXT T  
2140 SC=0:LOCATE 3,7:PRINT "  
"  
2150 LOCATE 3,7:PRINT "OUTRA VEZ " "  
2160 FOR T=1 TO 800:NEXT T  
2170 LOCATE 3,7:PRINT " " "  
2180 GOTO 1190
```

18

Papagaios



Vamos Imitar os Papagaios?

Aqui estão quatro papagaios, cada um de uma cor e com uma corda musical diferente no bico.

Quando o programa começa, um dos papagaios puxa a corda, produzindo uma nota musical. Você deve reproduzir essa nota. Em seguida, o computador mostra uma seqüência de dois papagaios (duas notas), que você também deve repetir, e assim por diante.

Os poleiros dos pássaros foram dispostos na tela de modo a corresponderem à posição das teclas de controle do cursor. Assim, quando você aperta a tecla com a seta para cima o papagaio de cima puxa uma corda, quando aperta a tecla com a seta para a esquerda é a vez do papagaio da esquerda e assim por diante. Como nos outros jogos, pode ser usado um *joystick* em lugar das teclas de controle do cursor.

Quando testamos o programa, ninguém conseguiu reproduzir uma

seqüência de mais de 25 notas; a média ficou por volta de 12. O programa foi feito para produzir seqüências de até 50 notas. Se você conseguir chegar ao final, parabéns! Você possui uma memória fantástica...ou será que é um andróide?

```

1000 REM <<< PAPAGAIOS - ANDY/JIM >>>
1010 X=RND(-TIME)
1020 DIM NO(50)
1030 SCREEN 1,0,0
1040 GOSUB 1710
1050 GOSUB 1820
1060 SC=0:HI=0
1070 GOSUB 2050
1080 FOR A=1 TO 4
1090 GOSUB 1530
1100 NEXT A
1110 PO=0
1120 REM <<< LOOP PRINCIPAL >>>
1130 N=INT(RND(1)*4)+1
1140 PO=PO+1
1150 NO(PO)=N
1160 GOSUB 2170:LOCATE 20,10:PRINT "Minha vez"
1170 FOR B=1 TO PO
1180 GOSUB 2220
1190 NEXT B
1200 FOR T=1 TO 500:NEXT T
1210 A$=INKEY$:IF A$<>"" THEN 1210
1220 B=1
1230 GOSUB 2170:LOCATE 20,10:PRINT "Sua vez"
1240 A$=INKEY$:IF A$="" THEN 1240
1250 AA=ASC(A$)
1260 IF AA=29 THEN A=4:GOSUB 1650:GOSUB
1590:FOR T=1 TO 100:NEXT T
1270 IF AA=28 THEN A=2:GOSUB 1650:GOSUB
1590:FOR T=1 TO 100:NEXT T
1280 IF AA=30 THEN A=1:GOSUB 1650:GOSUB
1590:FOR T=1 TO 100:NEXT T

```

```
1290 IF AA=31 THEN A=3:GOSUB 1650:GOSUB
 1590:FOR T=1 TO 100:NEXT T
1300 IF AA<28 OR AA>31 THEN 1240
1310 GOSUB 1650
1320 GOSUB 1530
1330 IF NO(B)=A THEN 1340 ELSE 1370
1340 IF B=PO THEN GOSUB 2170:LOCATE 20,
10:PRINT "Muito bem":LOCATE 16,12:PRINT
PO;"at agora"
1350 IF B=PO THEN FOR T=1 TO 1000:NEXT
T:GOSUB 2170:GOTO 1130
1360 B=B+1:GOTO 1240
1370 GOSUB 2170
1380 LOCATE 17,10:PRINT "Sequência":LOC
ATE 17,11:PRINT "errada":LOCATE 17,12:P
RINT "depois de";PO-1:LOCATE 17,13:P
RINT "jogadas"
1390 FOR T=1 TO 1000:NEXT T
1400 GOSUB 2170
1410 LOCATE 17,10:PRINT "A sequência":L
OCATE 16,12:PRINT "    era:"
1420 FOR T=1 TO 1000:NEXT T
1430 FOR B=1 TO PO
1440 GOSUB 2220
1450 NEXT B
1460 GOSUB 2170
1470 LOCATE 20,10:PRINT "Outra vez":X=R
ND(-TIME)
1480 FOR T=1 TO 1000:NEXT T
1490 GOSUB 2170
1500 GOTO 1110
1510 END
1520 REM <<< PAPAGAIO LEVANTADO >>>
1530 LOCATE PX(A),PY(A):PRINT CHR$(200)
;CHR$(201);CHR$(202)
1540 LOCATE PX(A),PY(A)+1:PRINT CHR$(20
3);CHR$(204);CHR$(205)
1550 LOCATE PX(A),PY(A)+2:PRINT CHR$(20
6);CHR$(207);CHR$(208)
```

```

1560 LOCATE PX(A),PY(A)+3:PRINT CHR$(20
9);CHR$(210);CHR$(211)
1570 RETURN
1580 REM <<< PAPAGAIO ABAIXADO >>>
1590 PLAY "L8"+CHR$(67+A)
1600 LOCATE PX(A)-1,PY(A)+2:PRINT CHR$(
212);CHR$(213);CHR$(214);CHR$(215)
1610 LOCATE PX(A)-1,PY(A)+3:PRINT CHR$(
209);CHR$(210)
1620 LOCATE PX(A),PY(A):PRINT CHR$(216)
1630 RETURN
1640 REM <<< PAPAGAIO EM BRANCO >>>
1650 LOCATE PX(A),PY(A):PRINT " "
1660 LOCATE PX(A),PY(A)+1:PRINT " "
1670 LOCATE PX(A)-1,PY(A)+2:PRINT " "
1680 LOCATE PX(A)-1,PY(A)+3:PRINT " "
1690 RETURN
1700 REM <<< DEFINIR CARACTERES >>>
1710 RESTORE 1860
1720 FOR C=1 TO 18
1730 READ CC
1740 FOR A=0 TO 7
1750 READ NN
1760 VPOKE CC*8+A,NN
1770 NEXT A
1780 NEXT C
1790 VPOKE &H2019,48:VPOKE &H201A,48:VP
OKE &H201B,96
1800 RETURN
1810 REM <<< INICIALIZAR VARIAVEIS >>>
1820 PX(1)=10:PX(2)=14:PX(3)=6:PX(4)=3
1830 PY(1)=5:PY(2)=13:PY(3)=15:PY(4)=7
1840 RETURN
1850 REM <<< CARACTERES >>>
1860 DATA 200,16,16,16,16,0,14,31,63
1870 DATA 201,0,0,28,126,127,207,207,20
7


```

```
1880 DATA 202,0,0,0,0,0,0,128,128
1890 DATA 203,63,49,40,8,16,33,35,3
1900 DATA 204,255,255,254,124,126,255,2
55,255
1910 DATA 205,0,0,0,0,0,128,192,192
1920 DATA 206,3,3,3,3,3,1,1,1
1930 DATA 207,223,191,191,223,223,239,2
47,251
1940 DATA 208,192,192,224,224,224,240,2
40,248
1950 DATA 209,1,0,0,0,56,127,0,0
1960 DATA 210,253,126,12,24,48,252,0,0
1970 DATA 211,248,120,60,28,14,6,3,0
1980 DATA 212,7,31,63,127,99,99,127,63
1990 DATA 213,144,225,247,251,139,115,2
51,255
2000 DATA 214,0,254,255,255,255,255,255
,255
2010 DATA 215,3,14,252,252,248,240,224,
128
2020 DATA 216,16,16,16,16,16,16,16,16
2030 DATA 217,255,170,85,170,85,170,85,
255
2040 REM <<< TELA >>>
2050 FOR Y=1 TO 4:LOCATE 10,Y:PRINT CHR
$(216):NEXT Y
2060 FOR Y=1 TO 12:LOCATE 14,Y:PRINT CH
R$(216):NEXT Y
2070 FOR Y=1 TO 14:LOCATE 6,Y:PRINT CHR
$(216):NEXT Y
2080 FOR Y=1 TO 6:LOCATE 3,Y:PRINT CHR$(
216):NEXT Y
2090 FOR X=0 TO 28:LOCATE X,0:PRINT CHR
$(217):NEXT X
2100 LOCATE 10,9:PRINT CHR$(217);CHR$(2
17);CHR$(217)
2110 LOCATE 14,17:PRINT CHR$(217);CHR$(2
17);CHR$(217)
2120 LOCATE 6,19:PRINT CHR$(217);CHR$(2
17);CHR$(217)
```

```
2130 LOCATE 3,11:PRINT CHR$(217);CHR$(2  
17);CHR$(217)  
2140 LOCATE 7,21:PRINT "PAPAGAIOS"  
2150 LOCATE 7,22:PRINT "-----"  
2160 RETURN  
2170 FOR Y=10 TO 15  
2180 LOCATE 17,Y:PRINT STRING$(12," ")  
2190 NEXT Y  
2200 RETURN  
2210 REM <<< MOVIMENTAR PAPAGAIO >>>  
2220 A=NO(B):GOSUB 1650:GOSUB 1590  
2230 FOR T=1 TO 100:NEXT T  
2240 GOSUB 1650:GOSUB 1530  
2250 RETURN
```

19

Pic Man



O Devorador de Biscoitos

Este é o primeiro dos três jogos de fliperama incluídos neste livro. Trata-se de uma variação de um jogo bem conhecido, em que você guia o herói por um labirinto, comendo biscoitos no caminho.

Como nos jogos de fliperama as coisas nunca são fáceis, existem alguns fantasminhas que gostam tanto de biscoitos quanto o seu herói. Além disso, são capazes de matá-lo se o alcançarem! O objetivo é comer o máximo possível de biscoitos em cada quadro, sem ser morto por um dos fantasmas. Quando os biscoitos acabam, aparece um novo quadro.

Como sempre, você pode usar o joystick ou as teclas de controle do cursor para controlar o movimento do *Pic Man*.

```
1000 REM <<< PIC MAN >>>
1010 SCREEN 1,2,0:CLEAR 500
```

```

1020 COLOR 15,0,0:CLS:KEY OFF
1030 RESTORE:DIM S$(11):HS=10000
1040 ON SPRITE GOSUB 1920
1050 ON INTERVAL=35 GOSUB 1430
1060 GOSUB 2320:DEFUSR=342
1070 GOTO 1590
1080 REM <<< ENTRADA DE DADOS >>>
1090 FOR I%=0 TO 2
1100 ST% = STICK(I%):IF ST% THEN I% = 2
1110 NEXT I%:RETURN
1120 A=USR(0):IF=ASC(INPUT$(1))
1130 RETURN
1140 REM <<< CARACTERE EM X,Y >>>
1150 C% = 6144+INT(Y%/8)*32+X%/8:RETURN
1160 REM <<< CARACTERE ACIMA >>>
1170 Y% = Y%-1:GOTO 1200
1180 REM <<< CARACTERE ABAIXO >>>
1190 Y% = Y%+16
1200 GOSUB 1150:CT% = C%+1:CL% = VPEEK(C%):
CR% = VPEEK(CT%):GOTO 1260
1210 REM <<< CARACTERE A ESQUERDA >>>
1220 X% = X%-2:GOTO 1250
1230 REM <<< CARACTERE A DIREITA >>>
1240 X% = X%+17
1250 GOSUB 1150:CT% = C%+32:CL% = VPEEK(C%)
:CR% = VPEEK(CT%)
1260 IF CL% = 251 THEN VPOKE C%,32:DT% = DT
%+1:IF PM% THEN PS=PS+50:IF PS>10000 TH
EN HS=PS
1270 IF CR% = 251 THEN VPOKE CT%,32:DT% = DT
%+1:IF PM% THEN PS=PS+50:IF PS>10000 T
HEN HS=PS
1280 IF CL% = 250 OR CR% = 250 THEN RETURN
1290 C% = 0:RETURN
1300 REM <<< ESCOLHER A DIREÇÃO >>>
1310 H% = INT(RND(-TIME)*3-1)*4:IF H% THE
N V% = 0:RETURN ELSE V% = INT(RND(-TIME)*3-
1)*4:RETURN
1320 REM <<< POSICIONAR SPRITES >>>

```

```

1330 PUT SPRITE 1,(X1%,Y1%),2,PB%
1340 PUT SPRITE 2,(X2%,Y2%),4,PB%
1350 PUT SPRITE 3,(X3%,Y3%),13,PB%
1360 PUT SPRITE 0,(X0%,Y0%),10,P0%
1370 RETURN
1380 REM <<< IMPRIMIR A CONTAGEM >>>
1390 LOCATE 22,0:PRINT USING "#####";HS
1400 LOCATE 22,1:PRINT USING "#####";PS
1410 RETURN
1420 REM <<< MOVIMENTAR O PIC MAN >>>
1430 PLAY "t255116o1c"
1440 T0%=C%:T1%=CT%:T2%=X%:T3%=Y%:T4%=C
L%:T5%=CR%:PM%=1:GOSUB 1090
1450 IF ST%=1 OR ST%=5 THEN P0%=0:H0%=0
:IF ST%=1 THEN V0%=-8:S0%=2 ELSE V0%=8:
S0%=4
1460 IF ST%=3 OR ST%=7 THEN P0%=0:V0%=0
:IF ST%=3 THEN H0%=8:S0%=3 ELSE H0%=-8:
S0%=1
1470 X%=X0%:Y%=Y0%
1480 IF V0%<0 THEN GOSUB 1170
1490 IF V0%>0 THEN GOSUB 1190
1500 IF H0%>0 THEN GOSUB 1240
1510 IF H0%<0 THEN GOSUB 1220
1520 IF C% THEN 1540
1530 X0%=X0%+H0%:Y0%=Y0%+V0%
1540 P0%=P0% XOR S0%
1550 IF X0%>232 THEN X0%=8
1560 IF X0%<8 THEN X0%=232
1570 C%=T0%:CT%=T1%:X%=T2%:Y%=T3%:CL%=T
4%:CR%=T5%:PM%=0:GOSUB 1390:GOTO 1360
1580 REM <<< INICIO DO JOGO >>>
1590 GOSUB 2030
1600 X0%=120:Y0%=144:H0%=-8:V0%=0:S0%=1
1610 GOSUB 1980
1620 P0%=1:PB%=5:SPRITE ON:INTERVAL ON
1630 GOSUB 1330

```

```

1640 GOSUB 1120
1650 GOSUB 1330
1660 IF DT%>=352 THEN INTERVAL OFF:GOSUB
    2110:DT%=0:GOTO 1600
1670 PB%>PB% XOR 3
1680 REM <<< MOVIMENTAR OS FANTASMAS >
>>>
1690 X%>=X1%:Y%>=Y1%
1700 IF H1%<0 THEN GOSUB 1220:GOTO 1740
1710 IF H1%>0 THEN GOSUB 1240:GOTO 1740
1720 IF V1%<0 THEN GOSUB 1170:GOTO 1740
1730 IF V1%>0 THEN GOSUB 1190
1740 IF C% THEN GOSUB 1310:H1%>=H%:V1%>=V
%:GOTO 1690
1750 X1%>=X1%+H1%:Y1%>=Y1%+V1%
1760 X%>=X2%:Y%>=Y2%
1770 IF H2%<0 THEN GOSUB 1220:GOTO 1810
1780 IF H2%>0 THEN GOSUB 1240:GOTO 1810
1790 IF V2%<0 THEN GOSUB 1170:GOTO 1810
1800 IF V2%>0 THEN GOSUB 1190
1810 IF C% THEN GOSUB 1310:H2%>=H%:V2%>=V
%:GOTO 1760
1820 X2%>=X2%+H2%:Y2%>=Y2%+V2%
1830 X%>=X3%:Y%>=Y3%
1840 IF H3%<0 THEN GOSUB 1220:GOTO 1880
1850 IF H3%>0 THEN GOSUB 1240:GOTO 1880
1860 IF V3%<0 THEN GOSUB 1170:GOTO 1880
1870 IF V3%>0 THEN GOSUB 1190
1880 IF C% THEN GOSUB 1310:H3%>=H%:V3%>=V
%:GOTO 1830
1890 X3%>=X3%+H3%:Y3%>=Y3%+V3%
1900 GOTO 1650
1910 REM <<< PERDE UMA VIDA >>>
1920 SPRITE OFF:INTERVAL OFF
1930 IF ABS(X0%-X1%)>15 AND ABS(X0%-X2%)
    >15 AND ABS(X0%-X3%)>15 OR ABS(Y0%-Y1%)
    >15 AND ABS(Y0%-Y2%)>15 AND ABS(Y0%-Y3%)
    >15 THEN GOSUB 1980:GOSUB 1330:SPRITE
    ON:INTERVAL ON:RETURN

```

```

1940 PLAY "t255146o3bagfedc"
1950 LF=LF-1:IF LF=0 THEN FS=0:GOSUB 14
00:RETURN 1590
1960 PUT SPRITE LF+3,(LF*16,0),0,3
1970 RETURN 1600
1980 X1%=24:Y1%=24:H1%=8:V1%=0
1990 X2%=216:Y2%=24:H2%=-8:V2%=0
2000 X3%=120:Y3%=168:H3%=8:V3%=0:IF RND
(-TIME)>.5 THEN H3%=-H3%
2010 RETURN
2020 REM <<< INICIALIZAR >>>
2030 GOSUB 2110:LF=5:DT%=0
2040 FOR I=4 TO 7:J=(I-3)*16
2050 PUT SPRITE I,(J,0),10,3
2060 NEXT I
2070 LOCATE 14,0:PRINT "RECORDE:";
2080 LOCATE 14,1:PRINT "PONTOS:";
2090 RETURN
2100 REM <<< DESENHAR LABIRINTO >>>
2110 WIDTH 29:LOCATE 0,2
2120 PRINT S$(0)
2130 PRINT S$(1):PRINT S$(1)
2140 PRINT S$(2):PRINT S$(2)
2150 PRINT S$(3):PRINT S$(3)
2160 PRINT S$(4):PRINT S$(4)
2170 PRINT S$(5):PRINT S$(6)
2180 PRINT S$(7):PRINT S$(8)
2190 PRINT S$(9):PRINT S$(9)
2200 PRINT S$(2)
2210 PRINT S$(3):PRINT S$(3)
2220 PRINT S$(10)
2230 PRINT S$(11):PRINT S$(11)
2240 PRINT S$(0);
2250 RETURN
2260 REM <<< DEFINIR AS FORMAS >>>
2270 SP$=""
2280 FOR J=1 TO 32
2290 READ H$:SP$=SP$+CHR$(VAL("&h"+H$))
2300 NEXT J

```

```

2310 RETURN
2320 FOR I=250 TO 251:K=I*8
2330 FOR J=0 TO 7
2340 READ H$:VPOKE J+K,VAL("&h"+H$)
2350 NEXT J,I:VPOKE 8223,96
2360 FOR I=0 TO 6
2370 GOSUB 2270:SPRITE$(I)=SP$
2380 NEXT I
2390 S1$=CHR$(250):S2$=CHR$(251)
2400 S3$=S1$+S1$:S4$=S2$+S2$
2410 S5$=S1$+S3$:S6$=S2$+S4$
2420 S7$=S3$+S3$:S8$=S4$+S4$
2430 S9$=S3$+S7$:S0$=S4$+S8$
2440 S$(0)=STRING$(28,S1$)
2450 S$(1)=S1$+S0$+S3$+S0$+S0$+S3$+S0$+
S1$
2460 S$(2)=S1$+S4$+S3$+S4$+S3$+S4$+S9$+
S4$+S3$+S4$+S3$+S4$+S1$
2470 S$(3)=S1$+S4$+S3$+S0$+S0$+S0$+S3$+
S4$+S1$
2480 S$(4)=S1$+S0$+S3$+S4$+S9$+S4$+S3$+
S0$+S1$
2490 S$(5)=S1$+S7$+S4$+S3$+S0$+S0$+S3$+
S4$+S1$+S7$
2500 S$(6)=SPACE$(1)+S0$+S3$+S8$+S0$+S3
$+S0$+SPACE$(1)
2510 S$(7)=SPACE$(1)+S0$+S3$+S4$+S9$+S4
$+S3$+S0$+SPACE$(1)
2520 S$(8)=S1$+S7$+S4$+S3$+S4$+S1$+SPAC
E$(4)+S1$+S4$+S3$+S4$+S7$+S1$
2530 S$(9)=S1$+S0$+S3$+S4$+S1$+SPACE$(4
)+S1$+S4$+S3$+S0$+S1$
2540 S$(10)=S1$+S4$+S7$+S4$+S7$+S9$+S4$+
S7$+S4$+S1$
2550 S$(11)=S1$+STRING$(26,S2$)+S1$
2560 RETURN
2570 REM <<< DADOS >>>
2580 DATA ff,c3,c3,c3,c3,c3,c3,ff
2590 DATA B1,0,0,0,0,0,0,B1

```


```

2600 REM
2610 DATA 3,f,3f,3f,7f,7f,ff,fe
2620 DATA fe,ff,7f,7f,3f,3f,f,3
2630 DATA c0,f0,fc,fc,fe,fe,ff,7f
2640 DATA 7f,ff,fe,fe,fc,fc,f0,c0
2650 DATA 3,f,3f,3f,1f,f,7,2
2660 DATA 2,7,f,1f,3f,3f,f,3
2670 DATA c0,f0,fc,fc,fe,fe,ff,7f
2680 DATA 7f,ff,fe,fe,fc,fc,f0,c0
2690 DATA 0,0,30,38,7c,7e,ff,fe
2700 DATA fe,ff,7f,7f,3f,3f,f,3
2710 DATA 0,0,c,1c,3e,7e,ff,7f
2720 DATA 7f,ff,fe,fe,fc,fc,f0,c0
2730 DATA 3,f,3f,3f,7f,7f,ff,fe
2740 DATA fe,ff,7f,7f,3f,3f,f,3
2750 DATA c0,f0,fc,fc,f8,f0,e0,40
2760 DATA 40,e0,f0,f8,fc,fc,f0,c0
2770 DATA 3,f,3f,3f,7f,7f,ff,fe
2780 DATA fe,ff,7e,7c,38,30,0,0
2790 DATA c0,f0,fc,fc,fe,fe,ff,7f
2800 DATA 7f,ff,7e,3e,1c,18,0,0
2810 DATA 0,18,25,47,f,f,9,9
2820 DATA 1f,1f,1f,3f,3f,7f,67,43
2830 DATA 0,18,64,e2,f0,f0,90,90
2840 DATA f8,f8,f8,fc,fc,fe,e6,c2
2850 DATA e0,18,5,7,f,f,9,9
2860 DATA 1f,1f,1f,3f,3f,7e,3c,18
2870 DATA 7,18,a0,e0,f0,f0,90,90
2880 DATA f8,f8,f8,fc,fc,7e,3c,18
2890 REM      ,
2900 DATA ff,c3,c3,c3,c3,c3,ff
2910 DATA 81,0,0,0,0,0,0,81

```

20

Frigger



Travessia Perigosa

Era uma vez, na Austrália, um pequeno canguru chamado Frigger. Frigger morava perto de um rio caudaloso, cheio de troncos flutuantes. Costumava ficar sentado do lado de fora da toca, contemplando a bela paisagem que o cercava. Um dia, porém, tudo mudou. Os humanos chegaram e construíram uma auto-estrada do outro lado do rio. A partir desse dia, sempre que Frigger e os amigos saem em busca de comida, têm que atravessar a estrada (evitando os carros) e o rio (pulando de tronco em tronco) para chegar em casa.

Só você, caro leitor, pode ajudar Frigger e seus amigos a voltarem para casa em segurança. Use o joystick ou as teclas de controle do cursor para guiar os cangurus e será recompensado com muitos pontos!

```

1000 REM <<< FRIGGER - ANDY/BOOTSY >>>
>
1010 SCREEN 1,0,0:CLEAR 1800
1020 COLOR 15,0,0:KEY OFF
1030 WIDTH 32:H5%=5:GOSUB 1670
1040 KA%=3:SC%=0
1050 GOSUB 1600:GOSUB 1540
1060 KX%=128:KY%=160:K%=1
1070 PUT SPRITE 0,(KX%,KY%),15,K%
1080 REM <<< LOOP PRINCIPAL >>>
1090 GOSUB 1350:IF (K% AND 1)=0 THEN GO
SUB 1480:GOTO 1090
1100 CH%=VPEEK(6144+(KX%+4)/8+INT((KY%+
4)/8)*32)
1110 IF CH%<>32 AND CH%<>240 AND (CH%<2
08 OR CH%>210) THEN 1170
1120 IF KY%=16 THEN 1290
1130 GOSUB 1310:IF ST% THEN GOSUB 1470
1140 IF CH%>207 AND CH%<211 THEN KX%=KX
%+(8 AND KY%=32)-(8 AND KY%=48):GOTO 10
70
1150 GOTO 1090
1160 REM <<< PERDE UMA VIDA >>>
1170 KA%=KA%-1:PLAY "o1164bagfedc"
1180 IF KA% THEN GOTO 1050
1190 PUT SPRITE 0,(0,208),0,0
1200 LOCATE 2,14
1210 PRINT "SUAS VIDAS ACABARAM !"
1220 LOCATE 2,16
1230 PRINT "APERTE O BOTAO OU A BARRA D
E"
1240 LOCATE 2,18
1250 PRINT "ESPAÇO PARA COMEÇAR DE NOVO
"
1260 GOSUB 1310:IF TG%=0 THEN 1260
1270 GOTO 1040
1280 REM <<< CONSEGUIU >>>
1290 PLAY "o1164cdefgab02cdefgab03cdefg
abo4cdefgab":SC%=SC%+1:GOTO 1050

```

```

1300 REM <<< ENTRADA DE DADOS >>>
1310 FOR I%=0 TO 2:ST%=$TICK(I%):TG%=$T
RIG(I%)
1320 IF ST% OR TG% THEN I%=2
1330 NEXT I%:RETURN
1340 REM <<< MOVIMENTO >>>
1350 L2$=RIGHT$(L2$,1)+LEFT$(L2$,31)
1360 L1$=RIGHT$(L1$,31)+LEFT$(L1$,1)
1370 C3$=RIGHT$(C3$,1)+LEFT$(C3$,31)
1380 C2$=RIGHT$(C2$,31)+LEFT$(C2$,1)
1390 C1$=RIGHT$(C1$,1)+LEFT$(C1$,31)
1400 LOCATE 0,4:PRINT L2$;
1410 LOCATE 0,6:PRINT L1$;
1420 LOCATE 0,14:PRINT C3$;
1430 LOCATE 0,16:PRINT C2$;
1440 LOCATE 0,18:PRINT C1$;
1450 RETURN
1460 REM <<< MOVIMENTO DO CANGURU >>>
1470 K%=$T%:M%=$T%
1480 K% =K% XOR 1
1490 KX% =KX% +(8 AND M% =3)-(8 AND M% =7)
1500 KY% =KY% +(8 AND M% =5)-(8 AND M% =1)
1510 PUT SPRITE 0,(KX%,KY%),15,K%
1520 RETURN
1530 REM <<< CONTAGEM, ETC. >>>
1540 IF SC%>HS% THEN HS% =SC%
1550 LOCATE 7,0:PRINT SC%
1560 LOCATE 16,0:PRINT HS%
1570 LOCATE 27,0:PRINT KAX%
1580 RETURN
1590 REM <<< TELA >>>
1600 LOCATE 0,1
1610 PRINT B1$;RV$;B2$
1620 LOCATE 0,20:PRINT B3$;B4$;
1630 LOCATE 1,0
1640 PRINT " PTOS:      REC:      VIDAS: "
1650 GOTO 1400
1660 REM <<< FORMAS >>>
1670 CX% =192

```

```

1680 READ I%:IF I%<0 THEN 1750
1690 FOR J%=0 TO 7:VPOKE C%*8+J%,I%
1700 READ I%:NEXT J%
1710 VPOKE 8192+C%/8,I%
1720 C%=C%+1
1730 IF C%=194 OR C%=203 OR C%=211 OR C%
    %=217 OR C%=226 OR C%=233 OR C%=241 THE
    N C%=(C%+7) AND 248
1740 GOTO 1680
1750 FOR I%=0 TO 7
1760 S$="" :FOR J%=0 TO 7:READ C%
1770 S$=S$+CHR$(C%):NEXT J%
1780 SPRITE$(I%)=S$ :NEXT I%
1790 CR$=CHR$(192)+CHR$(193)
1800 LY$=CHR$(202)+CHR$(201)+CHR$(200)
1810 BK$=CHR$(224)+CHR$(225)
1820 LG$=CHR$(208)+CHR$(209)+CHR$(210)
1830 C1$=CR$+CR$+SPACE$(8)+CR$+SPACE$(4)
    )+CR$+CR$+CR$+SPACE$(2)+CR$+SPACE$(2)+C
    R$
1840 C2$=LY$+SPACE$(12)+LY$+SPACE$(3)+L
    Y$+LY$+SPACE$(5)
1850 C3$=BK$+BK$+SPACE$(8)+BK$+BK$+BK$+
    BK$+BK$+SPACE$(8)+BK$
1860 L1$=LG$+STRING$(3,CHR$(248))+LG$+S
    TRING$(3,CHR$(248))+STRING$(3,CHR$(216))
    )+STRING$(2,CHR$(248))+LG$+STRING$(3,CH
    R$(248))+LG$+STRING$(2,CHR$(248))+STRIN
    G$(2,CHR$(216))+STRING$(2,CHR$(248))
1870 L2$=LG$+STRING$(3,CHR$(248))+LG$+S
    TRING$(3,CHR$(248))+STRING$(3,CHR$(216))
    )+STRING$(3,CHR$(248))+LG$+STRING$(3,CH
    R$(248))+STRING$(4,CHR$(216))+STRING$(2
    ,CHR$(248))
1880 RV$=STRING$(160,CHR$(248))
1890 B1$=STRING$(32,CHR$(232))
1900 FOR I%=1 TO 16 STEP 4
1910 B1$=B1$+STRING$(4,CHR$(232))+SPACE
    $(1)+STRING$(3,CHR$(232))


```

```
1920 NEXT I%
```

```
1930 BN$=LEFT$(BN$,63):BN$=BN$+CHR$(232)
)
1940 B2$=STRING$(160,CHR$(240))
1950 B3$=STRING$(64,CHR$(240))
1960 B4$=STRING$(32,CHR$(232))
1970 RETURN
1980 REM <<< DADOS DAS FORMAS >>>
1990 DATA 7,28,56,127,219,231,36,24,208
2000 DATA 128,64,32,252,218,229,38,24,2
08
2010 DATA 0,0,0,0,255,255,60,24,96
2020 DATA 64,64,64,64,255,255,128,0,96
2030 DATA 7,9,9,63,127,127,7,3,96
2040 DATA 191,109,182,109,218,125,171,9
5,196
2050 DATA 255,181,235,182,47,254,93,190
,196
2060 DATA 250,213,190,91,182,109,182,25
3,196
2070 DATA 0,0,67,172,240,192,255,0,20
2080 DATA 0,1,51,14,55,72,72,48,176
2090 DATA 96,200,244,12,154,164,228,24,
176
2100 DATA 255,255,255,255,255,255,255,2
55,192
2110 DATA 0,127,127,127,127,127,127,127
,160
2120 DATA 170,85,170,85,170,85,170,85,8
4
2130 DATA -1
2140 REM <<< DADOS DOS SPRITES >>>
2150 DATA 36,24,90,60,24,60,60,219
2160 DATA 36,24,24,60,90,189,126,36
2170 DATA 8,24,28,24,60,61,58,220
2180 DATA 4,12,14,24,60,188,184,94
2190 DATA 219,60,60,24,60,90,24,36
2200 DATA 36,126,189,90,60,24,24,36
2210 DATA 16,24,56,24,60,188,92,59
2220 DATA 32,48,112,24,60,61,29,122
```

21

Dinkey Kong



Salvando a Mocinha

Dinkey Kong pertence à família de jogos de flipperama em que o herói tem que chegar ao topo de um edifício, evitando uma série de obstáculos que aparecem no caminho.

Nesta versão, o monstro que aparece na parte superior da tela raptou a namorada do herói e a mantém prisioneira. Agora ele passa o tempo todo rolando grandes tonéis de aço, em uma tentativa de esmagar o herói.

O jogo tem três telas diferentes; se você conseguir chegar ao final, a primeira tela voltará a aparecer.

Os movimentos do herói são controlados com o auxílio do joystick ou das teclas de controle do cursor. Para pular, é só apertar o botão de tiro ou a barra de espaço.

```

1000 REM <<< DINKEY KONG- BOOTSY >>>
1010 SCREEN 1,2,0:COLOR 15,0,0:CLS
1020 KEY OFF:GOSUB 2140:HS% =1000
1030 SC% =0:LF% =3:PS% =0
1040 XP% =20:YP% =160
1050 GOSUB 1910:SB% =4:SM% =0
1060 IF YP% =32 THEN GOSUB 1860:SC% =(SC%
+1) AND 3:GOSUB 1040
1070 GOSUB 1400:IF C% THEN 1180
1080 X% =XP%:Y% =YP%
1090 IF TG% THEN GOSUB 1550:GOTO 1180
1100 IF ST% =1 THEN GOSUB 1500 ELSE GOSU
B 1490
1110 IF ST% =1 AND C% =0 AND CH% =240 THEN
UD% =-2:GOSUB 1630:GOTO 1180
1120 IF ST% =5 AND C% =0 AND CH% =240 THEN
UD% =2:GOSUB 1630:GOTO 1180
1130 IF ST% =3 AND CH% =32 THEN 1670
1140 IF ST% =3 THEN XP% =XP% +4:D% =2:IF SM
% AND 2 THEN SM% =0
1150 IF ST% =7 AND C% =32 THEN 1670
1160 IF ST% =7 THEN XP% =XP% -4:D% =-2:IF (
SM% AND 2)=0 THEN SM% =2
1170 SM% =SM% XOR 1:PUT SPRITE 0,(XP%,YP
%),15,SM%
1180 L% =0:FOR I% =0 TO 3:J% =XP%-BA%(I%,0
):K% =YP%-BA%(I%,1)
1190 IF ABS(J%)<12 AND ABS(K%)<12 THEN
L% =1:I% =3
1200 NEXT I%:IF L% THEN 1670
→ 1210 IF B% <3 THEN IF RND(-TIME)>.95 THE
N B% =B% +1:BA%(B%,0) =128:BA%(B%,1) =32
1220 IF B% =0 THEN 1060
→ 1230 SB% =SB% XOR 1:I% =1
1240 J% =BA%(I%,0):K% =BA%(I%,1)
→ 1250 IF K% =32 OR K% =96 OR K% =160 THEN J
% =J%-4 ELSE J% =J% +4
1260 IF K% <32 THEN 1280
1270 PUT SPRITE I%,(J%,K%),12,SB%

```

```

1280 X% = J% : Y% = K% : GOSUB 1490
1290 IF C% = 0 AND CH% = 32 THEN GOSUB 1820
: GOSUB 1310
1300 IF C% = 0 AND CH% = 240 THEN IF RND(-T
IME) > .66 THEN GOSUB 1820
1310 IF J% = 16 AND K% = 160 THEN 1350
1320 BAX(I%, 0) = J% : BAX(I%, 1) = K%
1330 I% = I% + 1 : IF I% > B% THEN 1060
1340 GOTO 1240
1350 FOR L% = I% TO B% : J% = BAX(L% + 1, 0) : BAX(
L%, 0) = J% : K% = BAX(L% + 1, 1) : BAX(L%, 1) = K% : I
F K% = 0 THEN 1370
1360 PUT SPRITE L%, (J%, K%), 12, SB%
1370 NEXT L%
1380 I% = I% - 1 : GOTO 1330
1390 REM <<< ENTRADA DE DADOS >>>
1400 C% = 1 : FOR I% = 0 TO 2 : ST% = STICK(I%) : T
G% = STRIG(I%)
1410 IF ST% OR TG% THEN I% = 2 : C% = 0
1420 NEXT I% : IF ST% = 5 THEN 1400
1430 RETURN
1440 A = USR1(0) : I$ = INPUT$(1)
1450 RETURN
1460 REM <<< CARACTERE EM X, Y >>>
1470 CH% = VPEEK(&H1800 + INT(Y%/8) * 32 + X%/8
) : RETURN
1480 REM <<< COM APOIO/SEM APOIO >>>
1490 Y% = Y% + 8
1500 X% = X% + 4 : Y% = Y% + 8 : GOSUB 1470 : C% = CH%
1510 X% = X% + 7 : GOSUB 1470
1520 IF C% <> CH% THEN RETURN
1530 C% = 0 : RETURN
1540 REM <<< PULAR >>>
1550 RESTORE 2640 : FOR I% = 1 TO 12
1560 READ VO% : XP% = XP% + D% : YP% = YP% + VO%
1570 PUT SPRITE 0, (XP%, YP%), 15, 0
1580 NEXT I%
1590 X% = XP% + 7 : Y% = YP% + 16
1600 GOSUB 1470 : IF CH% = 32 THEN 1670

```

```

1610 X%=XP%-7:Y%=YP%-16:RETURN
1620 REM <<< SUBIR ESCADA >>>
1630 FOR I%=1 TO 16:YP%=YP%+UD%
1640 PUT SPRITE 0,(XP%,YP%),15,0
1650 NEXT I%:RETURN
1660 REM <<< CAIR >>>
1670 PLAY "L6402BAGFEDC01BAGFEDC"
1680 FOR I%=YP% TO 192 STEP 2
1690 PUT SPRITE 0,(XP%,I%),15,SP%
1700 NEXT I%
1710 REM <<< PERDE VIDA >>>
1720 LF%=LF%-1:IF LF% THEN 1040
1730 LOCATE 4,10:PRINT "SUAS VIDAS ACA
BARAM"
1740 LOCATE 4,14:PRINT " APERTE O BOT
AO "
1750 PRINT SPACE$(22)
1760 LOCATE 4,16:PRINT "OU A BARRA DE E
SPAÇO "
1770 PRINT SPACE$(22)
1780 LOCATE 1,18:PRINT " PARA JOGAR
DE NOVO "
1790 GOSUB 1400:IF TG%=0 THEN 1790
1800 GOTO 1030
1810 REM <<< QUEDA DO BARRIL >>>
1820 FOR L%=1 TO 16:K%=K%+2
1830 PUT SPRITE I%,(J%,K%),12,SB%
1840 NEXT L%:RETURN
1850 REM <<< CONTAGEM >>>
1860 PS%=PS%+500
1870 IF PS%>HS% THEN HS%=PS%
1880 PLAY "L6404CDEFGAB05CBEFGAB06CDEFG
AB"
1890 RETURN
1900 REM <<< TELA >>>
1910 CLS:FOR I%=0 TO 4
1920 LOCATE 8-I%*2,I%*4+6
1930 PRINT STRING$(I%*4+12,CHR$(248))
1940 NEXT I%

```

```

1950 SL% = 10 - SC% * 2 : SH% = SC% * 2
1960 RESTORE 2620 : FOR I% = 1 TO SL%
1970 READ LX%, LY% : FOR J% = 1 TO 4
1980 LOCATE LX%, LY% : PRINT CHR$(240)
1990 LY% = LY% + 1 : NEXT J%, I%
2000 IF SH% = 0 THEN 2040
2010 FOR I% = 1 TO SH%
2020 READ LX%, LY% : LOCATE LX%, LY%
2030 PRINT SPC(2) : NEXT I%
2040 PUT SPRITE 0, (XP%, YP%), 15, 0
2050 C% = 224 : FOR I% = 2 TO 5 : FOR J% = 13 TO
15
2060 LOCATE J%, I% : PRINT CHR$(C%) ; : C% = C%
+ 1
2070 NEXT J%, I%
2080 LOCATE 4, 1 : PRINT PS% : LOCATE 1, 1 : PR
INT "PTS:"
2090 LOCATE 14, 1 : PRINT HS% : LOCATE 11, 1 :
PRINT "REC:"
2100 LOCATE 21, 1 : PRINT "VIDAS:" ; LF%
2110 FOR I% = 0 TO 4 : BA%(I%, 0) = 108 : NEXT I%
2120 RETURN
2130 REM <<< FORMAS >>>
2140 RESTORE
2150 FOR C% = 224 TO 235
2160 GOSUB 2240
2170 NEXT C%
2180 C% = 240 : GOSUB 2240
2190 C% = 248 : GOSUB 2240
2200 FOR S% = 0 TO 5
2210 GOSUB 2280
2220 NEXT S%
2230 RETURN
2240 CA% = C% * 8 : FOR I% = CA% TO CA% + 7
2250 READ J% : VPOKE I%, J% : NEXT I%
2260 READ J% : VPOKE C% / 8 + 8192, J%
2270 RETURN
2280 S$ = "" : FOR I% = 0 TO 31


```

```
2290 READ J%:S$=S$+CHR$(J%):NEXT I%
2300 SPRITE$(5%)=S$:RETURN
2310 REM <<< DINKEY KONG >>>
2320 DATA 0,0,16,44,62,28,56,112,96
2330 DATA 129,66,126,90,126,106,86,60,9
6
2340 DATA 0,0,8,52,124,56,28,14,96
2350 DATA 112,121,127,63,31,7,1,1,96
2360 DATA 60,255,255,255,231,215,235,21
3,96
2370 DATA 14,158,254,252,248,224,128,12
8,96
2380 DATA 1,3,15,15,31,63,62,62,96
2390 DATA 171,213,235,215,235,255,60,0,
96
2400 DATA 128,192,240,240,248,252,124,1
24,96
2410 DATA 30,15,15,3,59,127,127,57,96
2420 DATA 0,0,0,0,129,195,195,129,96
2430 DATA 120,240,240,192,220,254,254,1
56,96
2440 REM <<< CENARIO >>>
2450 DATA 129,255,129,129,129,255,129,1
29,128
2460 DATA 255,36,66,129,129,66,36,255,1
60
2470 REM <<< HOMEM >>>
2480 DATA 3,7,7,15,7,3,7,15,31,55,7,7,1
4,29,25,12
2490 DATA 128,192,224,128,192,128,192,2
24,240,176,128,128,128,128,192,224
2500 DATA 3,7,7,15,7,3,7,6,7,15,15,7,7,
7,13,14
2510 DATA 128,192,224,128,192,128,128,1
28,0,192,224,128,128,128,192,224
2520 DATA 1,3,7,1,3,1,3,7,15,13,1,1,1,1
,3,7
2530 DATA 192,224,224,240,224,192,192,2
40,248,236,224,224,112,184,152,48
```

```
2540 DATA 1,3,7,1,3,1,1,1,0,3,7,1,1,1,1,3
      ,7
2550 DATA 192,224,224,240,224,192,224,9
6,224,240,240,224,224,224,176,112
2560 REM <<< BARRIL >>>
2570 DATA 0,0,0,0,0,0,3,7,12,24,24,25,2
6,12,7,3
2580 DATA 0,0,0,0,0,0,192,224,48,88,152
,24,24,48,224,192
2590 DATA 0,0,0,0,0,0,3,7,12,26,25,24,2
4,12,7,3
2600 DATA 0,0,0,0,0,0,192,224,48,24,24,
152,88,48,224,192
2610 REM <<< ESCADA >>>
2620 DATA 10,6,19,10,6,14,23,18,4,18,21
,14,8,10,17,6,14,18,12,14
2630 REM <<< SALTO >>>
2640 DATA -5,-4,-3,-2,-1,0,0,1,2,3,4,5
2650 A$=INPUT$(1):RETURN
```

22

Banho de Palavras



Decifra-me ou eu te Molho!

Este jogo é uma versão mais divertida, menos cruel, do conhecido jogo da forca. Se o jogador não consegue descobrir a palavra depois de um número razoável de tentativas, as próprias letras erradas o empurram para dentro d'água!

As palavras podem ter até oito letras e não podem conter hifens. O programa original tem 60 palavras, mas não é difícil ampliar a lista ou trocar as palavras por outras. Quando o jogador acerta uma letra, ela aparece na posição que ocupa na palavra. Quando a letra está errada, o jogador é empurrado na direção da piscina. Quando a palavra é completada a tempo, o jogador volta à posição inicial e o jogo começa de novo com outra palavra.

Quando o jogador não consegue descobrir a palavra, ela não é mos-

trada. Como as palavras são escolhidas ao acaso, a mesma palavra pode voltar a aparecer mais tarde.

```

1000 REM <<< BANHO DE PALAVRAS - ISSI
>>>
1010 MAXFILES=2:OPEN "GRP:S" FOR OUTPUT
    AS 1
1020 GOSUB 1470
1030 W=INT(RND(-TIME)*60)+1
1040 W$=D$(W)
1050 GOSUB 1280
1060 GOSUB 1680
1070 FLAG=0
1080 IF T$=W$ THEN GOTO 1150
1090 IF LEN(G$)=10 THEN GOTO 1120
1100 GOTO 1060
1110 REM <<< PERDE >>>
1120 GOSUB 1520
1130 GOTO 1250
1140 REM <<< GANHA >>>
1150 OCT=8
1160 IF G$="" THEN GOTO 1230
1170 FOR N=LEN(G$) TO 1 STEP -1
1180 G$=RIGHT$(G$,N-1)
1190 M$="S11L640"+STR$(OCT)+"C":PLAY M$
1200 GOSUB 1810:PRINT " "
1210 FOR W=1 TO 250:NEXT W
1220 NEXT N
1230 PRESET (98,24):PRINT #1," PARABENS
!"
1240 PRESET (98,32):PRINT #1,"-----
"
1250 A$=INKEY$:IF A$="" THEN 1250
1260 BEEP:GOTO 1030
1270 REM <<< TELA >>>
1280 SCREEN 2,0,0:KEY OFF
1290 COLOR 15,1,1:CLS
1300 PRESET (74,0):PRINT #1,"BANHO DE P
ALAVRAS":COLOR 3

```

```

1310 A$="" :FOR N=1 TO 3:A$=A$+CHR$(&HCB
) +CHR$(&HCC):NEXT N
1320 PRESET (76,10):PRINT #1,A$
1330 PRESET (24,56):COLOR 13:O$=CHR$(1)
+CHR$(&HCC):PRINT #1,O$
1340 LINE (128,100)-(208,160),4,BF
1350 LINE (120,88)-(128,100),4
1360 LINE (216,88)-(208,100),4
1370 LINE (0,65)-(104,65),14
1380 LINE -STEP(20,20),14
1390 COLOR 11:PRESET (10,80):PRINT #1,"PALAVRA:"
1400 PRESET (40,120):T$="":G$=""
1410 COLOR 5:FOR N=1 TO LEN(W$)
1420 PRINT #1,"-":T$=T$+" "
1430 NEXT N
1440 PRESET (8,152):PRINT #1,"LETRA ?"
1450 RETURN
1460 REM <<< INICIALIZAÇÃO >>>
1470 RESTORE 1870
1480 DIM D$(60)
1490 FOR N=1 TO 60:READ D$(N):NEXT N
1500 RETURN
1510 REM <<< TCHIBUM! >>>
1520 FOR N=106 TO 152 STEP 2
1530 COLOR 1:PRESET (N-2,N-50)
1540 PRINT #1,CHR$(1)+CHR$(&HCB)
1550 COLOR 9:PRESET (N,N-48)
1560 PRINT #1,CHR$(1)+CHR$(&HCB):
1570 NEXT N
1580 RESTORE 1660
1590 FOR N=6 TO 13:READ A:SOUND N,A:NEX
T N
1600 LINE -STEP(-8,0),4
1610 FOR I=1 TO 15:J=-RND(-TIME)*20:K=R
ND(-TIME)*40-20
1620 LINE -STEP(K,J),4:LINE -STEP(-K,-J
),4
1630 NEXT I
1640 FOR W=1 TO 500:NEXT W

```


```

1650 RETURN
1660 DATA 15,135,16,16,16,0,16,0
1670 REM <<< ENTRADA DE DADOS >>>
1680 A$=INKEY$: IF A$="" THEN GOTO 1680
1690 A$=CHR$(ASC(A$) AND 223):PLAY "04S
11L16C"
1700 IF A$<"A" OR A$>"Z" THEN GOTO 1680
1710 FL=0:COLOR 9
1720 FOR N=1 TO LEN(W$)
1730 IF MID$(W$,N,1)=A$ THEN PRESET ((N
+4)*8,112):PRINT #1,A$:MID$(T$,N,1)=A$:
FL=1
1740 NEXT N
1750 IF FL=1 THEN PLAY "GAB":RETURN
1760 FL=0:FOR M=1 TO LEN(G$)
1770 IF MID$(G$,M,1)=A$ THEN FL=1
1780 NEXT M
1790 IF FL=1 THEN RETURN
1800 G$=A$+G$
1810 PRESET (24,56):COLOR 1:PRINT #1,O$
1820 PRESET (24,56):COLOR 13
1830 IF LEN(G$)/2=INT(LEN(G$)/2) THEN M
=&HCB ELSE M=&HCC
1840 O$=G$+CHR$(1)+CHR$(M):PRINT #1,O$;
1850 RETURN
1860 REM <<< DADOS >>>
1870 DATA ZUMBI,REPTO,BEZERRO,LIBELO,AS
CO,GUAPO,COLDRE,MURRO,HIATO,JOVEM
1880 DATA QUARTEL,MASTIM,BIGORNA,HONEST
O,ROBALO,GRAVIOLA,FICHA,JARRA,COLAPSO,P
OSTE
1890 DATA FEIXE,BILHAR,ANIMAL,SORTE,PIA
DA,MIASMA,BRUSCO,CARPA,HORTO,SUNGA
1900 DATA BOLDO,POTRO,FELIZ,ISTMO,CRESP
O,PLANTA,ANZOL,TURVO,CRIVO,DENGUE
1910 DATA GORDO,GRANEL,LERDO,ALGOZ,TRIA
GEM,GRAVAME,MAXIXE,CHUCHU,NABABO,OSTRA
1920 DATA TRAVA,VESTAL,BESTIAL,REPOUSO,
RAINHA,ALTAR,MODESTO,QUIABO,BOSQUE,LESMA

```

23

Aritmética



Contas e Mais Contas

Este programa pode gerar um número muito grande de problemas de aritmética. O próprio jogador seleciona o grau de dificuldade (fácil, médio, difícil) e a operação desejada (soma, subtração, multiplicação ou divisão). As perguntas e respostas aparecem em um quadro-negro.

No nível fácil, os problemas são todos de soma; o jogador tem que calcular o número total de "carinhás" que aparecem na tela. No nível médio de dificuldade, o jogador pode escolher uma das quatro operações e todas as respostas são números inteiros. Essas contas não são difíceis de fazer de cabeça. A maioria das pessoas só consegue resolver os problemas do nível mais difícil usando lápis e papel (ou, naturalmente, uma calculadora).

Para responder às perguntas, é preciso entrar com o resultado e aper-

tar a tecla **RETURN**. Isto quer dizer que se você descobrir que entrou com o número errado, ainda pode corrigi-lo, contanto que ainda não tenha apertado a tecla **RETURN**. O programa aceita até duas respostas erradas antes de revelar o resultado correto.

Depois de dez perguntas, o computador mostra na tela um “boletim” com o número de respostas corretas e a porcentagem de acertos.

```

1000 REM <<< ARITMETICA - ISSI >>>
1010 SCREEN 2,0,0:KEY OFF
1020 MAXFILES=2:OPEN "GRP:S" FOR OUTPUT
    AS 1
1030 GOSUB 1120
1040 COLOR 3:PRESET (75,60):PRINT #1,"1
> FACIL"
1050 COLOR 9:PRESET (75,80):PRINT #1,"2
> MEDIO"
1060 COLOR 11:PRESET (75,100):PRINT #1,
"3> DIFICIL"
1070 A$=INPUT$(1):A=ASC(A$)
1080 IF A<49 OR A>51 THEN GOTO 1070
1090 PLAY "L64CDECDE"
1100 LEV=A-48
1110 GOTO 1820
1120 COLOR 15,4,1:CLS
1130 COLOR 9:LINE (120,40)-(140,180)
1140 LINE (140,180)-(150,180):LINE(150,
180)-(130,30)
1150 LINE (130,30)-(110,30):LINE (110,3
0)-(90,180)
1160 LINE (90,180)-(100,180):LINE (100,
180)-(120,40)
1170 PAINT (120,35),9,15
1180 LINE (70,50)-(170,120),1,BF
1190 COLOR ,1
1200 RETURN
1210 REM <<< ENTRADA DE DADOS >>>
1220 G$=""
1230 PRESET (80,100):COLOR 15
1240 A$=INKEY$:IF A$="" THEN GOTO 1240
1250 PLAY "O4L16B"

```

```

1260 A=ASC(A$):IF (A>47 AND A<58) OR A=
45 THEN PRINT #1,A$;:G$=G$+A$:GOTO 1240
1270 IF A=13 THEN GOTO 1330
1280 IF A<>B OR G$="" THEN GOTO 1240
1290 LINE (70,90)-(170,120),1,BF
1300 G$=LEFT$(G$,LEN(G$)-1)
1310 PRESET (75,100):PRINT #1,G$;
1320 GOTO 1240
1330 IF VAL(G$)=ANS THEN CO=1 ELSE CO=0
1340 RETURN
1350 REM <<<< RESPOSTA >>>>
1360 FOR G=1 TO 2
1370 GOSUB 1220
1380 IF CO=1 THEN GOTO 1440
1390 PLAY "L802C01C"
1400 LINE (70,90)-(170,120),1,BF
1410 NEXT G
1420 PLAY "L1602FEDFEDFEDC"
1430 GOTO 1460
1440 PLAY "L1604CDECDECDEA"
1450 G=2:TT=TT+1
1460 LINE (70,50)-(170,120),1,BF:RETURN
1470 REM <<<< TIPO DE OPERAÇÃO >>>>
1480 LINE (70,50)-(170,120),1,BF
1490 COLOR 15:PRESET (75,60):PRINT #1,""
1> '+'
1500 COLOR 9:PRESET (75,75):PRINT #1,"2
> '-'
1510 COLOR 3:PRESET (75,90):PRINT #1,"3
> 'x'"
1520 COLOR 11:PRESET (75,105):PRINT #1,
"4> '/'"
1530 A$=INPUT$(1):A=ASC(A$)-48
⇒ 1540 IF A<1 OR A>4 THEN PLAY "L801C":GO
TO 1530
1550 DEF FN SUM=A+B:S$="+"
1560 IF A=2 THEN DEF FN SUM=A-B:S$="-"
1570 IF A=3 THEN DEF FN SUM=A*B:S$="x"
1580 IF A=4 THEN DEF FN SUM=A/B:S$="/" :
DIV=1

```

```

1590 RETURN
1600 REM <<< INICIALIZAÇÃO >>>
1610 LIM=B:DIV=0
1620 IF LEV=2 THEN LIM=12:GOSUB 1480
1630 IF LEV=3 THEN LIM=25:GOSUB 1480
1640 LINE (70,50)-(170,120),1,BF
1650 TT=0:RETURN
1660 REM <<< NÚMEROS >>>
1670 A=INT(RND(-TIME)*LIM)+1
1680 B=INT(RND(1)*LIM)+1
1690 IF DIV=1 THEN A=A*B
1700 LINE (70,50)-(170,120),1,BF
1710 COLOR 15:PRESET (72,62)
1720 IF LEV=1 THEN GOTO 1760
1730 PRINT #1,A;S$;B
1740 ANS=FN SUM
1750 RETURN
1760 COLOR 10:FOR N=1 TO A:PRINT #1,CHR$(1)+CHR$(66);:NEXT N:PRINT #1," +"
1770 PRESET (72,70)
1780 FOR N=1 TO B:PRINT #1,CHR$(1)+CHR$(65);:NEXT N:PRINT #1," ="
1790 ANS=A+B
1800 RETURN
1810 REM <<< LOOP PRINCIPAL >>>
1820 GOSUB 1610
1830 FOR P=1 TO 10
1840 GOSUB 1670:GOSUB 1360
1850 NEXT P
1860 LINE (70,50)-(170,120),1,BF
1870 COLOR 15
1880 PRESET (75,58)
1890 PRINT #1,"ACERTOU";TT
1900 PRESET (75,82)
1910 PRINT #1,"OU";TT*100/10;"%"
1920 PRESET (75,106)
1930 PRINT #1,"APERTE 'C'"
1940 A$=INPUT$(1):IF A$<>"C" AND A$<>"c"
" THEN GOTO 1940

```

1950 LINE (70,50)-(170,120),1,BF
1960 GOTO 1040

24

Hoje, o Brasil



Amanhã, o Mundo!

Este é um teste de geografia baseado no mapa do Brasil. O jogo consiste em identificar corretamente os estados e capitais indicados no mapa. O jogador pode escolher se o teste vai versar sobre cidades, estados ou ambos.

O programa pode ser usado como modelo para desenvolver um teste mais completo a respeito do Brasil ou, como a legenda acima sugere, um teste a respeito do mundo inteiro! Na verdade, a pesquisa dos dados a serem incluídos na listagem pode facilmente tornar-se mais educativa que o próprio programa...

```
1000 REM <<< HOJE O BRASIL >>>
1010 MAXFILES=2:OPEN "GRP:S" FOR OUTPUT
    AS 1
```

```

1020 COLOR 15,1,1:SCREEN 1,0,0:KEY OFF
1030 A$="ABCDEFGHIJKLMNPQRSTUVWXYZ"
1040 SC=0:LOCATE 8,0:PRINT "Hoje o Bras
il"
1050 B$="ABCDEFGHIJKLMNPQRSTUVWXYZ"
1060 LOCATE 7,2:PRINT "Amanhã o mundo!"
1070 COLOR 7
1080 LOCATE 1,5:PRINT "Faça sua escolha
:"
1090 LOCATE 2,7:PRINT "1...Identificar
o estado"
1100 LOCATE 2,9:PRINT "2...Identificar
a capital"
1110 LOCATE 2,11:PRINT "3...Identificar
o estado e"
1120 LOCATE 6,13:PRINT "a capital"
1130 A$=INKEY$:IF A$<"1" OR A$>"3" THEN
  GOTO 1130
1140 OP=VAL(A$)
1150 GOSUB 1980
1160 RESTORE 2690
1170 IF OP=3 THEN GOTO 1600
1180 REM <<< IDENTIFICAR CAPITAL/ESTAD
O >>>
1190 FOR C=1 TO 26
1200 M=INT(RND(-TIME)*LEN(B$))+1
1210 N=ASC(MID$(B$,M,1))-64
1220 B$=LEFT$(B$,M-1)+RIGHT$(B$,LEN(B$)
-M)
1230 FOR L=1 TO N
1240 READ N$,X,Y,CO$
1250 NEXT L
1260 RESTORE 2690
1270 PSET (INT(2.8*X),INT(3.5*Y))
1280 PUT SPRITE 1,(INT(2.8*X+2),INT(3.5
*Y-10)),9,1
1290 COLOR 7
1300 IF OP=1 THEN IF LEN(N$)<14 THEN PR
ESET (155,32):PRINT #1,"A capital é":PR

```

```
ESET (155,48):PRINT #1,N$ ELSE PRESET (155,16):PRINT #1,"A capital ":PRESET (155,32):PRINT #1,LEFT$(N$,4):PRESET (155,48):PRINT #1,RIGHT$(N$,LEN(N$)-4)
1310 IF OP=2 THEN IF LEN(CO$)<15 THEN P
RESET (148,32):PRINT #1,"O estado é":PR
ESET (148,48):PRINT #1,CO$ ELSE PRESET
(148,16):PRINT #1,"O estado ":PRESET (148,32):PRINT #1,LEFT$(CO$,14):PRESET (148,48):PRINT #1,RIGHT$(CO$,LEN(CO$)-14)
1320 TR=3
1330 PRESET (155,64):IF OP=1 THEN PRINT
#1,"O estado é" ELSE PRINT #1,"A capit
al é"
1340 GOSUB 2270:IF OP=1 THEN C$=I$ ELSE
CI$=I$
1350 IF OP=1 AND C$=CO$ THEN GOTO 1460
1360 IF OP=2 AND CI$=N$ THEN GOTO 1460
1370 TR=TR-1:PRESET (164,120):COLOR 9:P
RINT #1,"ERRADO!"
1380 IF TR=2 THEN PRESET (164,128):PRIN
T #1,"RESTAM 2":PRESET (164,136):PRINT
#1,"TENTATIVAS"
1390 IF TR=1 THEN PRESET (164,128):PRIN
T #1,"RESTA 1":PRESET (164,136):PRINT #
1,"TENTATIVA"
1400 COLOR 7
1410 IF TR=0 THEN GOTO 2180
1420 PLAY "EDC":FOR W=1 TO 500:NEXT W
1430 LINE (155,64)-(255,160),1,BF
1440 IF TR<1 THEN GOTO 2180
1450 GOTO 1330
1460 GOSUB 2110
1470 PRESET (169,96):COLOR 9:PRINT #1,"
CERTO !"
1480 PLAY "EFG":FOR W=1 TO 500:NEXT W
1490 SC=SC+TR
1500 PSET (INT(2.8*X),INT(3.5*Y)),1:GOS
UB 2110
```

```

1510 NEXT C
1520 SCREEN 1,0,0
1530 COLOR 15,1,1:CLS
1540 LOCATE 5,5:PRINT "Você fez";SC;"po
ntos"
1550 LOCATE 0,10:PRINT "Aperte 'J' para
jogar de novo"
1560 A$=INPUT$(1):A$=CHR$(ASC(A$) AND 2
23)
1570 IF A$<>"J" THEN GOTO 1560
1580 GOTO 1020
1590 REM <<< IDENTIFICAR CAPITAL E EST
ADO >>>
1600 FOR C=1 TO 26
1610 M=INT(RND(-TIME)*LEN(B$))+1
1620 N=ASC(MID$(B$,M,1))-64
1630 B$=LEFT$(B$,M-1)+RIGHT$(B$,LEN(B$)
-M)
1640 FOR L=1 TO N
1650 READ N$,X,Y,C$:COLOR 7
1660 NEXT L
1670 RESTORE 2690
1680 PSET (INT(2.8*X),INT(3.5*Y))
1690 PUT SPRITE 1,(INT(2.8*X+2),INT(3.5
*Y-10)),9,1
1700 PRESET (155,32):PRINT #1,"A capita
l é"
1710 TR=3
1720 GOSUB 2270:CI$=I$
1730 IF CI$=N$ THEN GOSUB 2110:PRESET (
164,96):COLOR 9:PRINT #1,"CERTO !":GOSU
B 2130:SC=SC+TR:GOSUB 2110:GOTO 1820
1740 TR=TR-1:PRESET (164,120):COLOR 6,1
:PRINT #1,"ERRADO !"
1750 IF TR=2 THEN PRESET (164,128):PRIN
T #1,"RESTAM 2":PRESET (164,136):PRINT
#1,"TENTATIVAS"
1760 IF TR=1 THEN PRESET (164,128):PRIN
T #1,"RESTA 1":PRESET (164,136):PRINT #
1,"TENTATIVA"

```

```

1770 GOSUB 2150:PRESET (164,32)
1780 LINE (155,64)-(255,160),1,BF
1790 IF TR<>0 THEN GOTO 1720
1800 GOSUB 2110:PRESET (148,24):PRINT #
1,"A capital é"
1810 PRESET (148,40):IF LEN(N$)<13 THEN
    PRINT #1,N$ ELSE PRINT #1,LEFT$(N$,13)
:PRESET (148,52):PRINT #1,RIGHT$(N$,LEN
(N$)-13)
1820 COLOR 7:PRESET (148,64):PRINT #1,""
O estado é"
1830 TR=3
1840 GOSUB 2270:CO$=I$
1850 IF CO$=C$ THEN GOSUB 2110:PRESET (
164,80):COLOR 9:PRINT #1,"CERTO !":GOSU
B 2130:SC=SC+TR:GOSUB 2110:GOTO 1950
1860 TR=TR-1
1870 PRESET (164,120):COLOR 6,1:PRINT #
1,"ERRADO !"
1880 IF TR=2 THEN PRESET (164,128):PRIN
T #1,"RESTAM 2":PRESET (164,136):PRINT
#1,"TENTATIVAS"
1890 IF TR=1 THEN PRESET (164,128):PRIN
T #1,"RESTA 1":PRESET (164,136):PRINT #
1,"TENTATIVA"
1900 GOSUB 2150
1910 LINE (145,72)-(255,160),1,BF
1920 IF TR<>0 THEN GOTO 1840
1930 GOSUB 2110:PRESET (148,64):PRINT #
1,"O estado é"
1940 PRESET (148,80):IF LEN(C$)<13 THEN
    PRINT #1,C$ ELSE PRINT #1,LEFT$(C$,13)
:PRESET (148,96):PRINT #1,RIGHT$(C$,LEN
(C$)-13)
1950 FOR FF=1 TO 1000:NEXT FF:PSET (INT
(2.8*X),INT(3.5*Y)),1:GOSUB 2110
1960 NEXT C:GOTO 1520
1970 REM <<< DESENHAR O MAPA >>>
1980 COLOR 5,1,1:SCREEN 2,0,0

```

```

1990 RESTORE 2960:S$="" :FOR N=0 TO 7:RE
AD A:S$=S$+CHR$(A):NEXT N:SPRITE$(1)=S$
2000 RESTORE
2010 PLAY "L1RRRR"
2020 PLAY "L4CL5FL8EL7FL8GL6AL8GL7AL8A+
L2B05L10C04L4F","V603R4L1FF","03R4AR304
CR303AR302F"
2030 READ X,Y:Y=INT(3.5*Y):X=5+INT(2.8*
X):PSET (X,Y)
2040 FOR N=1 TO 162
2050 READ X,Y:Y=INT(3.5*Y):X=5+INT(2.8*
X):L$="M"+STR$(X)+","+STR$(Y):DRAW L$
2060 NEXT N
2070 READ X,Y:Y=INT(3.5*Y):X=5+INT(2.8*
X):PSET (X,Y)
2080 FOR N=1 TO 6
2090 READ X,Y:Y=INT(3.5*Y):X=5+INT(2.8*
X):L$="M"+STR$(X)+","+STR$(Y):DRAW L$
2100 NEXT N
2110 LINE (148,0)-(255,191),1,BF
2120 RETURN
2130 PLAY "GGLBCDEF66":GOSUB 2160
2140 RETURN
2150 PLAY "L8CEGFEDC"
2160 FOR T=1 TO 800:NEXT T
2170 RETURN
2180 GOSUB 2110
2190 PRESET (148,24):COLOR 9
2200 PRINT #1,"A capital é"
2210 PRESET (148,40):PRINT #1,N$
2220 PRESET (148,56):PRINT #1,"O estado
é"
2230 PRESET (148,72):IF LEN(C0$)<13 THE
N PRINT #1,C0$ ELSE PRINT #1,LEFT$(C0$,
13):PRESET (148,88):PRINT #1,RIGHT$(C0$,
LEN(C0$)-13)
2240 FOR W=1 TO 800:NEXT W
2250 GOSUB 2110
2260 GOTO 1500

```

```

2270 COLOR 5
2280 I$="" :PRESET (148,80) :PRINT #1,>""
2290 A$=INKEY$ :IF A$="" THEN GOTO 2290
2300 BEEP :IF A$=CHR$(13) THEN RETURN
2310 IF A$>"Z" THEN A$=CHR$(ASC(A$)-32)
2320 IF ASC(A$)=100 OR ASC(A$)=128 THEN
    A$="A"
2330 IF ASC(A$)=144 OR ASC(A$)=145 THEN
    A$="A"
2340 IF ASC(A$)=105 OR ASC(A$)=129 THEN
    A$="I"
2350 IF ASC(A$)=110 OR ASC(A$)=115 THEN
    A$="O"
2360 IF ASC(A$)=106 OR ASC(A$)=130 THEN
    A$="O"
2370 IF ASC(A$)=98 OR ASC(A$)=112 THEN
    A$="E"
2380 IF ASC(A$)=99 OR ASC(A$)=108 THEN
    A$="A"
2390 X$="" :IF LEN(I$)<12 THEN W$=I$ ELS
E W$=LEFT$(I$,11) :X$=RIGHT$(I$,LEN(I$)-
11)
2400 COLOR 1 :PRESET (164,80) :PRINT #1,W$
2410 IF X$<>"" THEN PRESET (164,96) :PRI
NT #1,X$
2420 IF A$=CHR$(8) AND I$>"" THEN I$=LE
FT$(I$,LEN(I$)-1) :GOTO 2450
2430 IF A$=" " OR A$=". " THEN GOTO 2440
2440 I$=I$+A$.
2450 COLOR 5
2460 X$="" :IF LEN(I$)<12 THEN W$=I$ ELS
E W$=LEFT$(I$,11) :X$=RIGHT$(I$,LEN(I$)-
11)
2470 PRESET (164,80) :PRINT #1,W$
2480 IF X$<>"" THEN PRESET (164,96) :PRI
NT #1,X$
2490 GOTO 2290
2500 REM <<< MAPA DO BRASIL >>>

```

```

2510 DATA 7,8,7,8,8,8,9,7,10,8,11,9,12,
8,13,7.75,14,7.5,13,7
2520 DATA 12.5,6.75,12,5,13,5.25,14,5,1
5,4,16,4,17,3,18,4,18,5,18,6
2530 DATA 18,7,19,7,20,7,21,7,22,6,23,6
,24,6,25,6,26,6,26.75,5
2540 DATA 27.75,3.5,28,4,28.25,5.5,29,5
.5,29.5,7,28.5,8,28.5,9,29,10,30,11,31,
11,32,10
2550 DATA 33,9,34,10,35,10,36,10.75,37,
11,38,11.25,39,11.5,40,11.75,41,12,42,1
2.25
2560 DATA 43,12.5,44,12.75,45,13,46,14,
47,14.25,48.25,14.75,48.75,15.25,49,15.
75,49.5,16,49.75,17,49.25,18
2570 DATA 49,19,48,20,47,21,46,22,46,23
,45,24,45,25,44,26,44,27,44,28
2580 DATA 44,29,44,30,43,31,43,32,42,33
,42,34,42,35,41,35,41,36,40,36
2590 DATA 39,36,38,37,37,37,36,38,35,38
,34,39,34,40,33,41,33,42,33,43
2600 DATA 32,44,31,45,30,46,30,47,29,48
,29,49,28,50,28,49,27,48,26,47
2610 DATA 25,46,24,46,24,45,25,44,26,43
,27,42,27,41,27,40,26,39,25,38
2620 DATA 25,37,24,36,23,36,22,36,22,35
,22,34,22,33,22,32,22,31,21,30
2630 DATA 21,29,20,29,19,29,18,29,18,28
,18,27,17,26,16,26,15,26,14,25
2640 DATA 13,25,13,24,13,23,13,22,12,22
,11,23,10,24,9,24,8,24,7,24
2650 DATA 6,24,6,23,6,22,5,23,4,23,3,22
,2,21,2,20,2,19,3,18
2660 DATA 4,17,5,16,6,16.5,6.5,15,6.75,
14,7,13,7,12,8,11,8,10,7,9,7,8
2670 REM <<< ILHA DE MARAJO >>>
2680 DATA 29.5,8,31.75,8.5,32,9,31,10,2
9.5,9.5,29.5,8.25,29.5,8
2690 <<< ESTADOS >>>


```

2700 DATA PORTO ALEGRE,30.8,46.1,RIO GRANDE DO SUL
2710 DATA FLORIANOPOLIS,33.6,42.9,SANTA CATARINA
2720 DATA CURITIBA,32.6,39.9,PARANA
2730 DATA SAO PAULO,35.5,37.5,SAO PAULO
2740 DATA RIO DE JANEIRO,39,36.5,RIO DE JANEIRO
2750 DATA BELO HORIZONTE,38.5,33.5,MINAS GERAIS
2760 DATA VITORIA,42.8,34,ESPIRITO SANTO
2770 DATA SALVADOR,45.5,24.9,BAHIA
2780 DATA ARACAJU,46.5,22.5,SERGIPE
2790 DATA MACEIO,47.6,20.9,ALAGOAS
2800 DATA RECIFE,49.5,18.7,PERNAMBUCO
2810 DATA JOAO PESSOA,50,17,PARAIBA
2820 DATA NATAL,49.5,15.5,RIO GRANDE DO NORTE
2830 DATA FORTALEZA,44.6,13.2,CEARA
2840 DATA SAO LUIS,38,12,MARANHAO
2850 DATA BELEM,32.6,12,PARA
2860 DATA MACAPA,29,8.9,AMAPÁ
2870 DATA BOA VISTA,16.5,6.5,RORAIMA
2880 DATA MANAUS,18.5,13.5,AMAZONAS
2890 DATA RIO BRANCO,9.4,23,ACRE
2900 DATA PORTO VELHO,14,21.5,RONDONIA
2910 DATA CUIABA,24.5,28.5,MATO GROSSO
2920 DATA CAMPO GRANDE,26.5,34.5,MATO GROSSO DO SUL
2930 DATA GOIANIA,32.5,29.5,GOIAS
2940 DATA BRASILIA,34,28.5,DISTRITO FEDERAL
2950 DATA TERESINA,39.9,15.1,PIAUI

2960 DATA &H01,&H02,&H84,&H88,&HD0,&HE0,&HF0,&HFC

25

Gerador de Caracteres



Crie Seu Próprio Conjunto de Caracteres Gráficos

Criar caracteres pode ser tão divertido quanto jogar um jogo. Este programa utilitário permite redefinir os caracteres do MSX e combiná-los para gerar figuras mais complexas.

O programa mostra os caracteres originais e oferece ao usuário a oportunidade de adaptá-los, permutá-los ou criar caracteres totalmente novos. As definições dos caracteres podem ser copiadas e usadas nos mais diversos tipos de programas. Conjuntos completos de caracteres podem ser salvos em fita para uso futuro.

Embora alguns programas deste livro utilizem caracteres especiais, outros usam apenas os caracteres originais. A apresentação dos jogos do segundo tipo pode ser melhorada com o auxílio do gerador de caracteres.

Como o programa oferece um número relativamente grande de op-

ções, pode ser que você leve um certo tempo para aprender a usá-lo da forma mais eficiente. Não deixe de consultar a lista de instruções abaixo sempre que tiver alguma dúvida.

Uso do Gerador de Caracteres

O programa começa por mostrar o primeiro caractere escolhido, dentro de um quadriculado de 8 por 8. O usuário pode então escolher uma das opções abaixo, apertando a tecla com a letra correspondente à inicial da instrução:

Apagar: Apaga o caractere corrente.

Ler: Pede o número de um caractere. Depois que o usuário entra com o último algarismo, o caractere escolhido aparece dentro do quadriculado e pode ser editado, deslocado etc.

Editar: Coloca um cursor intermitente no canto superior esquerdo do quadriculado. O cursor pode ser movimentado com o auxílio do joystick ou das teclas de controle do cursor. O botão de tiro ou a barra de espaço serve para mudar um quadrado de "cheio" para "vazio" ou vice-versa. Para dar a figura por terminada, aperte a tecla **RETURN**. Em seguida, escolha a que número de código será associado o caractere que você acabou de criar.

Números: Nesta opção, o usuário escolhe o número de código a que o caractere será associado e depois entra com oito valores, correspondentes aos números binários que representam as oito linhas do quadriculado, começando pela linha de cima. Desta forma, o novo caractere é gerado diretamente a partir de dados numéricos. Para passar para outra opção, aperte a tecla **RETURN**.

Refletir: O computador pede o número de um caractere e depois pergunta se ele deve ser refletido em um eixo vertical (trocar direita por esquerda) ou horizontal (virar de cabeça para baixo). O resultado aparece no quadriculado. Para armazenar o novo caractere, basta escolher a opção Editar, apertar a tecla **RETURN** e entrar com o número de código escolhido. A mesma observação vale para as duas opções seguintes.

Vertical (Deslocamento): O computador pergunta se o deslocamento é para cima ou para baixo e desloca a figura uma linha para cima ou para baixo. A linha que desaparece na parte de cima torna a aparecer na parte de baixo e vice-versa.

Horizontal (Deslocamento): Como no caso do deslocamento vertical, excepto pelo fato de que o computador pergunta se o deslocamento deve ser para a direita ou para a esquerda.

Trocar: O computador pede dois números de código e troca as definições dos caracteres correspondentes.

Montar: Remove o quadriculado e cria um campo de 14 x 10 caracteres no qual o usuário pode observar o resultado de qualquer combinação de caracteres. Depois de fornecer o número de código de um caractere, o usuário pode deslocar esse caractere para qualquer posição dentro do campo, usando o *joystick* ou as teclas de controle do cursor. Para “fixar” um caractere na sua posição corrente, basta apertar o botão de tiro ou a barra de espaço. Para entrar com um novo caractere, aperte **RETURN** e, sem soltar a tecla, entre com o número de código do caractere desejado. Para sair desta opção, aperte **RETURN**.

Salvar: O computador pergunta com que nome o arquivo de caracteres deve ser salvo. Depois de entrar com o nome, aperte as teclas **PLAY** e **RECORD** do gravador e a tecla **RETURN** do computador.

Carregar: O computador pergunta qual o nome do arquivo a ser carregado. Se o usuário se limita a apertar a tecla **RETURN**, o computador carrega o primeiro arquivo que encontrar.

Nota: Só use as opções “Salvar” e “Carregar” se você estiver trabalhando com fita cassete.

```

1000 REM <<< GERADOR DE CARACTERES - A
NDY/BOOTSY >>>
1010 SCREEN 1,0,0:COLOR 15,4,4
1020 CLEAR 4000,&HEFFF:WIDTH 31
1030 DIM C$(8):DEFUSR=&HF000:DEFUSR1=34
2
1040 ON ERROR GOTO 3470
1050 GOSUB 3500
1060 FOR A=1 TO 8:C$(A)="00000000":NEYT
A
1070 YP=8
1080 GOSUB 2520
1090 GOSUB 2760
1100 ON YP GOSUB 2160,2470,1160,1730,34
20,1930,3100,2310,3310,1580,3020
1110 LOCATE 18,19:PRINT STRING$(10," ")

```

```

1120 LOCATE 18,20:PRINT STRING$(10," ")
1130 GOTO 1090
1140 END
1150 REM <<< EDITAR >>>
1160 XC=20:YC=0
1170 LOCATE XC,YC:PRINT CHR$(247):X=XC-
19
1180 FOR T=1 TO 50:NEXT T
1190 LOCATE XC,YC:PRINT CHR$(32)
1200 A=STICK(0) OR STICK(1)
1210 IF A=1 AND YC>0 THEN 1350
1220 IF A=3 AND XC<27 THEN 1390
1230 IF A=5 AND YC<7 THEN 1430
1240 IF A=7 AND XC>20 THEN 1470
1250 AA=STRIG(0) OR STRIG(1):IF AA THEN
1510
1260 A$=INKEY$:IF A$<>CHR$(13) THEN 117
0
1270 LOCATE XC,YC:IF MID$(C$(YC+1),X,1)
="1" THEN PRINT CHR$(248) ELSE PRINT CH
R$(246)
1280 GOSUB 2900
1290 FOR A=1 TO 8
1300 NU=VAL("&B"+C$(A))
1310 VPOKE CH*8+(A-1),NU
1320 NEXT
1330 LOCATE 18,19:PRINT STRING$(10," ")
:LOCATE 18,20:PRINT STRING$(10," "):LOC
ATE 18,21:PRINT STRING$(10," "):RETURN
1340 END
1350 Q$=MID$(C$(YC+1),X,1)
1360 LOCATE XC,YC
1370 IF Q$="1" THEN PRINT CHR$(248) ELS
E PRINT CHR$(246)
1380 YC=YC-1:GOTO 1170
1390 Q$=MID$(C$(YC+1),X,1)
1400 LOCATE XC,YC
1410 IF Q$="1" THEN PRINT CHR$(248) ELS
E PRINT CHR$(246)

```

```

1420 XC=XC+1:GOTO 1170
1430 Q$=MID$(C$(YC+1),X,1)
1440 LOCATE XC,YC
1450 IF Q$="1" THEN PRINT CHR$(248) ELSE
   PRINT CHR$(246)
1460 YC=YC+1:GOTO 1170
1470 Q$=MID$(C$(YC+1),X,1)
1480 LOCATE XC,YC
1490 IF Q$="1" THEN PRINT CHR$(248) ELSE
   PRINT CHR$(246)
1500 XC=XC-1:GOTO 1170
1510 X=XC-19:Q$=MID$(C$(YC+1),X,1)
1520 LOCATE XC,YC
1530 IF Q$="1" THEN PRINT CHR$(246):W$=
LEFT$(C$(YC+1),X-1)+"0"+RIGHT$(C$(YC+1),
,(8-X))
1540 IF Q$="0" THEN PRINT CHR$(247):W$=
LEFT$(C$(YC+1),(X-1))+ "1"+RIGHT$(C$(YC+
1),(8-X))
1550 C$(YC+1)=W$
1560 GOTO 1170
1570 REM <<< DESLOCAMENTO VERTICAL >>>
>
1580 LOCATE 18,19:PRINT "Subir/Descer":'
LOCATE 19,20:PRINT "S/D"
1590 A$=INKEY$: IF A$="" THEN 1590 ELSE
   IF A$<>"S" AND A$<>"s" AND A$<>"D" AND
A$<>"d" THEN 1590
1600 LOCATE 18,19:PRINT STRING$(12," ")
:LOCATE 18,20:PRINT STRING$(10," ")
1610 IF A$="S" OR A$="s" THEN 1670
1620 D$=C$(8):FOR A=7 TO 1 STEP -1
1630 C$(A+1)=C$(A):NEXT A
1640 C$(1)=D$
1650 GOSUB 2830
1660 RETURN
1670 D$=C$(1):FOR A=2 TO 8
1680 C$(A-1)=C$(A):NEXT A
1690 C$(8)=D$

```

```

1700 GOSUB 2830
1710 RETURN
1720 REM <<< DESLOCAMENTO HORIZONTAL >
>>>
1730 LOCATE 18,19:PRINT "Esq./Dir."
1740 LOCATE 19,20:PRINT "(E/D)"
1750 A$=INKEY$: IF A$="" THEN 1750
1760 IF A$<>"D" AND A$<>"d" AND A$<>"E"
AND A$<>"e" THEN 1750
1770 LOCATE 18,19:PRINT STRING$(10," ")
1780 LOCATE 18,20:PRINT STRING$(10," ")
1790 IF A$="D" OR A$="d" THEN 1860
1800 FOR A=1 TO 8
1810 D$=LEFT$(C$(A),1)
1820 E$=RIGHT$(C$(A),7)
1830 C$(A)=E$+D$
1840 NEXT A
1850 GOSUB 2830:RETURN
1860 FOR A=1 TO 8
1870 D$=RIGHT$(C$(A),1)
1880 E$=LEFT$(C$(A),7)
1890 C$(A)=D$+E$
1900 NEXT A
1910 GOSUB 2830:RETURN
1920 REM <<< REFLETIR >>>
1930 GOSUB 2900
1940 GOSUB 2320
1950 LOCATE 18,19:PRINT "horiz/vert"
1960 LOCATE 18,20:PRINT "H/V?";;
1970 CH=USR1(0):CH$=INPUT$(1)
1980 IF CH$="v" OR CH$="V" THEN 2070
1990 IF CH$<>"h" AND CH$<>"H" THEN 1970
2000 REM <<< REFLETIR EM UM EIXO VERTI
CAL >>>
2010 D$=C$(8):C$(8)=C$(1):C$(1)=D$
2020 D$=C$(7):C$(7)=C$(2):C$(2)=D$
2030 D$=C$(6):C$(6)=C$(3):C$(3)=D$
2040 D$=C$(5):C$(5)=C$(4):C$(4)=D$
2050 GOSUB 2830:RETURN

```

```

2060 REM <<< REFLETIR EM UM EIXO HORIZ
ONTAL >>>
2070 FOR A=1 TO 8
2080 D$=RIGHT$(C$(A),1):E$=LEFT$(C$(A),
1)
2090 F$=MID$(C$(A),2,1):G$=MID$(C$(A),7
,1)
2100 H$=MID$(C$(A),3,1):I$=MID$(C$(A),6
,1)
2110 J$=MID$(C$(A),4,1):K$=MID$(C$(A),5
,1)
2120 C$(A)=D$+G$+I$+K$+J$+H$+F$+E$
2130 NEXT A
2140 GOSUB 2830:RETURN
2150 REM <<< MONTAR >>>
2160 FOR Y=0 TO 7:LOCATE 19,Y:PRINT STR
ING$(10," "):NEXT Y
2170 XX=19:YY=0
2180 GOSUB 2900:IF CH=13 THEN GOSUB 247
0:RETURN
2190 PUT SPRITE 0,(XX*8+8,YY*8),15,0
2200 A=STICK(0) OR STICK(1)
2210 Y=STRIG(0) OR STRIG(1)
2220 CH$=INKEY$:IF CH$=CHR$(13) THEN 22
90
2230 IF A=1 THEN YY=YY-(1 AND YY>0)
2240 IF A=3 THEN XX=XX+(1 AND XX<27)
2250 IF A=5 THEN YY=YY+(1 AND YY<7)
2260 IF A=7 THEN XX=XX-(1 AND XX>19)
2270 IF Y THEN LOCATE XX,YY:PRINT CHR$(CH)
2280 GOTO 2190
2290 PUT SPRITE 0,(0,208),0,0:GOTO 2180
2300 REM <<< LER >>>
2310 GOSUB 2900
2320 FOR A=1 TO 8:C$(A)=""":NEXT A
2330 FOR A=0 TO 7
2340 AA=VPEEK(CH*8+A)
2350 B=AA AND 128:IF B>0 THEN LOCATE 20

```

```
,A:PRINT CHR$(248):C$(A+1)=C$(A+1)+"1"
ELSE C$(A+1)=C$(A+1)+"0":LOCATE 20,A:PR
INT CHR$(246)
2360 B=AA AND 64:IF B>0 THEN LOCATE 21,
A:PRINT CHR$(248):C$(A+1)=C$(A+1)+"1" E
LSE C$(A+1)=C$(A+1)+"0":LOCATE 21,A:PRI
NT CHR$(246)
2370 B=AA AND 32:IF B>0 THEN LOCATE 22,
A:PRINT CHR$(248):C$(A+1)=C$(A+1)+"1" E
LSE C$(A+1)=C$(A+1)+"0":LOCATE 22,A:PRI
NT CHR$(246)
2380 B=AA AND 16:IF B>0 THEN LOCATE 23,
A:PRINT CHR$(248):C$(A+1)=C$(A+1)+"1" E
LSE C$(A+1)=C$(A+1)+"0":LOCATE 23,A:PRI
NT CHR$(246)
2390 B=AA AND 8:IF B>0 THEN LOCATE 24,A
:PRINT CHR$(248):C$(A+1)=C$(A+1)+"1" EL
SE C$(A+1)=C$(A+1)+"0":LOCATE 24,A:PRIN
T CHR$(246)
2400 B=AA AND 4:IF B>0 THEN LOCATE 25,A
:PRINT CHR$(248):C$(A+1)=C$(A+1)+"1" EL
SE C$(A+1)=C$(A+1)+"0":LOCATE 25,A:PRIN
T CHR$(246)
2410 B=AA AND 2:IF B>0 THEN LOCATE 26,A
:PRINT CHR$(248):C$(A+1)=C$(A+1)+"1" EL
SE C$(A+1)=C$(A+1)+"0":LOCATE 26,A:PRIN
T CHR$(246)
2420 B=AA AND 1:IF B>0 THEN LOCATE 27,A
:PRINT CHR$(248):C$(A+1)=C$(A+1)+"1" EL
SE C$(A+1)=C$(A+1)+"0":LOCATE 27,A:PRIN
T CHR$(246)
2430 LOCATE 18,19:PRINT STRING$(10," ")
:LOCATE 18,20:PRINT STRING$(10," "):LOC
ATE 18,21:PRINT STRING$(10," ")
2440 NEXT A
2450 RETURN
2460 REM <<< APAGAR >>>
2470 FOR A=1 TO 8:C$(A)="00000000"
2480 NEXT A
```

```

2490 FOR Y=0 TO 7:LOCATE 19,Y:PRINT CHR$(32)+STRING$(S,CHR$(246)):NEXT Y
2500 RETURN
2510 REM <<< TELA >>>
2520 KEY OFF:A=200:FOR Y=0 TO 22
2530 LOCATE 0,Y:PRINT A;" ";CHR$(A);"
";A+23;" ";CHR$(A+23)
2540 A=A+1:NEXT Y
2550 VPOKE 246*8,255:VPOKE 247*8,255:VPOKE 248*8,0
2560 FOR B=1 TO 6:VPOKE 246*8+B,129:VPOKE 247*8+B,255:VPOKE 248*8+B,126
2570 NEXT B
2580 VPOKE 246*8+7,255:VPOKE 247*8+7,255:VPOKE 248*8+7,0
2590 A$=STRING$(8,CHR$(246))
2600 FOR Y=0 TO 7
2610 LOCATE 20,Y:PRINT A$
2620 NEXT Y
2630 LOCATE 19,8:PRINT "Montar"
2640 LOCATE 19,9:PRINT "Apagar"
2650 LOCATE 19,10:PRINT "Editar"
2660 LOCATE 19,11:PRINT "Hor.(Desl.)"
2670 LOCATE 19,12:PRINT "Carregar"
2680 LOCATE 19,13:PRINT "Refletir"
2690 LOCATE 19,14:PRINT "Números"
2700 LOCATE 19,15:PRINT "Ler"
2710 LOCATE 19,16:PRINT "Salvar"
2720 LOCATE 19,17:PRINT "Vert.(Desl.)"
2730 LOCATE 19,18:PRINT "Trocá"
2740 RETURN
2750 REM <<< ESCOLHER OPÇÃO >>>
2760 LOCATE 18,20:PRINT "> "+CHR$(29)+CHR$(29)+CHR$(29)+CHR$(29);
2770 A=USR1(0):A=ASC(INPUT$(1)) AND 223
2780 PRINT CHR$(A);:YP=0
2790 FOR I=6420 TO 6740 STEP 32
2800 IF A=VPEEK(I) THEN YP=(I-6420)/32+
1:I=6740

```

```
2810 NEXT I:IF YP=0 THEN 2760
2820 RETURN
2830 FOR B=1 TO 8
2840 FOR C=1 TO 8
2850 IF MID$(C$(B),C,1)="1" THEN LOCATE
    (19+C),(B-1):PRINT CHR$(248) ELSE LOCA
    TE (19+C),(B-1):PRINT CHR$(246)
2860 NEXT C
2870 NEXT B
2880 RETURN
2890 REM <<< ENTRADA DE CARACTERES >>>
>
2900 LOCATE 18,19:PRINT "Caractere":CH=
USR1(0):LOCATE 18,20:PRINT ">"
2910 LOCATE 19,20:PRINT "      ":LOCATE 19
,20:CC=0
2920 CH=ASC(INPUT$(1)):IF CH=13 AND CC=
0 AND YP=1 THEN RETURN
2930 IF (CH<48 OR CH>57) AND CH<>8 THEN
    2920
2940 IF CH=8 AND CC=0 THEN 2920
2950 IF CH=8 THEN PRINT CHR$(127);:CC=C
C-1:GOTO 2920
2960 PRINT CHR$(CH);:CC=CC+1
2970 IF CC<3 THEN 2920
2980 CH$=CHR$(VPEEK(6804))+CHR$(VPEEK(6
805))+CHR$(VPEEK(6806)):CH=VAL(CH$)
2990 IF CH<200 OR CH>255 THEN 2910
3000 RETURN
3010 REM <<< TROCAR >>>
3020 GOSUB 2900:C1=CH*8
3030 GOSUB 2900:C2=CH*8
3040 FOR CH=0 TO 7
3050 C3=VPEEK(C1):C4=VPEEK(C2)
3060 SWAP C3,C4
3070 VPOKE C1,C3:VPOKE C2,C4
3080 C1=C1+1:C2=C2+1:NEXT CH:RETURN
3090 REM <<< NUMEROS >>>
3100 GOSUB 2900
```


```

3110 FOR Y=0 TO 7:CH$=""
3120 LOCATE 19,20:PRINT STRING$(5," ")
3130 LOCATE 18,20:INPUT CH$
3140 IF CH$="" THEN Y=7:NEXT Y:RETURN
3150 CC=VAL(CH$):IF CC>255 THEN 3130
3160 VPOKE CH*8+Y,CC
3170 NEXT Y:GOTO 3110
3180 REM <<< SALVAR >>>
3190 REM Depois de salvos, os caracteres
3200 REM podem ser incluidos em outros
3210 REM programas. Para isso, acrescente
3220 REM as linhas abaixo no começo do
3230 REM seu programa:
3240 REM
3250 REM      10 SCREEN 1,
3260 REM      20 CLEAR 1000,&HEFFF
3270 REM      30 DEFUSR=&HF000
3280 REM      40 BLOAD "<name>"
3290 REM      50 A=USR(0)
3300 REM
3310 RESTORE 3480:FOR CH=&HF000 TO &HF0
0B
3320 READ CC:POKE CH,CC:NEXT
3330 A=USR(0)
3340 FOR CH=&HF000 TO &HF00B
3350 READ CC:POKE CH,CC:NEXT CH
3360 LOCATE 20,19:PRINT "<Nome Arq>"
3370 LOCATE 18,20:INPUT ">";CH$
3380 IF CH$="" THEN CH$="CHR"
3390 BSAVE CH$,&HF000,&HF300
3400 RETURN
3410 REM <<< CARREGAR >>>
3420 LOCATE 20,19:PRINT "<Nome Arq>"
3430 LOCATE 18,20:INPUT ">";CH$
3440 IF CH$="" THEN CH$="CHR"
3450 POKE &HF000,0:BLOAD CH$:IF PEEK(&H
F000) THEN A=USR(0)
3460 RETURN
3470 RESUME NEXT

```

```
3480 DATA 1,112,1,17,12,240,33,64,6,195
,89,0,1,112,1,17,64,6,33,12,240,195,92,
0
3490 REM <<< CRUZ >>>
3500 RESTORE 3540:CH$=""
3510 FOR CH=1 TO 8:READ CC
3520 CH$=CH$+CHR$(CC):NEXT CH
3530 SPRITE$(0)=CH$:RETURN
3540 DATA 24,24,24,255,255,24,24,24
```

Gerador de Figuras



A Sua Paleta Eletrônica

Aqui está um programa utilitário que o ajudará a dar vazão ao seu talento artístico. O programa funciona no modo SCREEN 2, que é o modo gráfico de alta resolução, e oferece opções que permitem desenhar figuras na tela do MSX com toda facilidade.

O programa começa por mostrar na tela uma caneta simulada. Qualquer movimento da “caneta”, executado com o auxílio do *joystick* ou das teclas de controle do cursor, produz uma linha na tela. Apertando uma vez o botão de tiro ou a barra de espaço, o símbolo da caneta é substituído pelo de uma borracha. A “borracha” pode ser usada para apagar qualquer desenho presente na tela. Quando o botão de tiro é apertado uma segunda vez, o símbolo da borracha é substituído por uma cruz. Isto indica que o cursor pode ser movimentado sem modificar o desenho. Quan-

do o botão de tiro é apertado uma terceira vez, a cruz é substituída pelo símbolo da caneta. Assim, para selecionar qualquer um dos três modos, basta apertar o botão de tiro (ou a barra de espaço) um número apropriado de vezes.

Enquanto o programa está no modo de movimento (símbolo em forma de cruz), várias opções adicionais podem ser usadas. Experimente usar a tecla **C** para desenhar um círculo, **B** para desenhar um retângulo, **L** para traçar uma reta e **INS** para colorir o interior de uma figura. Nos três primeiros casos, aperte a tecla **HOME** para indicar o ponto inicial da reta, o centro do círculo ou um dos vértices do retângulo; em seguida, movimente o cursor para outra posição, de modo a indicar o ponto final da reta, um dos pontos da circunferência do círculo, o vértice oposto do retângulo ou o ponto final da reta e aperte a tecla **C**, **B** ou **L**, dependendo da opção escolhida. Para colorir uma figura, posicione o cursor no interior da figura (que deve ser fechada, caso contrário a cor "vazará" para o resto da tela) e aperte a tecla **INS**.

A qualquer momento é possível mudar a cor da "caneta" usando a tecla **RETURN**. Cada vez que a tecla é apertada, o computador seleciona uma cor diferente, que é indicada pela cor do cursor.

Para escolher a cor do fundo, aperte **SLCT**.

Para salvar o desenho em fita, aperte **GRAPH-S** (antes disso você deve apertar as teclas **PLAY** e **RECORD** do gravador).

Para carregar um desenho na memória do computador, aperte **GRAPH-C** e depois aperte a tecla **PLAY** do gravador (o computador carregará a primeira figura que encontrar na fita).

Para apagar a tela, aperte **DEL** duas vezes.

Com um pouco de prática, logo você estará criando belas figuras na tela do seu MSX.

Nota: Só use as opções para salvar e carregar desenhos se você estiver trabalhando com fita cassete.

```

1000 REM <<< GERADOR DE FIGURAS - BOOT
SY >>>
1010 SCREEN 2,2,0:CLEAR 300,&HD7F3:COLO
R 15,0,0:CLS
1020 DIM S$(2):PC=15:BC=0:L=1:X=128:Y=9
6:RESTORE
1030 GOSUB 1790
1040 GOTO 1420
1050 REM <<< ENTRADA DE DADOS >>>
1060 C=1
1070 FOR I=0 TO 2

```

```

1080 I$=INKEY$:IP=0:IF I$>"" THEN IP=AS
C(I$)
1090 ST=STICK(I):TG=STRIG(I):IF ST OR T
G OR IP THEN C=0:I=2
1100 NEXT I:S=ST OR TG
1110 RETURN
1120 REM <<< COLOCAR SPRITE NA TELA >>
>>
1130 XS=X-4:YS=Y-4
1140 PUT SPRITE 0,(XS,YS),PC,0
1150 RETURN
1160 REM <<< FUNDO >>>
1170 BC=BC+1 AND 15:COLOR,,BC
1180 RETURN
1190 REM <<< CANETA >>>
1200 PC=PC+1 AND 15:COLOR PC
1210 GOSUB 1130
1220 RETURN
1230 REM <<< RETA/CIRCULO/RETANGULO >>
>>
1240 IF CR OR L=0 AND X0=X AND Y0=Y OR
L AND IP<>11 THEN RETURN
1250 IF L THEN BEEP:X0=X:Y0=Y:L=0:RETUR
N
1260 IF IP=76 OR IP=108 THEN LINE (X0,Y
0)-(X,Y),PC
1270 IF IP=66 OR IP=98 THEN LINE (X0,Y0
)-(X,Y),PC,B
1280 IF IP=67 OR IP=99 THEN CIRCLE (X0,
Y0),SQR((ABS(X0-X))^2+(ABS(Y0-Y))^2),PC
1290 L=1:RETURN
1300 REM <<< COLORIR >>>
1310 PAINT (X,Y),PC:RETURN
1320 REM <<< APAGAR >>>
1330 TC=PC
1340 PC=0:GOSUB 1130
1350 GOSUB 1060:IF IP AND IP<>127 THEN
1400
1360 PC=15:GOSUB 1130

```


```

1370 GOSUB 1060:IF C THEN 1340
1380 IF IP<>127 THEN 1400
1390 CLS
1400 PC=TC:GOSUB 1130:RETURN
1410 REM <<< DESENHAR >>>
1420 GOSUB 1060:IF C THEN 1420
1430 X=X+(1 AND ST>1 AND ST<5 AND X<255
)-(1 AND ST>5 AND X>0)
1440 Y=Y+(1 AND ST>3 AND ST<7 AND Y<191
)-(1 AND (ST>0 AND ST<3 OR ST>7) AND Y>
0)
1450 GOSUB 1130
1460 CR=CR+(1 AND TG) AND 3
1470 CR=CR-(CR AND CR=3)
1480 IF TG THEN SPRITE$(0)=S$(CR)
1490 IF CR=1 THEN PSET(X,Y),PC
1500 IF CR=2 THEN PRESET(X,Y)
1510 IF S THEN 1420
1520 REM <<< COMANDO >>>
1530 RESTORE 2010:CN=0
1540 READ CC:CN=CN+1:IF CC=0 THEN 1410
1550 IF CC<>IP THEN 1540
1560 ON CN GOSUB 1170,1200,1240,1240,13
30,1310,1240,1240,1240,1240,1240,1240,1
590,1690
1570 GOTO 1410
1580 REM <<< SALVAR >>>
1590 RESTORE 2020:DEFUSR=&HD7F4
1600 FOR I=&HD7F4 TO &HD7FF
1610 READ H$:POKE I,VAL("&h"+H$)
1620 NEXT I
1630 A=USR(0)
1640 BSAVE "picpat",&HD800,&HF000
1650 POKE &HD7FC,&H20:A=USR(0)
1660 BSAVE "piccol",&HD800,&HF000
1670 RETURN
1680 REM <<< CARREGAR >>>
1690 RESTORE 2030:DEFUSR=&HD7F4
1700 FOR I=&HD7F4 TO &HD7FF

```

```
1710 READ H$:POKE I,VAL("&h"+H$)
1720 NEXT I
1730 BLOAD "picpat"
1740 A=USR(0)
1750 BLOAD "piccol"
1760 POKE &HD7F9,&H20:A=USR(0)
1770 RETURN
1780 REM <<< DEFINIR SPRITES >>>
1790 FOR I=0 TO 2
1800 S$="" :FOR J=1 TO 32
1810 READ H$:S$=S$+CHR$(VAL("&h"+H$))
1820 NEXT J
1830 S$(I)=S$
1840 NEXT I
1850 SPRITES$(0)=S$(1):CF=1
1860 GOSUB 1130
1870 RETURN
1880 REM <<< DADOS >>>
1890 DATA 10,10,10,fe,10,10,10,10,0
1900 DATA 0,0,0,0,0,0,0,0,0
1910 DATA 0,0,0,0,0,0,0,0,0
1920 DATA 0,0,0,0,0,0,0,0,0
1930 DATA 0,0,0,10,c,7,7,3
1940 DATA 1,0,0,0,0,0,0,0,0
1950 DATA 0,0,0,0,0,0,80,c0
1960 DATA f4,fa,fd,7f,3e,1c,0,0
1970 DATA 0,0,0,e,f,7,3,1
1980 DATA 0,0,0,0,0,0,0,0,0
1990 DATA 0,0,0,0,0,80,c0,e0
2000 DATA f0,78,3c,1e,f,0,0,0
2010 DATA 24,13,11,12,127,18,66,67,76,9
8,99,108,210,200,0
2020 DATA 1,0,18,11,0,d8,21,0,0,c3,59,0
2030 DATA 1,0,18,11,0,0,21,0,d8,c3,5c,0
```

Monitor da Memória



Revelando os Segredos Mais Íntimos do Seu Micro

Este programa utilitário o ajudará a conhecer melhor a memória do MSX. O programa permite observar as rotinas da ROM ou qualquer programa que esteja carregado na RAM. O conteúdo da memória pode ser também modificado ou salvo em fita.

Uso dos Comandos

Observar a Memória. O computador pede os endereços inicial e final, que devem ser fornecidos em hexadecimal. O conteúdo da memória é então mostrado na tela, em blocos de dez posições. Para passar para o bloco seguinte, basta apertar a barra de espaço.

Teste da Soma. O computador pede os endereços inicial e final e calcula a soma dos endereços e dos conteúdos de todas as memórias especificadas. Esta opção é usada para verificar se houve algum erro durante a digitação de um programa cuja listagem inclua o valor desta soma.

Modificar a Memória. O computador pede os endereços inicial e final e mostra o conteúdo do primeiro endereço. Para mudar o conteúdo da memória, basta entrar com um número e apertar **RETURN**. Se o número não for permitido, o computador indicará este fato e não fará nenhuma modificação. Para deixar o conteúdo da memória como está e passar para o endereço seguinte, basta apertar a tecla **RETURN**.

Atenção: A modificação da memória deve ser feita com muita cautela, pois todo o funcionamento do computador pode ser afetado.

Salvar em Fita. O computador pede os endereços inicial e final e o nome do arquivo. A operação só é executada quando o usuário aperta a tecla **RETURN**. Antes de escolher esta opção, as teclas **RECORD** e **PLAY** do gravador devem ser apertadas.

Carregar da Fita. Escolha esta opção e aperte a tecla **PLAY** do gravador. O computador carregará o primeiro arquivo que encontrar.

Para desistir de qualquer opção, aperte **CTRL-D**.

Nota: Só use as opções para salvar e carregar dados da memória se você estiver trabalhando com fita cassete.

```

1000 REM <<< MONITOR DA MEMORIA - BOOT
SY >>>
1010 SCREEN 0:KEY OFF
1020 GOTO 1180
1030 Q=0:IP$=INPUT$(1):IP=ASC(IP$):IF I
P>64 THEN IP=IP AND 223:IP$=CHR$(IP)
1040 IF IP=4 THEN Q=1
1050 RETURN
1060 H$=""
1070 GOSUB 1030:IF Q=1 THEN RETURN
1080 IF IP=8 AND H$>"" THEN H$=LEFT$(H$,
,LEN(H$)-1):PRINT CHR$(127);
1090 IF IP=13 THEN RETURN
1100 IF (IP<48 OR IP>57) AND (IP<65 OR
IP>70) THEN GOTO 1070
1110 IF LEN(H$)=4 THEN GOTO 1070
1120 H$=H$+IP$:PRINT IP$;:GOTO 1070
1130 PRINT "DE:";:GOSUB 1060:IF Q=1 THE
N RETURN

```

```

1140 FH=VAL("&h"+H$)
1150 PRINT TAB(12) "AT:"; :GOSUB 1060:IF
   Q=1 THEN RETURN
1160 TH=VAL("&h"+H$)
1170 RETURN
1180 PRINT "Monitor da Memória":PRINT
1190 PRINT
1200 PRINT " (A)judar - este menu"
1210 PRINT " (O)bservar a memória"
1220 PRINT " (T)este da soma"
1230 PRINT " (M)odificar a memória"
1240 PRINT " (S)alvar em fita"
1250 PRINT " (D)esistir"
1260 PRINT:PRINT " <Ctrl-D> para desistir":PRINT:PRINT
1270 PRINT ":";
1280 GOSUB 1030
1290 PRINT CHR$(IP);
1300 RESTORE:K=0:FOR I=1 TO 6
1310 READ J:IF IP=J THEN K=I:I=6
1320 NEXT
1330 IF K=0 THEN PRINT "?":GOTO 1270
1340 PRINT:ON K GOSUB 1370,1430,1540,16
70,1540,1360
1350 GOTO 1270
1360 GOTO 1200
1370 GOSUB 1130:IF Q=1 THEN PRINT:RETUR
N
1380 PRINT TAB(24) "SOMA:";
1390 IF TH<FH THEN PRINT "?":RETURN
1400 H=0:FOR I=FH TO TH:H=H+PEEK(I)
1410 H=H MOD 256:NEXT I
1420 H$=HEX$(H):PRINT STRING$(2-LEN(H$))
,"0")+H$":RETURN
1430 GOSUB 1130:PRINT:IF Q=1 THEN RETUR
N
1440 IF TH<FH THEN TH=FH+7
1450 L=0:FOR I=FH TO TH STEP 8:L=L+1
1460 H$=HEX$(I):PRINT ":";STRING$(4-LEN
(H$), "0");H$;" ";

```

```

1470 FOR J=0 TO 7
1480 H$=HEX$(PEEK(I+J)):PRINT STRING$(2
-LEN(H$),"0");H$;" ";:NEXT J
1490 FOR J=0 TO 7:K=PEEK(I+J)
1500 IF K>31 THEN PRINT CHR$(K); ELSE P
RINT ".";
1510 NEXT J:PRINT:IF L=23 THEN I$=INPUT
$(1):L=0:IF ASC(I$)=17 THEN RETURN
1520 NEXT I
1530 RETURN
1540 S$=IP$
1550 INPUT "Nome do arquivo: ";F$
1560 IF F$="" THEN RETURN
1570 Q=0:IF S$="S" THEN GOSUB 1130
1580 PRINT:IF Q=1 THEN RETURN
1590 IF TH<FH AND S$="S" THEN PRINT ":"?
":RETURN
1600 IF S$="S" THEN PRINT "Aperte PLAY/
REC e depois qualquer tecla";
1610 IF S$="C" THEN PRINT "Aperte PLAY
e depois qualquer tecla"
1620 I$=INPUT$(1):PRINT:IF ASC(I$)=17 T
HEN RETURN
1630 ON ERROR GOTO 1660
1640 IF S$="S" THEN BSAVE F$,FH,TH
1650 IF S$="C" THEN BLOAD F$
1660 RETURN
1670 GOSUB 1130:PRINT:IF Q=1 THEN RETUR
N
1680 IF TH<FH THEN PRINT ":"?":RETURN
1690 FOR I=FH TO TH:H$=HEX$(I)
1700 PRINT ":";STRING$(4-LEN(H$),"0");H
$;" ";:H$=HEX$(PEEK(I))
1710 PRINT STRING$(2-LEN(H$),"0");H$;""
";
1720 H$=""
1730 GOSUB 1030:IF Q=1 THEN I=TH:NEXT:P
RINT:RETURN
1740 IF IP=0 AND H$>"" THEN H$=LEFT$(H$
,LEN(H$)-1):PRINT CHR$(127);

```

```
1750 IF IP=13 THEN PRINT IP$:IF H$>"" T  
HEN POKE I,VAL("&h"+H$):NEXT:RETURN ELS  
E NEXT:RETURN  
1760 IF LEN(H$)=2 THEN GOTO 1730  
1770 IF (IP<48 OR IP>57) AND (IP<65 OR  
IP>70) THEN GOTO 1730  
1780 H$=H$+IP$:PRINT IP$::GOTO 1730  
1790 DATA 84,79,83,77,67,65
```

Observações Finais

Quando nos dispusemos a escrever este livro, fizemos questão de incluir apenas jogos que fossem realmente interessantes. Isto naturalmente levou a programas compridos, em que a probabilidade de cometer erros de digitação é relativamente elevada e esses erros, quando ocorrem, não são fáceis de localizar. Os parágrafos seguintes foram escritos com o objetivo de ajudá-lo caso um dos programas, depois de carregado, não funcione ou funcione de forma inesperada.

Conselhos Gerais

1. Peça a um amigo para verificar a listagem com você. Leia a listagem do livro em voz alta e compare-a com a que aparece na tela do computador.
2. Não se esqueça de que, no formato SCREEN Ø, deve haver uma correspondência direta entre as listagens, linha por linha.
3. Cuidado para não confundir a letra I com o algarismo 1.
4. Cuidado para não trocar as letras maiúsculas por minúsculas e vice-versa.
5. Cuidado para não trocar ponto-e-vírgula por dois pontos e vice-versa.
6. Verifique se o número de espaços está correto, usando as linhas vizinhas como referência para contar os espaços.
7. Verifique novamente os números das linhas. Não confunda a continuação de uma linha com uma linha nova.
8. Verifique de novo as linhas de dados. É muito fácil pular um número em uma linha de dados. Para ter certeza de que isto não ocorreu, conte o número de vírgulas e compare-o com o da listagem.
9. Verifique se não se esqueceu de digitar nenhum sinal de aspas.
10. Será que existe mesmo um erro no programa? Torne a ler as instruções do jogo. Muitas vezes, depois que você entra com um dado pelo teclado, o programa não faz nada até a tecla RETURN ser apertada.

Outras Observações

1. Os três programas de jogos de dados (*Corrida de Dados*, *Olhos de Cobra* e *Craps*) só funcionam corretamente quando estão simultaneamente presentes na memória do computador. Por essa razão, é aconselhável salvá-los em fita como se fossem um único programa.
2. Alguns programas param de funcionar quando você entra com um dado diferente do previsto, como uma letra em lugar de um número. Seria possível prever todas as respostas “tolas” e colocar mensagens apropriadas no programa, mas isso aumentaria consideravelmente o tamanho das listagens. Ao escrevermos os jogos, partimos do princípio de que o usuário não estaria a fim de “derrubar” o programa! Assim sendo, se ocorre um erro logo após você entrar com um dado, leia de novo as instruções do jogo para ver se você não está entrando com um dado inapropriado.
3. Todos os programas, com exceção de *Senha*, funcionam igualmente bem em receptores coloridos ou preto-e-branco. Para jogar *Senha* em receptores preto-e-branco, entre com S quando o computador perguntar: “CORES OU SÍMBOLOS (C/S) ?”. Nesse caso, as cores do jogo serão substituídas por símbolos.
4. Antes de carregar um programa novo no computador, desligue-o e torne a ligá-lo. Com isso, evitará que ocorra um erro se o programa anterior tiver modificado o conteúdo de alguma memória da parte de controle do computador.

Como Localizar Erros

Se o programa continua apresentando erro mesmo depois de você seguir as diretrizes acima, o jeito é arregaçar as mangas e partir para uma busca mais direta. Eis alguns conselhos úteis:

1. Se uma mensagem de erro indica que o erro ocorreu em uma determinada linha mas a linha corresponde exatamente à listagem original, o engano pode estar na linha em que é calculado o valor de uma das variáveis usadas na linha em que o erro ocorre. Verifique todas essas linhas.
2. Se a mensagem de erro for mais específica, como “ÍNDICE FORA DO LIMITE” (índice de uma matriz maior que o permitido) ou “FUNÇÃO ILEGAL” (parâmetro incorreto de uma função), procure localizar a variável responsável pelo erro. Em caso de dúvida, use comandos dire-

tos, como “PRINT A”, “PRINT B” etc., para ver na tela qual era o valor das variáveis suspeitas no momento em que ocorreu o erro.

3. Acompanhe o fluxo lógico das instruções do programa. O comando “TRON” mostra a ordem em que as instruções são executadas, revelando assim qualquer erro de endereçamento em instruções como “GOTO”, “GOSUB” etc.

4. Trabalhe de forma metódica. Examine separadamente cada parte do programa. Interrompa a execução quantas vezes for necessário, incluindo instruções “STOP” em pontos estratégicos do programa. Aproveite para observar os valores intermediários das variáveis, usando o comando “PRINT” na forma direta.


5. Suprima temporariamente algumas linhas do programa acrescentando “REM” no início da linha. Verifique de que forma a execução é afetada.

6. Torne a digitar as linhas suspeitas! Em alguns casos, a linha aparece corretamente na tela, mas foi guardada de forma truncada na memória do computador.







7. Use as instruções “ERROR” e “ON ERROR GOTO”, da forma descrita no manual do computador.

8. Vá para a cama. Depois de uma boa noite de sono, será bem mais fácil localizar o erro!

Apêndice 1: Gráficos para Desenhar Sprites e Caracteres



DEFINIÇÕES



Use uma cópia Xerox desta página para não sujar o livro.

Apêndice 2: Gráfico das Coordenadas da Tela

	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16	17	18	19	20	21	22	23	24	25	26	27	28	29	30	31	32	33	34	35	36	37	38	39	40
1																																								
2																																								
3																																								
4																																								
5																																								
6																																								
7																																								
8																																								
9																																								
10																																								
11																																								
12																																								
13																																								
14																																								
15																																								
16																																								
17																																								
18																																								
19																																								
20																																								
21																																								
22																																								
23																																								
24																																								

	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16	17	18	19	20	21	22	23	24	25	26	27	28	29	30	31	32	33	34	35	36	37	38	39	40
1																																								
2																																								
3																																								
4																																								
5																																								
6																																								
7																																								
8																																								
9																																								
10																																								
11																																								
12																																								
13																																								
14																																								
15																																								
16																																								
17																																								
18																																								
19																																								
20																																								
21																																								
22																																								
23																																								
24																																								

SCREEN 0 - Texto - 40 x 24 caracteres

SCREEN 1 - Texto - 32 x 24 caracteres

SCREEN 2 - Gráficos de Alta Resolução - 256 x 192 pontos

SCREEN 3 - Gráficos de Baixa Resolução - 64 x 48 pontos

Use uma cópia Xerox desta página para não sujar o livro.

Apêndice 3: Mapa da Memória do MSX

ÁREA DOS PROGRAMAS DO USUÁRIO	ÁREA DE TRABALHO DO BASIC CONTROLE DE E/S ARMAZENAMENTO DE STRINGS INICIALMENTE 200 USE CLEAR PARA MUDAR PILHA ÁREA LIVRE ARMAZENAMENTO DE MATRIZES ARMAZENAMENTO DE VARIÁVEIS TEXTO DE PROGRAMAS EM BASIC	& HFFF & HF380 & H8000 & H0000
	BASIC MSX (32K)	

Apêndice 4: Tabela de Cores do MSX

CORES		CORES	
0	Transparente	8	Vermelho
1	Preto	9	Vermelho claro
2	Verde	10	Amarelo
3	Verde claro	11	Amarelo claro
4	Azul escuro	12	Verde escuro
5	Azul claro	13	Roxo
6	Vermelho escuro	14	Cinza
7	Azul celeste	15	Branco

Impressão e Acabamento
GRAPHOS
Rua Santo Cristo, 70/78 — Tel.: 223-1272

Aqui está uma coleção de jogos interessantes e divertidos para os computadores da família MSX. Há programas para todos os gostos: jogos de aventura, como o *Marie Celeste*, baseado na lenda do navio fantasma; jogos clássicos, como *Star Trek* e *Super Breakout*; jogos de estratégia, como damas e xadrez; jogos de dados, como *Craps*; jogos de raciocínio, jogos de fliperama e jogos educativos. O livro inclui também uma série de programas utilitários que ajudarão o leitor a criar desenhos originais na tela e a desenvolver seus próprios programas, além de um mapa da memória e gráficos das coordenadas de tela e para a geração de caracteres. Esta obra se destina a todos os usuários do MSX, desde os principiantes até programadores experientes.

Todos os jogos foram escritos na linguagem BASIC e cuidadosamente testados.

O Autor

Jim Gregory é Diretor-Gerente da Mr. Micro, uma das maiores *software houses* da Inglaterra, especializada em jogos para microcomputadores. Já escreveu vários artigos para as revistas *Personal Computer World*, *Leisure and Electronics Trader* e *Computer and Video Games*.

PROGRAMAS DE JOGOS SUPERINTERESSANTES PARA O SEU MICROCOMPUTADOR!

PROGRAMAS
DE JOGOS DE TERROR
Jenny Tyler e Chris Oxlade

PROGRAMAS
DE JOGOS DE GUERRA
Daniel Isaaman e Jenny Tyler

PROGRAMAS
DE JOGOS DE ESPIONAGEM
Jenny Tyler e Chris Oxlade

PROGRAMAS
DE JOGOS ESPACIAIS
Jenny Tyler

PROGRAMAS
DE JOGOS DE HORROR
Jenny Tyler